

# 微电子器件与工艺模拟实验讲义

高云、杨维明、叶葱、汪宝元

湖北大学物理学与电子技术学院

## 实验目录

<b>Experiment 1.</b>	<b>Thin Film Resistor: Creating an Thin Film Resistor Using ATHENA</b>	<b>2</b>
<b>Experiment 2.</b>	<b>Zener Diode</b>	<b>10</b>
<b>Experiment 3.</b>	<b>MS Junction: Creating a M-S Junctiong Using ATLAS</b>	<b>12</b>
<b>Experiment 4.</b>	<b>NMOS Device: Creating a NMOS Device using ATHENA</b>	<b>15</b>
<b>Experiment 5.</b>	<b>NMOS Device: NMOS Device simulation using ATLAS</b>	<b>30</b>
<b>Experiment 6.</b>	<b>NMOS Devise: Creating a NMOS Devise using DEVEDIT</b>	<b>39</b>
<b>Experiment 7.</b>	<b>MESFET Device</b>	<b>50</b>
<b>Experiment 8.</b>	<b>BJT Device</b>	<b>66</b>
<b>Experiment 9.</b>	<b>Solar Cell</b>	<b>79</b>
<b>Experiment 10.</b>	<b>TFT</b>	<b>92</b>
 <b>Appendix A:</b>	 <b>ATLAS 中常用 STATEMENTS 及其部分参数</b>	 <b>100</b>
<b>Appendix B:</b>	<b>IC 制造工艺简述</b>	<b>107</b>


## Experiment 1. Thin Film Resistor: Creating an Thin Film Resistor Using ATHENA

基础知识

e.g. p 型 Si 上制作一个 n+ poly-Si 型电阻

根据电阻公式:

$$R = r \cdot \frac{l}{S} = r \cdot \frac{l}{hw} \quad (1.1)$$



进一步,

$$R = \left(r \cdot \frac{1}{h}\right) \cdot \frac{l}{w} = R_0 \cdot \frac{l}{w} \quad (1.2)$$

其中,  $R_0$  称作方块电阻. 也就是说, 薄膜电阻为方块电阻乘以长宽比.

控制电阻材料的电阻率和材料厚度, 可以得到适度的方块电阻, 在平面工艺中, 控制长宽比, 就可以得到需要的电阻阻值.

目标结构




图 1.1 完整的薄膜电阻结构

结构参数

衬底:  $0.6\mu\text{m} \times 0.8\mu\text{m}$ , Si, p 型, 浓度  $1 \times 10^{14} \text{cm}^{-3}$

氧化层: 厚度  $0.04\mu\text{m}$

多晶硅层: 厚度 0.2um, n 型, 浓度  $1 \times 10^{15} \text{cm}^{-3}$

模拟程序

## 1. 定义网格

a. 在 DECKBUILD 下运行 ATHENA.

b. 在文本窗口输入如下语句:

**go athena**

c. 打开网格定义窗口: 右键点击 Commands—>选 Mesh Define...

d. 在打开的 Mesh Define 界面, Direction 一栏默认为 X, Location 一栏输入 0.00 (输入后切记记得按回车键. 后面的也是如此. 不再赘述.), Spacing 一栏输入 0.10, Comment 一栏输入 Non-uniform Grid (0.6um×0.8um), 如下图:




图 1.2 定义网格参数

e. 单击 Insert 按钮.

f. 重复上面两步, 实现 Location=0.20 时 Spacing=0.01; Location=0.60 时 Spacing=0.01.

g. 现在设置 Y 方向上的网格, 在 Direction 一栏单击 Y. 其他步骤同 d, e 步骤, 实现如下设置:




图 1.3 Y 方向网格定义

- h.单击 View...按钮,即可看到网格分布窗口.(注意总共产生了 1786 个格点和 3404 个三角形.)
- i.单击 WRITE.在 DECKBUILD 中看到的语句将如下所示:

```
go athena
#Non-uniform Grid (0.6um×0.8um)
line x loc=0.00 spac=0.10
line x loc=0.20 spac=0.01
line x loc=0.60 spac=0.01
#
line y loc=0.00 spac=0.008
line y loc=0.2 spac=0.01
line y loc=0.5 spac=0.05
line y loc=0.8 spac=0.15
```

## 2. 定义衬底

- a.打开 ATHENA Mesh Initialize 菜单:右键点击 Commands—>Mesh Initialize.
- b.在弹出的窗口作出如下选择:

Material: Silicon  
Orientation: 100  
Impurity: Boron  
Concentration: By Concentration  
1.0e14  
Dimensionality: 2D  
Grid scaling factor: 1.0  
Comment: Initial Silicon

## 3. 生成氧化层

接下来,我们将要在硅表面上在 **1000℃**、**3%HCL**、**1 个大气压**条件下采用**干氧法**持续 **30 分钟**成长一层氧化层.

先后选择 “Commands—>Process—>Diffuse” 打开 ATHENA Diffuse 菜单.选择或输入以下:

Time(minutes): 30  
Temperature(C):1000  
Ambient: Dry O2  
Gas pressure (atm): 1.0  
Hcl %:3  
Comment: Gate Oxidation

## 4. 提取氧化层厚度

依次选择 “Commands—>Extract” 进入 ATHENA Extract 菜单.选择或输入以下:

Extract: Material thickness  
Name: Oxidethick  
Material: SiO~2  
Extract location: X=0.3




图 1.4 高亮氧化层厚度语句

提取氧化硅的厚度:419.388 埃.

## 5. 优化氧化层厚度.

先后单击“Main Control—>Optimizer”即进入 DECKBUILD Optimizer 界面.

在 Mode 为 Setup 下,选择最大误差为 1%——“Maximum error (%)=1”(Stop criteria 一栏下).

改 Mode 为 Parameters.在本例中,栅氧厚度优化参数选择温度和压强.为此,我们得回到 DECKBUILD 中并高亮“diffuse time=30 temp=1000 dryo2 press=1.00 hcl.pc=3”(如下图).然后再回到 Optimizer,依次选择“Edit—>Add”,选 temp 和 press 项,单击 Apply.




图 1.5 高亮扩散语句

改 Mode 为 Targets.Optimizer 将用 DECKBUILD 中 Extract 语句定义优化目标.因此,再次回到 DECKBUILD 文本窗口并高亮 extract 一句.在再回到 Optimizer 时,选“Edit—>Add”,在 Target value 栏输入目标值“400”.





图 1.6 高亮优化目标

接着 Mode 选择 Graphics, 点击 Optimize, 即可从图中看见优化的全过程:



### 1.7 优化过程

接着, Mode 选择 parameters 点击 Edit 菜单选择“copy to Deck”, 优化后的温度和气压值将拷贝至文本窗口中。

## 6. 沉积多晶硅层

接下来,用共形沉积法沉积 0.2 微米厚的多晶硅.步骤如下:

- 首先打开 ATHENA Deposit 菜单——依次选 “Process—>Deposit—>Deposit...” 即可。

b.选择或输入以下:

Type: Conformal

Thickness (um):0.2

Total number of grid layers: 10

## 7. 离子注入

a.选 “Commands—>Process—>Implant...” 打开 ATHENA Implant 菜单.

b.选择或输入以下:

Impurity: Arsenic

Dose (ions/cm2): 1.0e15

Energy (KeV): 40

Model: Dual Pealson

Tilt (degrees): 7

Rotation (degrees): 30

Material type: Crystalline

## 8. 退火处理

先后选择 “Commands—>Process—>Diffuse” 打开 ATHENA Diffuse 菜单.在时间、温度、环境气氛、气压、Diffusion 模型栏分别输入或选择 2、1000、氮气、1、Fermi.

## 9. 提取方块电阻阻值

a.打开 ATHENA Extract 菜单——依次选择 “Commnds—>Extract” 即可.

b.选择提取参量: Sheet resistance.其他变量选择如下:




图 1.8 提取方块电阻时的参数选择

c.单击 WRITE 及 Cont.可以得出方块电阻为 77.8174ohm/square.(如下图所示.)






图 1.9 提取出方块电阻值的语句显示

## 10. 刻蚀出多晶形状

a.刻蚀掉  $X=0.15$  左右的多晶硅.首先打开 ATHENA Etch 菜单,然后选择或输入以下:

Etch Method: Geometrical

Geometrical type: Left

Material location (um): 0.15

Comment: Polysilicon Definition

b.刻蚀掉  $x=0.45$  左右的多晶硅.方法同上.

刻蚀多晶硅后的图形如图 1-10 所示.




图 1.10 刻蚀多晶硅后的结构图




图 1.11 刻蚀氧化层后的结构图

## 11. 氧化多晶硅

在打开的 ATHENA Diffuse 菜单中,使用或修改的参数如下(在以下的步骤中,略去打开 ATHENA 扩散、氧化、刻蚀等菜单的过程):

Time (minutes) : 3

Temperature (c): 900

Ambient: Wet O2

Gas pressure (atm): 1.00

Diffusion: fermi  
Oxidation: Compressible  
Comment: Polysilicon Oxidation

## 12. 刻蚀出电极接触孔

在 ATHENA Etch 菜单,使用或修改的参数如下:

Etch Method: Geometrical  
Geometrical type: Any shape  
Material: Oxide  
Arbitrary points: x=0.25 y=-0.25  
                                  x=0.35 y=-0.25  
                                  x=0.35 y=-0.2  
                                  x=0.25 y=-0.2  
Comment: Polysilicon Definition

## 13. 沉积金属电极

打开的 ATHENA Deposit 菜单,选择或修改的参数如下:

Type: Conformal  
Material: Aluminum  
Thickness (um): 0.05  
Total number of grid layers: 10




图 1.12 沉积铝后的结构图




图 1.13 完整的薄膜电阻的结构

## 14. 刻蚀出电极的形状

a.刻蚀掉 X=0.2 以左的铝.打开 ATHENA Etch 菜单,然后选择或输入以下:

Etch Method: Geometrical  
Geometrical type: Left  
Material: Aluminum  
Etch location (um): 0.2

b.刻蚀掉 X=0.4 以右的铝.方法同上.结果如图 1.13 所示

## Experiment 2. Zener Diode

### 基础知识

**齐纳二极管**,又称稳压二极管,是一种硅材料制成的**面接触型晶体二极管**,简称稳压管.稳压管与普通二极管有着相类似的伏安特性——其正向特性为指数曲线;所不同的是,当稳压管外加反向电压大到一定程度时则击穿,击穿区的曲线很陡,几乎平行于纵轴,表现出很好的稳压特性,因而广泛应用于稳压电源与限幅电路之中.

### 模拟程序

```
#
go athena
# 网格定义
line x loc=0.00 spac=0.2
line x loc=1 spac=0.1
line x loc=1.1 spac=0.02
line x loc=2 spac=0.25
#
line y loc=0.00 spac=0.02
line y loc=0.2 spac=0.1
line y loc=0.4 spac=0.02
line y loc=2 spac=0.5

# 衬底定义
init silicon c.phos=5.0e18 orientation=100

# 形成氧化层
deposit oxide thick=0.50 divisions=5
etch oxide left p1.x=1

# 离子注入与退火
implant boron dose=1.0e15 energy=50 pearson tilt=7 rotation=0 amorph
method fermi compress
diffus time=30 temp=1000 nitro press=1.00

# 提取结深和方块电阻
extract name="xj" xj material="Silicon" mat.occno=1 x.val=0 junc.occno=1
extract name="rho" sheet.res material="Silicon" mat.occno=1 x.val=0 region.occno=1

# 氧化层刻蚀
etch oxide all

# 形成 Al 接触
deposit alum thickness=0.2 div=3
```

```
etch alum right p1.x=1.0
```

```
# 电极定义
```

```
electrode name=anode x=0.0
```

```
electrode name=cathode backside
```

```
# 结构保存&绘画
```

```
structure outf=diodeex05_0.str
```

```
tonyplot diodeex05_0.str -set diodeex05_0.set
```

```
go atlas
```

```
# 模型定义
```

```
models bipolar bbt.std print # print 是指将模型参数打印出来
```

```
impact selb # 用于击穿分析的 Selberherr 碰撞电离模型.
```

```
# 数值方法定义
```

```
method newton trap maxtrap=10 climit=1e-4
```

```
solve init #初值
```

```
# 保存&绘画 IV 图
```

```
log outf=diodeex05.log
```

```
solve vanode=-0.25 vstep=-0.25 vfinal=-10 name=anode
```

```
tonyplot diodeex05.log -set diodeex05_log.set
```

```
# 从 IV 图中提取器件参数
```

```
extract init infile="diodeex05.log"
```

```
extract name="bv" min(v."anode")
```

```
extract name="leakage" x.val from curve(v."anode",abs(i."anode")) where y.val=1e-10
```

```
# 退出模拟
```

```
quit
```

模拟结果

xj=0.437199 um from top of first Silicon layer

X.val=0

rho=171.057 ohm/square X.val=0

bv=-10

leakage=-5.68681

IV 特性如图 2.1.




图 2.1 齐纳二极管 IV 曲线图

## Experiment 3. MS Junction: Creating a M-S Junctiong Using ATLAS

基础知识

金半(金属-半导体)接触基于两种模型:肖特基势垒和欧姆接触.肖特基势垒指一具有大的势垒高度(也就是 $\phi_{Bn}$ 或 $\phi_{Bp} \gg kT$ ),以及掺杂浓度比导带或价带上态密度低的金属-半导体接触.肖特基二极管是基于肖特基势垒的二极管.

目标参数

尺寸:12um×5um

结构:p(1e19)n(5e16)n+(1e20)

模拟程序

# 网格定义

go atlas

mesh space.mult=1.0

#

x.mesh loc=0.00 spac=0.5

x.mesh loc=3.00 spac=0.2

x.mesh loc=5.00 spac=0.25

x.mesh loc=7.00 spac=0.25

x.mesh loc=9.00 spac=0.2

x.mesh loc=12.00 spac=0.5

#

y.mesh loc=0.00 spac=0.1

y.mesh loc=1.00 spac=0.1

y.mesh loc=2.00 spac=0.2

y.mesh loc=5.00 spac=0.4

# 区域定义

region num=1 silicon

# 电极定义

electr name=anode x.min=5 length=2

electr name=cathode bot

# 形成 N 型 Si (N-epi doping )

doping n.type conc=5e16 uniform

# 形成 P 型 Si 保护环 (Guardring doping )

doping p.type conc=1e19 x.min=0 x.max=3 junc=1 rat=0.6 gauss

doping p.type conc=1e19 x.min=9 x.max=12 junc=1 rat=0.6 gauss

```
# 形成 N+背面场
doping      n.type  conc=1e20  x.min=0
x.max=12 y.top=2 y.bottom=5 uniform
```

```
# 保存&绘画二极管结构
save outf=Schottky_Diode.str
tonyplot Schottky_Diode.str
```

```
# 计算模型定义
model conmob fldmob srh auger bgn
contact name=anode workf=4.97
```

```
# 提供初值
solve init
```

```
# 数值方法定义
method newton
```




图 3.1 肖特基二极管结构图

```
# 绘画 IV 曲线
log outfile=SD_IV.log
solve vanode=0.05 vstep=0.05 vfinal=1 name=anode
tonyplot SD_IV.log
```

```
# 求能带图
#get energy band
solve init
output qfn qfp con.band val.band
save outfile=SD_band.str
tonyplot SD_band.str
```

```
# 退出
quit
```

运行结束之后,就可以得到二极管结构图 Schottky\_Diode.str,IV 曲线图 SD\_IV.log,以及其能带图 SD\_band.str.

#### 结构图:

在打开的 Schottky\_Diode.str 中,在图中任一处通过依次选择“鼠标右键—>display—>show mesh 图标 (第一个图标,表示网格分布)—>Apply”可观察网格分布.

Display 中常用其他图标:第四个——show contours(默认的是净掺杂浓度分布);倒数第四个——show junctions(显示结的位置);倒数第三个——show electrodes(显示电极的位置).

选择第四个图标后,得到一个颜色鲜艳的图像,如图 3.2 所示.其中不同颜色代表不同掺杂浓度,左下方为其数值.

在菜单栏中依次选择“Tools—>Cutline”按照需要观察不同 LINE 的分布变化,现在用最后一

个图标,用键盘敲入画出点(2,0)到点(2,5),以及点(6,0)点到(6,5).最后结果如图 3.1.




图 3.2 肖特基二极管的掺杂分布图

#### IV 曲线图:

在得到的 IV 曲线图中,通过依次选择“鼠标右键—>display—>Y Quantities—>Anode Current”将 y 轴坐标换成正极电流,应用后就可以看到二极管的 IV 曲线如图.




图 7.3 肖特基二极管的 IV 曲线图




图 7.4 肖特基二极管的能带图

#### 能带图:

通过依次单击“鼠标右键—>display—>轮廓图标(第四个)—>右键 Define—>Selected—

>Quantity—>Conduction Band Energy—>Apply” 可以看到二极管的到导带能量图.然后用 “Tools—>Cutline” 画线(6,0)至(6,5),在得到的图中再通过 “右键—>Display—>Valence Band Energy” 最后可得而二极管的能带图如图.

## Experiment 4. NMOS Device: Create an NMOS device using ATHENA

### 4.1 基本说明:

在本章中,用户将学习到创建一个典型 MOSFET 结构的基本操作.这些操作包括:

- ◇ 创建一个好的模拟网格
- ◇ 保形沉积( conformal deposition)
- ◇ 几何刻蚀
- ◇ 氧化,扩散,退火以及离子注入
- ◇ 结构操作
- ◇ 保存与下载结构

所有这些操作,都单独在 ATHENA 工艺模拟器中进行.

### 4.2 创建一个初始结构

#### 4.2.1 定义初始矩形网格

- a.清理当前文本窗口:右键点击 File—>选 Empty Document.
- b.把 ATHENA 作为模拟器——在文本窗口输入如下语句:

**go athena**

接下来,我们将定义初始的矩形网格.需要注意的是,正确的定义网格在工艺模拟中至关重要.格点数直接影响到器件的精度和时间.一个精细的网格应该出现在关键区域,如离子注入的地方,pn 结形成的区域,或者光模拟中的光敏部分.

- c.打开网格定义窗口:右键点击 Commands—>选 Mesh Define...

#### 4.2.2 创建一个 0.6um\*0.8um 模拟区域及一个非均匀网格

在打开的 Mesh Define 界面,进行如下操作:


- a.Direction: X(默认项).
- b.单击 Comment 输入 “Non-Uniform Grid (0.6um\*0.8um)” .
- c.单击 Location,输入 0;单击 Spacing,输入 0.1.
- d.单击 Insert.
- e.采用同样的方法设置 x=0.2 时 spac=0.01;x=0.6 时 spac=0.01.
- f.创建 Y 方向上的网格:在 Direction 一栏单击 Y,设置 y=0.00 时 spac=0.008;y=0.2 时 spac=0.01;y=0.5 时 spac=0.05;y=0.8 时 spac=0.15.

我们将注意到网格最密的地方是在表面(y=0um 至 y=0.2um 的地方).因为在后面这个区域将形成 NMOS 的表面有源区.

- g.预览矩形网格:单击 View....(注意总共产生了 1786 个格点和 3404 个三角形.)

- h.写入文本窗口:单击 WRITE.





### 4.3 定义衬底

a. 打开 ATHENA Mesh Initialize 菜单: 右键点击 Commands—>Mesh Initialize.

b. 在弹出的窗口作出如下选择:

Material: Silicon

Orientation: 100

Impurity: Boron

Concentration: By Concentration, 1.0, 14

Dimensionality: 2D

Comment: Initial Silicon Structure with <100> Orientation (可选项, 建议新手使用,)

这样一来, 一个尺寸  $0.6\mu\text{m} \times 0.8\mu\text{m}$ 、均匀掺杂硼  $1 \times 10^{14} \text{atom/cm}^3$ 、(100)方向的硅衬底就创建出来了。(Si (100) 衬底: 具有低界面电荷、低界面缺陷的优点, CMOS 集成电路一般采用(100)晶向的硅材料)

c. 单击 WRITE

d. 单击 RUN

DECKBUILD 自动产生一临时结构文件 struct outfile=.history01.str. 采取以下方法还可查看其具体结构: 鼠标高亮 “.history01.str” —> 右键点击 Tools—> 选择 Plot 及 Plot Structure, 稍等一会儿, 就可看到结构图. 默认情况下, 它只显示 regional 和 material 信息.

如想查看网格或其它信息, 右键点击 plot—>Display, 选择第一个图标 (Mesh) 然后点击 Apply 即可看到其网格分布. Tonyplot: Display (2D Mesh) 窗口其他图标分别表示: Edges、Regions (or Material)、Contour、Light、Vectors、Junctions、Electrodes、Threed、Draw (?).

### 4.4 生成栅氧层

接下来, 我们将要在硅表面上在  $950^\circ\text{C}$ 、3% HCL、1 个大气压条件下采用干氧法持续 11 分钟成长一层栅氧层.

先后选择 “Commands—>Process—>Diffuse” 打开 ATHENA Diffuse 菜单. 在时间、温度、环境栏分别输入或选择 11、950、Dry O2. (注意: 输入数字后均需按回车键.) 选择 Gas Pressure (atm) 和 HCL%, 输入预先设定好的值 1 和 3. Comment 一栏输入 “gate Oxidation”. 然后写入 DECKBUILD.

单击 **Cont** 按钮继续模拟. 之后将生成 str 文件.

接着绘画 str 文件.history02.str——高亮 “.history02.str” 并选择 Plot 及 Plot Structure 即可看到最新模拟结构.

### 4.5 提取栅氧层厚度

下来将提取在氧化过程中栅氧层的厚度. 将使用到 DECKBUILD 中 Extract 语句.

Extract 可非常快速容易地整合和控制某值或曲线. 你可以创建你自己定制的表情式或选择工

艺和器件模拟器提供的标准程序.也即,你可选一标准的表达式并修改使之满足自己的需要.

依次选择“Commands—>Extract”进入 ATHENA Extract 菜单.选择或输入以下:

Extract: Material thickness

Name: Gateoxide

Material: SiO~2

Extract location: X

Value: 0.3

单击 WRITE 及 Cont 按钮继续模拟,将可看到提取的栅氧层厚度.厚度约在 131 埃.

在一步我们将用 DECKBUILD 的 **Optimizer** 对栅氧层的厚度进行优化.

## 4.6 优化栅氧层厚度

在本节我们将学习怎样使用 **Optimizer** 对栅氧层工艺参数进行优化.假设栅氧层目标厚度为 100 埃.为了达到此目的,扩散温度和气压均需调节.

先后单击“Main Control—>Optimizer”即进入 DECKBUILD Optimizer 界面.

在 Mode 为 Setup 下,选择最大误差为 1%——“Maximum error (%)=1”(Stop criteria 一栏下).

改 Mode 为 Parameters.在本例中,栅氧厚度优化参数选择温度和压强.为此,我们得回到 DECKBUILD 中并高亮“diffuse time=11 temp=950 dryo2 press=1.00 hcl.pc=3”(如图 2.1).然后再回到 Optimizer,依次选择“Edit—>Add”,选 temp 和 press 项,单击 Apply.

改 Mode 为 Targets.Optimizer 将用 DECKBUILD 中 Extract 语句定义优化目标.因此,再次回到 DECKBUILD 文本窗口并高亮 extract 一句(如图 2.2).在再回到 Optimizer 时,选“Edit—>Add”,在 Target value 栏输入目标值“100”.

为了监视整个优化过程,在 Mode 栏选 Graphics 并单击 Optimize.

整个优化过程将不断调整温度和压强以最终达到目标值.

最终 Optimizer 将收敛于温度、压强分别约为 926℃、0.98 时,提取的栅氧厚度为 100.209 埃.接着我们将这个值复制到 DECKBUILD 中以得到目标值——改 Mode 为 Parameters,选“Edit—>Copy to Deck”即可.




图 4.1 高亮 diffuse 一句




图 4.2 高亮 extract 一句

## 4.7 离子注入

离子注入是半导体掺杂的主要方法.因为离子注入在小的关键尺寸(critical dimension,CD)和浅掺分布,高剂量,斜注入(tilted implant)和其他先进技术有着广泛应用,因此,对例子注入过程的模拟就显得非常重要了.

在本例中,我们将采用一种注入硼  $9.5 \times 10^{11} \text{cm}^{-2}$ 、能量 10KeV、离子倾斜角度为  $7^\circ$ 、旋转  $30^\circ$  的阈值电压调整注入.

具体过程如下:

a.选 “Commands—>Process—>Implant...” 打开 ATHENA Implant 菜单.

b.选择或输入以下:

impurity: Boron

Dose:  $9.5 \times 10^{11}$

Energy: 10

Model: Daul Pearson

Tilt: 7

Rotation: 30

Material Type: Crystalline

Commnt: Threshold Voltage Adjust implant

c.单击 WRITE 及 Cont

## 4.8 用 TONYPLOT 分析硼杂质分布

a.用 TONYPLOT 绘画 str 文件 “.history05.str”.接着选 “Plot—>Display...”

b.单击 Contours 图标(第四个).

c.依次选择 “Define—>Contours...” ,Quantity 选 Boron.

d.单击 Apply.

硼杂质的分布图如图所示




图 4.3 离子注入后硼杂质浓度分布图

接下来,我们用二维结构图上 cutline 工具创建一个硼杂质分布的一维横截图.

- 选 “Tools—>Cutline” 打开 Cutline 菜单.
- 选第一个图标(默认选项).
- 鼠标左键点击氧化层上方并拖拉至结构的底部.硼杂质一维分布图就如图所示.




图 4.4 在结构上画垂直横截线

## 4.9 共形沉积(conformal deposition)多晶硅

共形沉积可用来产生多层结构.共形沉积是最简单的沉积模型,可用于对沉积层形状并无特别要求的任意情形.

NMOS 工艺中多晶硅厚度为 2000 埃.

a.首先打开 ATHENA Deposit 菜单——依次选 “Process—>Deposit—>Deposit...” 即可.

b.选择或输入以下:

Type:conformal (默认选项)

Material: Polysilicon

Thickness: 0.2

Total number of grid layers: 10

Comment: Conformal Polysilicon Deposition

c.单击 WRITE 及 Cont

d.绘画结构.结构如下.




图 4.5 多晶硅的共形沉积

## 4.10 简单几何刻蚀

接下来是多晶硅栅极的定义.在本例中,多晶硅栅极边缘在  $x=0.35\mu\text{m}$  处,中心在  $x=0.6\mu\text{m}$  处.因此,我们将刻蚀  $x=0.35\mu\text{m}$  以左的区域.

a.依次选择 “Commands—>Process—>Etch—>Etch...” 打开 ATHENA Etch 菜单.

b.选择或输入以下:

Etch Method: Geometrical (默认选项)

- Geometrical type: Left
  - Material: Polysilicon
  - Etch location (um): 0.35
  - Comment: Polysilicon Definition
  - c.单击 WRITE 及 Cont.
- 得到的结构图如下.

图 4.6 刻蚀多晶硅以形成栅极

## 4.11 多晶硅氧化

下一步是多晶硅氧化以为多晶硅离子注入掺杂作准备.氧化条件为在 **900℃,1 个大气压**下用**湿氧法**进行 **3 分钟**.

因为氧化是在未损伤的图案化的(非平面的)多晶硅上进行,所以产用的方法为 fermi 和 compress 方法.fermi 方法是用在掺杂浓度小于  $1 \times 10^{20} \text{cm}^{-3}$  的未损伤的衬底而 compress 方法用于模拟非平面结构氧化及二维氧化.

- a.首先打开 ATHENA Diffuse 菜单:依次选择 “Commands—>Process—>Diffuse...” 即可.

- b.选择和输入以下:

Time: 3  
 Temperature: 900  
 Ambient: Wet O2  
 Gas pressure: 1  
 HCL(uncheck)  
 Diffusion Models: Fermi  
 Oxidation Models: Compressible  
 Comment: Polysilicon Oxidation

- c.单击 WRITE 及 Cont
- 得到的结构如图.




图 4.7 多晶硅栅的氧化

## 4.12 多晶硅掺杂

多晶硅氧化后,下一步就是用磷掺杂多晶硅以形成 n+多晶硅栅.在这里,磷剂量为  $3 \times 10^{13} \text{cm}^{-2}$ ,注入能量为 20KeV.

- a.首先打开 ATHENA Implant 菜单:依次选择 “Commands—>Process—>Implant...” 即可.

- b.选择和输入以下:

Impurity: Phosphorus  
 Dose:  $3 \times 10^{13}$

Energy: 20

Tilt: 7

Rotation: 30

Model: Dual Pearson

Material Type: Crystalline

Comment: Polysilicon Doping

c.单击 WRITE 及 Cont

d.绘画结构,并观察其净掺杂分布——“Display (2D Mesh)”—>Contours icon (第四个) —>Apply”.

e.观察磷分布——先打开 Contours 窗口:“Display (2D Mesh)”—>Contours icon—>Apply—>Define (右键)—>Contours...” ,然后选 Quantity 为 Phosphorus,最后单击 Apply 及 Dismiss.




图 4.8 多晶硅离子注入后净掺杂分布

磷杂质浓度分布如下图所示.






图 4.9 磷杂质浓度分布

### 4.13 隔离氧化层的淀积 (Spacer Oxide Deposition)

源、漏离子注入之前必须进行隔氧淀积。本例中隔氧淀积厚度为 0.12  $\mu\text{m}$ 。

a. 首先打开 ATHENA Deposit 菜单: 依次选择 “Commands —> Process —> Deposit —> Deposit...” 即可。

b. 选择和输入以下:

Material: Oxide

Thickness: 0.12

Total number of grid layer: 10

Comment: Spacer Oxide Deposit

c. 单击 WRITE 及 Cont




图 4.10 隔

氧淀积后的结构

### 4.14 侧墙 (Sidewall Spacer) 氧化隔离的形成

a. 打开 ATHENA Etch 菜单



b.选择或输入以下:

Material: Oxide


Thickness: 0.12

Comment: Spacer Oxide Etch

c.单击 WRITE 及 Cont

图

2.11 干刻形成边墙隔氧



## 4.15 源/漏极注入与退火

为了形成 NMOS 的 n+源/漏,在本例中,采用的是**砷**,其剂量为  $5 \times 10^{15} \text{cm}^{-3}$ ,注入能量为 **50KeV**:

a.打开 ATHENA Implant 菜单

b.选择或输入以下:

Implant: Arsenic

Dose:  $5 \times 10^{15}$

Energy: 50

Tilt: 7

Rotation: 30

Material Type: Crystalline

Comment: Source/Drain Implant

c.单击 WRITE

接下来在 **900℃**、**1 个大气压**、**氮气**氛围下退火 **1 分钟**:

a.打开 ATHENA Diffuse 菜单

b.选择或输入以下:

Time: 1

Temperature: 900

Ambient: Nitrogen

Gas Pressure: 1

Diffusion Models: Fermi

Oxidation Models(uncheck)

Comment: Source/Drain Annealing

c.单击 WRITE 及 Cont

用 TONYPLOT 绘画该结构的掺杂分布如图 4.12.

图 4.12 源/漏离子注入与退火后净掺杂分布

下一步,我们想观察退后前后净掺杂分布.为了达到此目的,

a.在上图窗口中,依次选“File—>Load Structure...”,在 filename 一项键入“.history12.str”,随后右键点击 Load 并选 Overlay.这样一来,“.history12.str”将覆盖于“.history12.str”之上.

b.依次单击“Tools—>Cutline...”打开 Cutline 窗口.

c.选择键盘图标,输入如下 X、Y 值:

Start X: 0.1 Y: -0.05

End X: 0.1 Y: 0.2

d.按回车键,在弹出的信息确认框中单击 Confirm 按钮即可看到退火前后一维浓度分布图 4.13.




图 4.13 退火前后净掺杂浓度分布

## 4.16 金属的淀积

Athena 中的电极可以是任意金属、硅化物或多晶硅区域.唯一的特例是背电极可以安置在结构的底部而不需考虑那里是否存在金属.在本例中,NMOS 的金属化是先形成源/漏的接触窗口,然后沉积 Al 和 Al 图案化.

为了形成源/漏的接触窗口,x=0.2um 左右的氧化层将被刻蚀掉.

a.打开 ATHENA Etch 菜单

b.选择或输入以下:

Geometrical type: Left

Material: Oxide

Etch location: 0.2

Comment: Open Contact Window

c.单击 WRITE 及 Cont

用 TONYPLOT 绘画其结构图如图 4.14 所示。

图 4.14 金属化前形成接触窗口

接下来用 ATHENA Deposit 沉积 0.03um 厚的 Al.

- a. 首先打开 ATHENA Deposit 菜单.
- b. 选择或输入以下:  
Material: Aluminum  
Thickness: 0.03  
Total number of grid layer: 2  
Comment: Aluminum Deposition
- c. 单击 WRITE 及 Cont




图 4.15 Half NMOS 结构的铝沉积  
最后用 Etch 刻蚀 x=0.18um 以右的铝层:

- a. 打开 ATHENA Etch 菜单.
- b. 选择或输入以下:  
Geometrical type: Right  
Material: Aluminum  
Etch location: 0.18  
Comment: Etch Aluminum
- c. 单击 WRITE 及 Cont




图 4.16 Half NMOS 结构的铝刻蚀

## 4.17 提取器件参数

在本节中,我们将提取 Half NMOS 结构的一些器件参数.这些参数包括:结深、n++源/漏方块电阻、隔氧下 LDD 方块电阻、长沟道阈值电压.所有这些参数,均可用 DECKBUILD 中的 Extract 提取.

### 4.17.1 提取结深

a.打开 ATHENA Extract 菜单——依次选择“Commnds—>Extract”即可.

b.选择或输入以下:

Extract: Junction depth

Name: nxj

Material: Silicon

Extract location: X=0.2

c.单击 WRITE.

文件窗口将出现:

```
extract name="nxj" xj material="Silicon" mat.occno=1 x.val=0.2 junc.occno=1
```

在该 **extract** 语句中,**name="nxj"**是 n 型源/漏结深;**xj** 表明结深将要被提取;**material="Silicon"** 指包含结的物质为硅;**mat.occno=1** 表明提取结深的位置为第一层硅处;**x.val=0.2** 指提取源/漏结深于  $x=0.2\mu\text{m}$  处;**junc.occno=1** 表明提取结深在第一个结处.在更复杂的结构中,相同的物质层中有多于 1 个结.例如,一个 n+源/漏区位于 p 型阱和 n 型衬底中的结构,就有两个结.




图 4.17 多结结构

### 4.17.2 提取 N++源/漏方块电阻

为了提取方块电阻,先再次激活 ATHENA Extract 菜单,然后选择或输入以下:

Extract: Sheet resistance

Name: n++ sheet resistance

Extract location: X=0.05

接着单击 WRITE.文本窗口将出现如下文字:

```
extract name="n++ sheet res" sheet.res material="Silicon" mat.occno=1 x.val=0.05 region.occno=1
```

在该语句中, **sheet.res** 为即将提取的是方块电阻;**mat.occno=1** 和 **region.occno=1** 定义物质和区域发生层数均为 1;**x.val=0.05** 表明提取的位置为  $x=0.05\mu\text{m}$  处.

提取的结果数据保存在 **results.final** 中.

### 2.17.3 提取 LDD 方块电阻

为了提取隔氧下的 LDD 方块电阻,由前面模拟得到的结构,X=0.3 是合理的.提取过程如上,需改变的只是 Extract 和 Name 改为 ldd sheet res 以及 Extract location 的 X 量改为 0.3.

### 2.17.4 提取长沟道阈值电压

为了提取  $x=0.5\mu\text{m}$  处 NMOS 的长沟道阈值电压:

a.打开 ATHENA Extract 菜单.

b.选择或输入以下:

Extract: QUICKMOS 1D Vt

Device type: NMOS

Name: 1dvt

Qss:  $1\text{e}10$

Extract location: X=0.5

c.单击 WRITE 与 Cont.

在这里需要特别指出的是,Qss 为陷阱电荷(单位为  $\text{cm}^{-2}$ ).默认情况下,衬底 0V,栅压为 0-5V,步长 0.25V,器件温度为 300 Kelvin.

## 2.18 镜像得到完整 NMOS 结构

前面得到的只是 Half NMOS 结构,为了得到完整结构,接下来需对该结构进行 Mirror.该步骤必须在输出结构或命名电极之前完成.具体过程如下:

a.依次选 “Commands—>Structure—>Mirror” 打开 ATHENA Mirror 菜单.

b.在弹出的窗口中选 Right.

c.单击 WRITE 及 Cont.

完整结构如图 4-18.

从这个图中可以看出,右半部分结构完全是左半部分的复制,具体的量包括网格,掺杂值,等等.




图 4.18 完整 NMOS 结构(定义电极前)

## 4.19 定义电极

为了便于在器件模拟器 ATLAS 中设置偏压,NMOS 晶体管的电极必须进行定义:

a.依次选 “Commands—>Structure—>Electrode...” 打开 ATHENA Electrode 菜单.

b.选择或输入以下:

Electrode Type: Specified Positon

Name: Source

X Position: 0.1

c.单击 WRITE.

类似地,可以定义漏极位于  $x=1.1$ 、栅极位于  $x=0.6$  处.

最后定义背电极.在 ATHENA 中,背电极可以位于结构的底部而不需要那里是否有金属存在.为了定义背电极,在打开的 Electrode 中,选 Electrode Type 为 Backside,然后在 Name 一栏键入 backside 即可.

## 4.20 保存 ATHENA 结构文件

虽然 DECKBUILD 的历史功能 (history function),可保存每个工艺步骤的结构文件,但在很多情况下,我们需要的是独立保存和初始化结构.理由如下:

- i. 历史文件的个数有限制 (默认是 25 个)
- ii. 没必要在每次 DECKBUILD 文本运行后保存几十个历史文件 (每个历史文件占几十或几百 K 字节)

- iii. 通常用户想保存的只是关键步骤后的文件（如最后得到的结构文件）  
为了保存或装载一个结构，
- 依次选 “Commands—>File I/O...” 打开 ATHENA File I/O 菜单.
  - 单击 Save 按钮并命名文件为 nmos.str
  - 单击 WRITE 和 Cont.
  - 用 TONYPLOT 打开 nmos.str 结构文件.选择电极按钮以观察漏极、源极、栅极和背电极.




图 4.19 完整 NMOS 结构

## Experiment 5. NMOS Device: NMOS Device simulation using ATLAS

### 5.1 ATLAS 总揽

ATLAS 是一个基于物理的二维器件模拟器.它可预测特定半导体结构的电学行为,并了解器件运行时内部物理机制.

在 SILVACO's VIRTUAL WAFER FAB 模拟环境下,ATLAS 可单独使用或者是作为一个核心工具.在预测电路性能各过程变量影响的先后顺序中,器件模拟将拟合器件模拟和 SPICE 模型提取结果.

#### 5.1.1 ATLAS 输入和输出

图 3.1 显示的是出入 ATLAS 信息的类型.大部分 ATLAS 模拟使用两个输入:一个包含等待执行的命令文本文件,以及一个定义结构的结构文件.

ATLAS 产生三种输出文件.运行输出 (run-time output) 提供仿真运行过程的指南,并告知错误信息和警告信息出现的位置;日志文件 (log files) 保存器件分析时所有终端的电流电压;解文件 (solution files) 保存器件处于某一偏压下解变量相关的二维、三维数据.




图 5.1 ATLAS 的输入与输出

#### 5.1.2 ATLAS 的命令顺序

在 ATLAS 中,每个输入文件必须包括以正确顺序排列的五组语句(排列顺序如图 3.2).如果顺序错误,通常会产生一个错误信息,程序运行停止,或将导致程序的不正确运行.



Group		Statements
1. Structure Specification	————	MESH REGION ELECTRODE DOPING
2. Material Models Specification	————	MATERIAL MODELS CONTACT INTERFACE
3. Numerical Method Selection	————	METHOD
4. Solution Specification	————	LOG SOLVE LOAD SAVE
5. Results Analysis	————	EXTRACT TONYPLCT

图 5.2 ATLAS 命令组以及每组主要的语句

### 5.1.3 开始运行 ATLAS

为了在 DECKBUILD 下运行 ATLAS,键入

deckbuild - as&

几秒延时后,DECKBUILD 将如图 3.3 所示,从 DECKBUILD 下方可以看出,现在已进入 ATLAS 而不是 ATHENA 模块.




图 5.3 DECKBUILD 窗口下 ATLAS 模拟器

### 5.1.4 用 ATLAS 定义结构

在 ATLAS 中,一个器件结构可以用三个不同方式来定义:

1.从某一文件中读入已经存在的结构.这个结构可以由其他程序(如 ATHENA 或者 DEVEDIT)创建.

2.利用 DECKBUILD 的自动界面特性,从 ATHENA 或 DEVEDIT 转换而来.

3.用 ATLAS 命令构建一个结构.

相对于第三种方法,第一和第二种方法使用起来更便利,故而更受青睐.在本手册中,将使用第二种方法.

## 5.2 NMOS 结构的 ATLAS 模拟

在本手册中,我们将实现如下模拟:

1.  $V_{ds}=0.1V$  时  $I_d$ - $V_{gs}$  关系曲线.
2. 提取一些器件参数,如  $V_t$ ,  $\beta$  和  $\theta$ .
3.  $V_{gs}=1.1V, 2.2V, 3.3V$  时  $I_d$ - $V_{ds}$  关系曲线.

在第二章用 ATHENA 创建了 NMOS 结构,接下来,将模拟 NMOS 器件的电学特性.

## 5.3 创建 ATLAS 输入 DECK 文件

为了开始 ATLAS 模拟器,键入语句:

```
go atlas
```

为了载入由 ATHENA 产生的结构文件 “nmos.str”:

在 ATLAS Commands 菜单中,依次选择 “Structure—>Mesh...” 打开 ATLAS Mesh 菜单.

- a. 在 Type 一栏,单击 Read from file.
- b. 在 File name 一栏,填入结构文件名 “nmos.str”.
- c. 单击 WRITE 按钮.

## 5.4 模型定义命令组

因为 NMOS 结构已经在 ATHENA 中创建,故跳过结构定义命令组.在模型定义命令组中,我们必须分别用 Model 语句、Contact 语句和 Interface 语句定义模型、接触特性以及界面属性.

### 3.4.1 模型定义

对于简单的 MOS 模拟,推荐使用模型 SRH 和 CVT. SRH 是 Shockley-Read-Hall 复合模型而 CVT 是 Lombardi 反型层模型(参见 ATLAS User's Manual vol.1 pp.3-43),它设置一个包含浓度、温度、平行场、垂直场依赖的通用迁移率模型.为了定义这两个模型:

- a. 依次选择 “Commands—>Models—>Models...” 打开 ATLAS Model 菜单.

- b. 选择以下选项:

Category: Mobility

Mobility: CVT

Print Model Status: Yes

为了改变 CVT 模型默认值(如果必要的话):

1. 单击 Define Parameters 及 CVT..., 即可看到 CVT 菜单.
2. 修改参数后单击 Apply.

为了增加复合模型:

- a. 在 Category 一栏选择 Recombination., 将看到三个复合模型: Auger, SRH(Fixed Lifetimes) 和 SRH(Conc. Dep. Lifetimes).

- b. 选择 SRH(Conc. Dep. Lifetimes).

- c. 单击 WRITE.

### 5.4.2 定义接触性质

电极与半导体的接触在默认情况下被假设成欧姆接触.如果定义了功函数,电极就被看作是肖特基接触. Contact 语句用来定义一个或多个电极的金属功函数.为了定义 n 型多晶硅的功函数:

- a. 依次选择 “Commands—>Models—>Contacts...” 打开 ATLAS Contact 菜单.

- b.输入或选择如下:  
Electrode name: gate  
n-poly
- c.单击 WRITE.

### 5.4.3 定义界面性质

为了定义 NMOS 结构的界面性质,将使用 **Interface** 语句.该语句用来定义半导体/绝缘体界面的界面电荷密度和表面复合速度.为了定义 Si-SiO<sub>2</sub> 界面处存在面密度为  $3 \times 10^{10} \text{cm}^{-2}$  的固定电荷:

- a.依次选择 “Commands—>Models—>Interface” 打开 ATLAS Interface 菜单.
- b.在 Fixed Charge Density 一栏输入 3e10.
- c.单击 WRITE.

## 5.5 数值方法选择命令组

接下来,我们为模拟选择数值方法的类型.在解半导体器件问题时可以使用不同的方法.对于 MOS 结构,将使用解耦合(GUMMEL)和全耦合(NEWTON)方法.简单说来,解耦合方法(如 GUMMEL 方法)将在保持其他变量不变的情况下轮流解每个未知量,这个过程一直持续到得出一个稳定解.全耦合方法(如 NEWTON 方法)是同时解系统中所有的未知量.Method 语句归纳如下:

- a.依次选择 “Commands—>Solutions—>Method..” 打开 ATLAS Method 菜单.
- b.在 Method 一栏同时选择 Newton 和 Gummel.
- c.Max.num of iterations 默认值为 25.如有必要可改变.
- d.单击 WRITE.

同时选择 Newton 和 Gummel 将导致解答过程先进行 Gummel 循环,如果还没收敛的话,接着用 Newton 方法求解.

## 5.6 解定义命令组

在解命令组里,Log 语句用来保存 ATLAS 计算产生的包括所有终端性质的日志文件,Solve 语句用来解不同偏压时的情形,Load 语句用来载入所有解文件(solution files).

### 3.6.1 Vds=0.1V 时 Id-Vgs 曲线

在本手册中,我们期望得到 NMOS 结构的一条简单的 Vds=0.1V 时 Id-Vgs 曲线,为了达到此目的:

- a.依次选择 “Commands—>Solutions—>Solve...” 打开 ATLAS test 菜单.
- b.单击 Prop...按钮,激活 ATLAS Solve properties 菜单.
- c.在 Log file 一栏输入日志文件名 “nmos1\_” 并单击 OK 按钮.
- d.接着,将鼠标移至工作表区域,单击鼠标右键,选择 “Add new row” 如图 3.4 所示.
- e.把鼠标移至 gate 参数,接着单击鼠标右键.在一列电极名称中选择电极 drain.然后将 Initial Bias 设置为 0.1 并单击 WRITE.
- f.接着,再次将鼠标移至工作表区域,单击鼠标右键,选择 “Add new row” .
- g.在 “gate” 一行,将鼠标移至 CONST,单击鼠标右键,选择 “VAR1”,并将最终偏压 Final Bias 设置为 3.3,步长 Delta 设置为 0.1.



图 5.4 增加新的一行

h.单击 **WRITE**.在 **DECKBUILD** 文本窗口将看到如下语句:

```
solve init
solve vdrain=0.1
log outf=nmos1_0.log
solve name=gate vgate=0 vfinal=3.3 vstep=0.1
```

在上面语句中,solve init提供零偏压(或热平衡)下电势和载流子浓度的初始估计值.解完零偏压情形,第二个 solve 语句也即“solve vdrain=0.1”将在漏极上设置一直流偏压.如果某一特别的电极的偏压没有被任意 Solve 语句定义,则该电压为零.因此,没必要用 Solve 语句明确地定义所有电极的偏压.

第三个语句是一个 Log 语句:“log outf=nmos1\_0.log”.该语句将 ATLAS 计算的所有的模拟结果都保存在 nmos1\_0.log 这个日志文件中.这些结果包括直流模拟时的每个电极的电流电压.为了终止终端性质保存到这个文件,可使用另一个 Log 语句“log off”或使用另一个 log 文件名.

最后一个 Solve 语句“solve name=gate vgate=0 vfinal=3.3 vstep=0.1”将使栅压以步长 0.1V 从 0V 变化到 3.3V.注意该语句中的 name 参数必不可少且电极名称是区分大小写的.

## 5.6.2 提取器件参数

一些器件参数如  $V_t$ ,Beta 和 Theta 的提取,可用 ATLAS Extract 菜单来实现:

- 依次选择“Commands—>Extracts—>Device...”打开 ATLAS Extraction 菜单.
- 在默认情况下, $V_t$  在 Test name 一栏中被选择.当然,用户可以修改默认的提取表达式.
- 单击 **WRITE** 按钮, $V_t$  Extract 语句将出现在 **DECKBUILD** 文本窗口中.

重复上述步骤,将第二步中的  $V_t$  依次改为 Beta、Theta.

在运行(run)模拟之前,我们需要用 Tonyplot 语句绘画出模拟结果.为了自动绘画出 Id-Vgs 关系曲线,在最后的 Extract 语句后键入以下 Tonyplot 语句即可:

```
tonyplot nmos1_0.log
```

现在,我们可以开始运行模拟.单击 **DECKBUILD** 上 run 按钮开始器件模拟.一旦模拟完成,TONYPLOT 将自动绘制 Id-Vgs 曲线,曲线图如下.




图 5.5 NMOS 器件的  $I_d$ - $V_{gs}$  关系曲线

同样地,提取的器件参数(即  $V_t$ 、 $\beta$  和  $\theta$ )可以从 DECKBUILD 的运行输出窗口看到,如图 5.6.

```

next    line    stop    cont    run    quit    Line: 50
paste   init    pause   clear   restart  kill    Stop: None
4 7 G    -14.3*   -22.9*
Electrode  Va(V)  Jn(A/um)  Jp(A/um)  Jc(A/um)  Jt(A/um)
source     0.000e+00 -5.847e-05 -2.976e-27 -5.847e-05 -5.847e-05
drain      1.000e-01 5.847e-05 2.019e-23 5.847e-05 5.847e-05
gate       3.300e+00 -0.000e+00 -0.000e+00 -0.000e+00 0.000e+00
backside   0.000e+00 -1.882e-14 -7.709e-16 -1.959e-14 -1.959e-14
Total                                -1.254e-13

Time for bias point: 24.00 sec.
Total time: 1067.96 sec.

ATLAS>
EXTRACT> init inf="nmos1_0.log"
EXTRACT> extract name="vt"
(xintercept(maxslope(curve(abs(v."gate"),abs(i."drain")))) -
abs(ave(v."drain"))/2.0)
vt=0.203864 V
EXTRACT> extract name="beta"
slope(maxslope(curve(abs(v."gate"),abs(i."drain")))) *
(1.0/abs(ave(v."drain")))
beta=0.000280309 A/V2
EXTRACT> extract name="theta" ((max(abs(v."drain")) *
$"beta")/max(abs(i."drain"))) - (1.0 / (max(abs(v."gate")) - ($"vt")))
theta=0.156394 1/V
EXTRACT> quit

ATLAS> ## tonyplot nmos1_0.log
Finished executing line 49
ATLAS

```

图 5.6 显示提取器件参数的 DECKBUILD 运行输出窗口

### 5.6.3 用 Log,Solve 和 Load 语句产生曲线族

本手册的下一个目标是产生  $V_{gs}=1.1V$ 、 $2.2V$  和  $3.3V$ ,  $V_{ds}$  从  $0V$  变化到  $3.3V$  时的  $I_d$ - $V_{ds}$  曲线族.为了防止先前产生的“nmos1\_0.log”保存后面的终端性质,我们需要增加使用另一个 Log 语句:

log off

为了产生曲线族,首先,我们需要用 DECKBUILD:ATLAS Test 菜单获取每个  $V_{gs}$  时的解:

- 依次选择“Commands—>Solutions—>Solve...”打开 DECKBUILD:ATLAS Test 菜单.
- 单击 Prop...按钮,激活 ATLAS Solve properties 菜单.
- 在 Write mode 一栏选 Line 而不是 Test 并单击 OK.
- 为栅极设置偏压如图 3.7 所示.




图 5.7 为栅极设置偏压

- 单击 WRITE 按钮,DECKBUILD 文本窗口将显示如下 Solve 语句:

solve vgate=1.1

- 为了保存该解的输出在一个 ATLAS solution 文件中,附加上“outf=solve1”:

solve vgate=1.1 outf=solve1

对于栅压为  $2.2V$  和  $3.3V$ ,重复以上步骤并将 solve 文件分别保存至“solve2”以及“solve3”中.看到的将是如下的 solve 语句:

solve vgate=2.2 outf=solve2

solve vgate=3.3 outf=solve3

接下来,我们将再次使用 ATLAS Test 菜单设置 Solve 语句,使漏压从  $0V$  逐渐变化到  $3.3V$ .为了达到此目的,

- 依次点击“Commands—>Solutions—>Solve...”打开 DECKBUILD:ATLAS Test 菜单.
- 在 ATLAS Test 菜单中,单击 Prop...按钮,激活 ATLAS Solve properties 菜单.
- Write mode 选 Test.
- log file 文件名改为 nmos2\_.
- 单击 OK.
- 在工作表区域,在 Name 一栏改 gate 为 drain,Type 一栏改 CONST 为 VAR1,Initial Bias,Final Bias 和 Delta 分别设置为  $0$ 、 $3.3$  以及  $0.3$ .

- 单击 WRITE.

接下来,我们将利用 Load 菜单载入解文件(solution file),即栅压为  $1.1V$  的 Solve1 文件.并将它取代“solve init”语句.

- 高亮语句“solve init”.
  - 依次点击“Commands—>Solutions—>Load...”打开 DECKBUILD:ATLAS Load 菜单.
  - 在 File name 一栏输入 solve1.
  - 在 Format 一栏选择 SPICES.
  - 单击 WRITE 按钮,将出现一个提示框.接着单击“Yes,replace selection”按钮.
- Load 语句将取代 solve init 语句.因此,从 Load 语句(即 load infile=solve1)到 Solve 语句这几句

语句将产生  $V_{gs}=1.1V$  时的  $I_d$ - $V_{gs}$  曲线.为了产生  $V_{gs}=2.2V$  及  $3.3V$  时的  $I_d$ - $V_{gs}$  曲线,只需简单的拷贝这三句语句然后:

- 改变 Load 语句中输入文件名——改 solve1 为 solve2 和 solve3.
- 改变 Solve 语句中 log 文件名——改 nmos2\_0.log 为 nmos3\_0.log 和 nmos4\_0.log.
- 改后看到的语句如图 3.8 所示.

为了在一张图中绘出曲线族,即三个 log 文件,键入如下 Tonyplot 语句:

```
tonyplot -overlay nmos2_1.log nmos2_2.log nmos2_3.log
-set nmos.set
```

在该语句中,**-overlay** 是将 3 个 log 文件重叠在一张图上; **-set nmos.set** 是使用这个文件来在 TonyPlot 中显示前面的 optoex08\_2.str 文件, -set 文件中存储了一些相应的信息.




图 5.8 产生  $I_d$ - $V_{ds}$  曲线族的语句

#### 5.6.4 退出模拟

最后,键入如下语句退出模拟:

```
quit
```

现在可以单击 DECKBUILD 上 Cont 按钮继续运行器件模拟.一旦模拟完成,TONYPLOT 将自动绘画  $I_d$ - $V_{ds}$  曲线族,如图所示:




图 5.9 NMOS 的输出特性



## Experiment 6. NMOS Devise: Creating a NMOS Devise using DEVEDIT

### 6.1 DEVEDIT 总揽

DEVEDIT 是一个器件结构编辑器.它可以对一个已存在的结构生成一个新的网格,修改一个器件,或者从头开始创建一个器件.这些器件接着可以用在 SILVACO 二维和三维模拟器中.DEVEDIT 可以借助图形用户界面使用(Graphical User Interface ,GUI)或者可作为 DECKBUILD 下的一个模拟器.

#### 6.1.1 问题

DEVEDIT 之前的器件模拟器的一个局限在于网格数不够或者网格质量不好.如果一个网格包含太多钝角三角形或者三角形的数目不够多(网格太过粗糙),结果将不精确甚至根本没有结果而言;而如果一个网格包含太多三角形(网格太过精密),模拟时间将大大增加.因为大部分的模拟器的模拟时间将随三角形数目(或格点数)的增加而成几何增长,因此,把三角形的数目控制在一个合适的数目范围内是非常必要的.采用类似于 SSUPREM4 的模拟器创建一个非均匀网格将是一件费时费力的事.

#### 6.1.2 解决方案

DEVEDIT 解决这些问题的方法是容许创建结构或以 SILVACO 标准结构文件形式读入 DEVEDIT.接下来,包含在文件中的网格可以用 MeshBuild 运算法则来取代.精炼网格可以通过设置描述关键区域的参数或简单地指出需要需要精炼的区域. 在创建一个结构的过程中,器件定义可以简单地通过在屏幕上绘画来实现.

#### 6.1.3 什么时候使用 DevEdit

当想进行如下操作时使用DEVEDIT:

- ◇ 定义一个**屏幕上交互式**(a device interactively on the screen)的器件,用于后面的器件或工艺模拟.
- ◇ 当工艺模拟器创建的网格不适用于器件模拟器的时候,重新定义器件结构的网格.
- ◇ 在工艺或器件模拟过程中,如果已定义的网格不再适用于下一步模拟,重新定义网格.

#### 6.1.4 什么时候不使用 DevEdit

当进行如下操作时不宜使用DEVEDIT:

- ◇ 取代需要高精度的数字工艺模拟
- ◇ 为一维器件结构网格定义

### 6.2 开始使用 DEVEDIT

DEVEDIT可以从UNIX或DECKBUILD开始运行.DEVEDIT可读的文件类型有两种:一是 SILVACO标准结构文件形式(在所有SILVACO模拟器中通用);二是命令形式(创建结构的DEVEDIT命令).结构文件形式包括三角形、掺杂值、边界等信息;而命令形式一般用在从头开始创建器件网格.

为了在GUI模式下开始使用DEVEDIT,将使用一个UNIX语句:

```
devedit &
```

接着可以看到DEVEDIT基本窗口显示如图6.1:




图 6.1 DEVEDIT基本窗口

## 6.3 用 DEVEDIT 创建一个 MOS 器件

### 6.3.1 工作区域

为了设置工作区域:

1. 右键单击 **Regions** 菜单, 在打开的下拉框中, 选择 **Resize Work Area**, 在屏幕右方将看到一个新的面板.
2. 设置  $y$  的最小值为 -1.0 (半导体将从  $y=0$  开始). 记得输入值后单击回车键.
3. 设置  $x$  的最小值为 0.
4. 设置  $x$  和  $y$  的最大值均为 2.0.
5. 单击 **Apply**. 看到的将如图 6.2 所示.




图 6.2 调整大小后的区域

### 6.3.2 增加硅基区域

为了增加一个硅基区域:

- 1.在 Regions 菜单下拉框中选择 Add Regions...,打开的 Add Regions 面板将如图 6.3 所示,默认情况下,Material 这一栏位硅.
- 2.移动鼠标至绘画窗口的(0.0,0.0)点.鼠标的位置将显示在面板右侧.
- 3.先后按住和释放鼠标左键.随着鼠标的移动,将有一个与第一点相连的线.把鼠标移到(2.0,0.0),再次按住和释放鼠标左键.在(2.0,2.0)和(0.0,2.0)处重复以上操作.
- 4.现在应该可以看到一个三角形的三边.稍后将自动增加第四边(参见图 6.4).
- 5.改变当前编辑模式:鼠标右键单击 New Polygon,选择 Set Base Impurity.
- 6.在 Doping Type 一栏,改 Silicon Donor 为 Silicon Acceptor.
- 7.在 Boron 一栏键入 1e15.如图 6.5.
- 8.单击 Apply.衬底区域显示为橙色.




图 6.3 增加区域面板




图 6.4 区域绘画




图 6.5 设置衬底掺杂

### 6.3.3 其他区域

为了增加一个绝缘区域:

- 1.在 Regions 菜单下拉框中选择 Add Regions....
- 2.在 Material 菜单下拉框中选择 Silicon Oxide.
- 3.画点(0,0),(0,-0.1),(2,-0.1)和(2,0).
- 4.单击 Apply.

氧化硅区域将显示如图 6.6 所示.

### 6.3.4 增加接触

为了增加一些接触:

- 1.在 Regions 菜单下拉框中选择 Add Regions....
  - 2.选择 Electrode 复选框.
  - 3.单击 Electrode names,选择 Source.
  - 4.在 Material 菜单下拉框中选择 Aluminum.
  - 5.画点(0,0),(0,-0.2),(0.3,-0.2)和(0.3,0).区域重合将无关紧要,因为新的区域将刻蚀掉旧的区域.
  - 6.选择 Electrode 复选框.
  - 7.单击 Apply.源极将出现在 DEVEDIT 窗口中.如图 4.7 所示.
- 为了增加第二个电极:
- 8.在 Regions 菜单下拉框中选择 Add Regions....
  - 9.选择 Electrode 复选框.
  - 10.单击 Electrode names,选择 drain.其他地方应该已经设置完成.
  - 11.画点(1.7,0),(1.7,-0.2),(2,-0.2)和(2,0).

12.单击 Apply.




图 6.6 增加一个氧化硅区域




图 6.7 源极区

### 6.3.5 增加衬底接触

为了增加衬底接触:

- 1.在 Regions 菜单下拉框中选择 Substrate Electrode....
- 2.选择 Substrate Electrode Exists 复选框.
- 3.单击 Apply.

### 6.3.6 增加栅极

最后,增加栅极:

- 1.在 Regions 菜单下拉框中选择 Add Regions....
- 2.选择 Electrode 复选框.
- 3.单击 Electrode names,选择 gate.

- 4.在 Material 菜单下拉框中选择 Polysilicon.
- 5.画点(0.7,-0.1),(0.7,-0.5),(1.3,-0.5)和(1.3,-0.1).
- 6.单击 New Polygon,选择 Set Base Impurities.
- 7.在 Arsenic 一栏输入  $1e+19$ .
- 8.单击 Apply.将看到多晶硅栅极区,如图 6.8 所示.




图 6.8 多晶硅栅极区

### 6.3.7 增加一些掺杂

既然所有区域已经定义完成,现在需要增加一些掺杂:

- 1.在 Impurities 菜单下拉框中选择 Add Impurity...,Add Impurity 面板将出现在窗口右边.
- 2.在 Add Impurity 面板中,在 Impurity 菜单中选择 Arsenic.
- 3.输入点(0,0)和(0.4,0.0).这将是浓度峰值区.
- 4.在 Peak Concentration 和 Reference Value 栏分别输入  $1e+20$  和  $1e+15$ .因为硅衬底掺杂水平为  $1e+15$ ,reference value 描述的是不同掺杂浓度的结深(junction value).掺杂的参数设置具体见图 6.9.




图 6.9 增加掺杂面板

为了设置滚降函数(roll off function):

1.打开Y Rolloff下拉框,选择“Gaussian (Dist)” (而不是Gassian).该函数设置峰值浓度与参考值之间存在一定距离.

2.在Y方向Distance一栏输入0.5.

3.打开X Rolloff下拉框,选择“Error Fuction (Dist)”.

4.在X方向Distance一栏输入0.3.

5.单击Apply,User Added Impurities将出现在右下方.

为了使漏极下方有着相同的杂质分布:

1.打开Impurity下拉框选择Add Impurity....

2.输入点(1.6,0)和(2,0).其余设置保持不变.

3.单击Apply.

为了显示掺杂分布:

1.在Show:Net Doping下拉框选择Fine(如果有时间,你可以选择Very Fine),净掺杂分布将如图4.10所示.

2.为了增加轮廓图,打开Contour Legend下拉框,选择所需观察的选项.(参见图6.10)

注意:在继续下去之前,为了防止下一步工作后缓慢刷屏,设置Show:Net Doping菜单为Off,

## 63.8 创建网格

在创建网格之前,首先必须设置网格:

1.在Mesh下拉列表中选择Mesh Parameters...,Mesh Parameters面板将会出现.

2.设置Base Mesh Height和Base Mesh Width均为0.2微米.

3.单击Apply.(网格直在后面才创建起来.)

接着,设置refinement参数:

1.在Mesh下拉列表中选择Refinement On Quantities....该面板控制的区域将有一个更精密的基于杂质浓度值的网格.

2.在Min.Mesh Spacing一栏设置为0.02微米(默认值).

3.在Refinement Quantities区域的Add下拉列表中选择Net Doping,

4.单击Done.

注意:杂质的精炼(refining on impurities)仅限于半导体区域.

最后创建网格:

在Mesh菜单的下拉列表中选择Meshbuild.DEVEDIT现在将创建一个网格结构.在创建网格过程中将可以看到一个Cancel按钮.它的作用是防止应用不合理的网格参数和产生过多的三角形.






图 6.10 净掺杂分布

### 6.3.9 保存 SILVACO 标准结构文件

- 1.在 File 菜单下拉列表中选择 Save as....
- 2.定义将保存的文件名为“nmos.str”.如图 6.11 所示.
- 3.单击 Save Structure.

### 6.3.10 保存 DEVEDIT 文件

- 1.在 File 菜单下拉列表中选择 Save as....
- 2.定义文件名为“mos.de”.
- 3.单击 Save Comments.

注意:几乎由 DEVEDIT 创建的所有文件都应该以两个文件形式保存——一个命令文件和一个 SILVACO 标准结构文件.命令文件包含着比 SILVACO 标准接轨文件更多的结构信息.

为了读入一个命令文件,键入如下命令即可:

Devedit test.de &




图 6.11 保存 DEVEDIT 结构文件

## Experiment 7. MESFET Device

### 基础知识

MESFET 具有与 MOSFET 相似的电流-电压特性. 然而在器件的栅电极部分, MESFET 利用金属-半导体的整流接触取代了 MOSFET 的 MOS 结构; 另外, 在源极与漏极部分, MOSFET 以欧姆接触取代 MOSFET 中的 p-n 结. 也就是说, **MESFET 中共有三个金-半接触**: 一个肖特基接触作为栅极, 以及两个当作源极与漏极的欧姆接触. 基本结构如图 7.1.

与 JFET 相比, MESFET 有两大优点: (1) MESFET 的工艺简单. M-S 工艺允许把沟道长度做得更短, 使得结电容更小, 有利于提高器件的开关速度和工作频率; (2) MESFET 多用砷化镓材料制作, 砷化镓的电子迁移率大约是硅的六倍, 因此可以制造出高频、高速器件.




图 7.1 GaAs MESFET 的基本结构

### Example 1: 离子注入 GaAs MESFET 制作工艺及 $V_t$ 测量

该例采用 ATHENA 的 FLASH 模块和 ATLAS 的 BLAZE capability 来模拟 MESFET 的制造过程及其电学分析.

命令 `"contact name=gate work=4.87"` 用来设置栅极功函数. 源/漏接触假设为欧姆接触. 必须注意的是即使在 ATHENA 模拟中使用物质钛, 这个 contact 语句也是必需的. ATLAS 计算一个 Schottky 势垒的高度采用如下的公式:

$$\text{barrier height} = (\text{metal work function} - \text{semiconductor electron affinity})$$

解 MESFET 阈值电压的第一步为获取所有接触零偏压时的初始解. 然后漏极电压设置为 0.1V. LOG 语句打开一个 log 文件, 并把所有的终端特性存储在这个文件当中. 最终的 SOLVE 语句使得栅压从零到 -3V 变化. 注意 ATLAS 在任何时候都需要先求零初始解. 因此栅压应该从零到所需值间变化. 如果所需的栅压中有正有负, 零初始解需要求两次.

### 模拟程序

```
go athena
# GaAs MESFET fabrication and analysis using DevEdit
# 网格定义
line x loc=-1.5 spac=0.2
line x loc=-.7 spac=0.1
line x loc=-.5 spac=0.05
line x loc=0.0 spac=0.1
line x loc=0.5 spac=0.05
```

```

line x loc=0.7 spac=0.1
line x loc=1.5 spac=0.2
#
line y loc=0.00 spac=0.02
line y loc=2.00 spac=0.5

# 衬底定义: 半绝缘GaAs衬底, p型
init gaas c.beryllium=1.0e13 orient=100 space.mult=1

#
implant beryllium energy=100 dose=2e11
implant silicon energy=100 dose=1e12          # n型
diffus time=10 temp=850

# 沉积和刻蚀钛形成金属栅极
deposit titanium thick=.3 divisions=10
etch titanium right p1.x=0.5
etch titanium left p1.x=-0.5

# 形成侧墙
deposit oxide thick=0.35 divisions=8
etch oxide thick=.4

# 注入前重新定义网格: 为了使得杂质的 profile 更好, 仿真更精确或者速度更快.
# regrid before implant
go devedit
base.mesh height=0.08 width=0.08          # 提供最粗糙的网格 (0.08×0.08)
bound.cond apply=false max.ratio=300      # 边界条件
bound.cond when=automatic max.slope=28 rnd.unit=0.001 line.straightening=2 align.points
      constr.mesh max.angle=90 max.ratio=300 max.height=1 max.width=1 \
      min.height=0.0001 min.width=0.0001
      constr.mesh type=Semiconductor default
constr.mesh type=Insulator default
constr.mesh type=Metal default
imp.refine min.spacing=0.02                # 定义最小网格间隙
imp.refine imp="net doping" sensitivity=1

#
mesh
# 源漏注入
go athena
implant silicon energy=50 dose=1e13

# 非掺杂区域网格最少化

```

```

# 非注入区域减少网格
go devedit
base.mesh height=0.4 width=0.4
bound.cond apply=false max.ratio=300
bound.cond when=automatic max.slope=28 rnd.unit=0.001 line.straightening=2 \
    align.points
constr.mesh max.angle=90 max.ratio=300 max.height=1 max.width=1 \
    min.height=0.0001 min.width=0.0001
constr.mesh type=Semiconductor default
constr.mesh type=Insulator default max.angle=180
constr.mesh type=Metal default
imp.refine min.spacing=0.03
imp.refine imp="net doping" sensitivity=.5
mesh
go athena
diff time=10 temp=850
# 形成源/漏接触:欧姆接触
deposit aluminum thick=.2 divisions=4
etch aluminum start x=-1 y=10
etch cont x=-1 y=-10
etch cont x=1 y=-10
etch done x=1 y=10

# 电极定义
electrode name=source x=-1.4
electrode name=drain x=1.4
electrode name=gate x=0.0

#
go devedit
base.mesh height=0.1 width=0.1
bound.cond apply=false max.ratio=300
bound.cond when=automatic max.slope=28 rnd.unit=0.001 line.straightening=2 \
    align.points
constr.mesh max.angle=90 max.ratio=300 max.height=1 max.width=1 \
    min.height=0.0001 min.width=0.0001
constr.mesh type=Semiconductor default
constr.mesh type=Insulator default max.angle=180
constr.mesh type=Metal default
imp.refine min.spacing=0.03
imp.refine imp="net doping" sensitivity=1

mesh
structure outfile=Example 1_0.str

```

```
tonyplot Example 1_0.str -set Example 1_0.set
```

```
go atlas
```

```
# 设置栅极功函数
```

```
contact name=gate work=4.87
```

```
# 肖特基接触
```

```
# 定义GaAs中电子空穴寿命及计算模型
```

```
material material=GaAs taun0=1.e-8
```

```
taup0=1.e-8
```

```
models conmob fldmob srh optr print
```

```
# 计算方法定义
```

```
method newton trap
```

```
solve vdrain=0.1
```

```
# 求解IV曲线并保存到log文件中
```

```
log outf=Example 1.log master
```

```
solve vgate=0.0 vstep=-0.1 vfinal=-3
```

```
name=gate
```

```
# 从IV曲线中提取出Vt
```

```
extract init infile="Example 1.log"
```

```
extract name="vt"  
(xintercept(maxslope(curve((v."gate"),(i."drain")))))
```

```
save outfile=Example 1_1.str
```

```
tonyplot Example 1.log
```

```
quit
```




图 7-2 MESFET 结构图

运行结果

1. 结构图如图 7-2 所示

2. Id-Vg 曲线图如图 7-3 所示,

提取出的阈值电压:vt=-0.399607 V




图 7.3 MESFET 的  $I_d$ - $V_g$  曲线图

## Example 2: Epitaxial GaAs MESFET S-parameter

在该例中将使用:

- √ 用 ATLAS 语句定义 MESFET 结构
- √ 设置 GaAs 迁移率模型和栅极功函数
- √  $V_{gs}=0$  时的  $I_d/V_{ds}$  特性
- √ 在某一直流偏压下的不同频率的交流分析
- √ 从 ATLAS 到 s-parameters 电容和电导数据的转换
- √ 用 TonyPlot 绘画 MESFET s-parameters 的 Smith Charts

**REGION** 语句用来定义两个 GaAs 区域,这两个区域在后面将进行不同浓度的掺杂。**ELECTRODE** 语句定义电极的名字和位置.栅极功函数用 **CONTACT** 语句来设置.每个 GaAs 区域均匀掺杂.一层 n 型有源区位于轻掺 p 型衬底之上.

初始态的模拟所用的数值方法(numerical methods)是保守的.语句"method gummel newton"定义了每个偏压点模拟采用的方法是解耦合(gummel)方法.如果计算不收敛,这将转换到全耦合 (newton) 方法.对于复杂的器件而言,这个方法求初始偏置时最严谨的方法.它将花费更多的 CPU 运行时间,也有可能在高电流时出现一些问题.如果问题发生了,可以尝试使用语句"method newton".

直流模拟是利用 **SOLVE** 语句通过变化漏压至 3.0V.在这之后,一个 log 文件打开,交流分析开始模拟.频率变化是这样实现:

**solve ac.analysis direct frequency=1.e9 fstep=2.e9 nfsteps=20**

其中,参数 **ac**:交流分析;**direct**:定义一个健全的交流解答器(solver)用于高频分析,同时分配比直流模

拟更多的内存.如果你的系统没有足够的虚拟内存,模拟将停止并出现错误信息.所需内存的大小依赖于网格节点的数目;**frequency** 的值为交流信号初始频率的大小;**fstep** 为步长频率;**nfsteps** 为步数.

ATLAS 中的交流分析计算基于直流的小交流信号电流的实部和虚部.从这些电流 ATLAS 可计算出每对电极间的电容电导.**C drain>gate** 表示漏-栅电容, **-1\*C gate>gate** 表示总的栅极电容.

s-parameters (或者 Z, X 及 ABCD parameters)的计算是通过在进行交流分析时设置 log 语句的 '**s.param**' (或者 y.param 等等)实现.最重要的参数要数\*port 参数.s-parameter 分析的前提假设是器件是一个四终端两端口器件.其输入输出指定各自的电极.用户必需正确定义所有的端口.在本例中,输入端口为栅极,输出端口为漏极.

模拟程序

```
go atlas
Title   MBE Epitaxial GaAs MESFET - S parameters calculation
# 网格定义
mesh   space.mult=1.0
#
x.mesh loc=0.00 spac=0.3
x.mesh loc=2.3 spac=0.02
x.mesh loc=2.7 spac=0.02
x.mesh loc=5 spac=0.3
#
y.mesh loc=0.00 spac=0.01
y.mesh loc=0.04 spac=0.03
y.mesh loc=0.12 spac=0.02
y.mesh loc=6 spac=1.0

# 区域定义
region num=1 GaAs x.min=0 x.max=5 y.min=0 y.max=0.12
region num=2 GaAs x.min=0 x.max=5 y.min=0.12 y.max=6.12

#电极定义
elec num=1 name=source   x.min=0.0 y.min=0.0 x.max=1.0 y.max=0.0
elec num=2 name=drain    x.min=4.0 y.min=0.0 x.max=5.0 y.max=0.0
elec num=3 name=gate     x.min=2.35 length=0.3

# 掺杂定义
doping region=1 uniform conc=1.0e17 n.type   # 衬底
doping region=2 uniform conc=1.0e15 p.type   # 沟道

# 设置接触、GaAs 迁移率及复合模型、材料特性
contact num=3 work=4.77                      # 肖特基接触
models region=1 print conmob fldmob srh optr
models region=2 srh optr
```



```
material region=2
```

```
# 设置计算方法
```

```
# Solution - use gummel newton for initial then switch  
to full newton
```

```
method gummel newton
```

```
# 求解栅压为零时的输出特性
```

```
solve vgate=0
```

```
save outfile=mesfetex02_1.str
```

```
tonyplot mesfetex02_1.str -set mesfetex02_0.set
```

```
solve vdrain=0.025 vstep=0.025 vfinal=0.1 name=drain
```

```
method newton
```

```
solve vdrain=0.2 vstep=0.1 vfinal=0.6 name=drain
```

```
solve vdrain=0.8 vstep=0.2 vfinal=3 name=drain
```

```
# 交流小信号分析
```

```
# Small signal ac analysis with s-parameter calculation
```

```
log outf=mesfetex02_ac.log s.param inport=gate
```

```
outport=drain width=100
```

```
solve ac.analysis direct frequency=1.e9 fstep=2.e9
```

```
nfsteps=20
```

```
tonyplot mesfetex02_ac.log -set mesfetex02_1.set
```

```
tonyplot mesfetex02_ac.log -set mesfetex02_2.set
```

```
quit
```

运行结果:

1. 结构图如图 7-4 所示

2. 交流电导如图 7-5 所示

3. 能带图如图 7-6 所示

由能带图可得势垒高度为:  $0.7-0.09=0.61\text{V}$




图 7.4 MESFET 结构图




图 7.5 MESFET 交流电导




图 7.6 MESFET 的能带图

### Example 3: 能量平衡模型和漂移扩散模型对比

在本例中,将分别用能量平衡(Energy Balance,EB)模型和漂移扩散(Drift-Diffusion,DD)模型对 0.2um MESFET 结构进行电学分析.具体包括:

- √ 用 ATLAS 句法进行结构定义
- √ 能量平衡模型的定义
- √  $V_{gs}=0.0V$  时  $I_d/V_{ds}$  的模拟
- √ 用漂移扩散模型重新模拟并比较结果

能量平衡模型对物理器件效应(特别是速度过冲和非局部碰撞电离效应,经典的漂移扩散模型并没涉及到此效应,在亚微米 MESFET 的模拟中,将观察到这两个效应)提供一个更精确的描述.本例击中考虑 GaAs 的速度过冲效应.非局部碰撞电离可以在击穿模拟中看到.因为漂移扩散模型并没有解释速度过冲效应,其结果是该模型将严重低估电流.当沟道长度减小时,该矛盾将变得更加严重.

能量平衡模型通过 **model hcte.el** 定义.因为这是个单极器件,仅选择电子能量平衡.能量平衡模型的一个重要参数是弛豫时间,它可通过 **material** 语句来定义.

数值方法选择牛顿方法——**method newton**.如果最初模拟出现收敛问题,the block method may also be added.

第二个 Atlas 与第一个 Atlas 相同,只是没了语句 **models hcte.el**;语句 **output e.velocity** 将电子速度信息保存到解文件(solution files)中.

模拟程序

```
go atlas
title GaAs MESFET Simulation with EB and DD models
#
# 第一部分:网格定义
mesh space.mult=2.0
#
x.mesh loc=0.00 spac=0.02
x.mesh loc=0.2 spac=0.01
x.mesh loc=0.4 spac=0.01
x.mesh loc=0.6 spac=0.02
#
y.mesh loc=0.00 spac=0.005
y.mesh loc=0.2 spac=0.02
y.mesh loc=1 spac=0.1
#
# 第二部分:结构和模型定义
region num=1 material=GaAs
#
elec num=1 name=source x.min=0.0 y.min=0.0 x.max=0.1 y.max=0.
elec num=2 name=drain x.min=0.5 y.min=0.0 x.max=0.6 y.max=0.
elec num=3 name=gate x.min=0.2 length=0.2
#
doping uniform conc=1.e15 p.type # 衬底
doping uniform conc=1.e17 n.type y.min=0 y.max=0.12 # 沟道
doping uniform conc=5.e18 n.type x.left=0. x.right=0.1 y.min=0 y.max=0.05
doping uniform conc=5.e18 n.type x.left=0.5 x.right=0.6 y.min=0 y.max=0.05 # S/D 接触
#
contact num=3 work=4.87
#
#####
# EB calculation #
#####
#
models print conmob fldmob evsatmod=1 hcte.el
material taurel.el=1.e-12 taumob.el=1.e-12 vsat=1.e7 # 弛豫时间、饱和速度
#
# 第三部分:解初始解
solve init
save outf=mesfetex03.str
tonyplot mesfetex03.str -set mesfetex03_0.set
#
# 第四部分:求解输出曲线
```

```

method newton maxtrap=6
output e.velocity
#
probe n.mob max dir=0 name="mux"
probe n.mob max dir=90 name="muy"
log outf=mefetex03_1.log
#
solve vdrain=0.0 vstep=0.05 vfinal=0.5 name=drain
method newton
solve vdrain=0.6 vstep=0.1 vfinal=2 name=drain      # Vd
save outf=mefetex03_1.str
#
#####
#          DD calculation          #
#####
#
go atlas
# 第一部分:重载网格、结构定义
mesh infile=mefetex03.str
#
contact num=3 work=4.87
models print conmob fldmob evsatmod=1
material taurel.el=1.e-12 taumob.el=1.e-12
vsat=1.e7
#
# 第二部分:输出曲线
method newton maxtrap=6
output e.velocity
probe n.mob max dir=0 name="mux"
probe n.mob max dir=90 name="muy"
log outf=mefetex03_2.log
solve vdrain=0.0 vstep=0.05 vfinal=0.5 name=drain
method newton
solve vdrain=0.6 vstep=0.1 vfinal=2 name=drain
save outf=mefetex03_2.str
tonyplot mefetex03_1.log -overlay mefetex03_2.log -set mefetex03_log.set
quit

```




图 7.7 MESFET 结构图

运行结果

1. 结构图如图 7-7 所示
2. 不同模型下的 X 方向电子迁移率如图 7-8 所示
3. 不同模型下的 Y 方向电子迁移率如图 7-9 所示




图 7.8 EB 和 DD 模型下 X 方向电子迁移率比较图




图 7.9 EB 和 DD 模型下 Y 方向电子迁移率比较图

#### 4. 不同模型下的输出曲线




图 7.9 EB 和 DD 模型下输出曲线比较图

#### Example 4:直流分析深能级体陷阱

接下来的三个例子将模拟存在陷阱态 MESFET 结构的模拟.该例包括以下几个部分:

- √ 形成 MESFET 结构并掺杂(用 FLASH)
- √ 考虑陷阱的直流模拟  $I_d/V_{gs}$  特性
- √ 考虑陷阱的栅极关断时的瞬态模拟
- √ 考虑陷阱的频域交流模拟
- √ 考虑陷阱的  $s$  参数提取

**TRAP** 语句用来设置体陷阱态的参数.定义多能态的语句有很多,在这里仅用一个能态.陷阱类型可以设置成施主或受主.陷阱的能量水平用 **e.level** 来设置.等价的设置同样可用,比方说用寿命取代横截面积.

直流模拟:设置漏极电压 0.5V 然后使栅压从 0 逐渐变到-1.0V.

比较有/无陷阱的模拟结果发现,施主陷阱导致正的阈值电压变得更大.

模拟程序

```

go athena
# 网格定义
line x loc=0.00 spac=0.10
line x loc=2 spac=0.10
#
line y loc=0.1 spac=0.01
line y loc=0.5 spac=0.1

# 衬底定义
init gaas c.silicon=1.0e11

# 外延层沉积
# an active layer of 0.1um of n-type GaAs is deposited
deposit gaas thick=0.10 c.silicon=2.0e17 divisions=15 dy=0.005 ydy=0.00

# 形成掩蔽层
deposit nitride thick=0.5 divisions=4
etch nitride left p1.x=0.2
etch nitride right p1.x=1.8
relax y.min=0.2

# 形成源/漏
implant silicon dose=1.0e15 energy=30
etch nitride all

# 形成接触
deposit alum thickness=0.01 div=1
etch alum start x=0.2 y=-5
etch cont x=0.7 y=-5
etch cont x=0.7 y=5
etch done x=0.2 y=5
etch alum start x=1.3 y=-5
etch cont x=1.8 y=-5
etch cont x=1.8 y=5
etch done x=1.3 y=5

# 电极定义
electrode name=source x=0.1
electrode name=gate x=1.0
electrode name=drain x=1.9

# 结构保存&绘画
structure outf=mefetex04_0.str
tonyplot mefetex04_0.str -set mefetex04_0.set

```

# 图 7.10

```

go atlas
# 定义接触&模型
contact name=gate work=4.87
models print fldmob srh
#饱和速度、电子空穴 SRH 复合寿命
material vsat=0.92e7 taun0=1e-9 taup0=1e-9
#TRAP 语句用来设置体陷阱态的参数.
trap acceptor e.level=0.45 density=1.0e16 degen=12 sign=1e-14 sigp=3.e-14 fast # 受主陷阱
#
method newton trap
solve prev
solve vdrain=0.01
solve vdrain=0.05
solve vdrain=0.2
solve vdrain=0.5
#
log outf=mefetex04_1tr.log master
solve vgate=0.0 vstep=-0.1 vfinal=-1.0 name=gate
save outf=mefetex04_2tr.str
tonyplot mefetex04_2tr.str -set mefetex04_1.set

```

```

go atlas
contact name=gate work=4.87
models print fldmob srh
#
material vsat=0.92e7 taun0=1e-9 taup0=1e-9
#
method newton trap
solve prev
solve vdrain=0.01
solve vdrain=0.05
solve vdrain=0.2
solve vdrain=0.5
#
log outf=mefetex04_1.log
solve vgate=0.0 vstep=-0.1 vfinal=-1.0 name=gate
#  $V_g=0 \sim -0.1V$ 
save outf=mefetex04_2.str
tonyplot mefetex04_1.log -overlay mefetex04_1tr.log # 图 7.11

```

quit

运行结果

1. 结构图如图 7-10 所示




图 7.10 MESFET 结构图



## 2. 转移曲线




图 7.11 有无深能级 Id-V<sub>gs</sub> 曲线比较图

### Example 5: 瞬态分析深能级体陷阱

本例基于例 4 的结构,所不同的是,在瞬态模式下模拟有陷阱时的情况.

漏压设置为 0.5V.栅压从 0 变化到 -2.0V.参数 **ramptime** 设置栅压变化时间,而 **tstop** 则设置模拟时间.参数 **dt** 用来设置瞬态的第一个时间步长,其他的时间步长则自动计算.在大部分情形下,dt 值为 ramptime/100 是合适的.

与例 4 不同的地方:

#### ① 结构方面

```
log outf=***.log
```

```
solve vgate=-2.0 ramptime=1e-9 dt=1e-11 tstop=1000e-9
```

(例四为:solve vgate=0.0 vstep=-0.1 vfinal=-1.0 name=gate)

#### ② 特性方面

```
go atlas
```

```
#
```

```
contact name=gate work=4.87
```

```
models print fldmob srh
```

```

material vsat=0.92e7 taun0=1e-9 taup0=1e-9
trap acceptor e.level=0.45 density=1.0e16 degen=12 sign=1e-14 sigp=3.e-14 fast
#
method newton trap
solve prev
solve vdrain=0.01
solve vdrain=0.05
solve vdrain=0.2
solve vdrain=0.5
#
log outf=***tr.log
solve vgate=-2.0 ramptime=1e-9 dt=1e-11 tstop=1000e-9
(例四为:solve vgate=0.0 vstep=-0.1 vfinal=-1.0 name=gate )

```

运行结果

### 1. 瞬态响应




图 7.12 有无深能级的瞬态响应比较

## 2. 转移曲线




图 7.13 有无深能级的转移曲线比较

### Example 6: 交流分析深能级体陷阱

结构基于例四.交流模拟的初始点与前两个例子相同:栅极接地,漏压为 0.5V.交流信号轮流加在各个接触上.交流信号的频率从 1MHz 变化大 10GHz.参数 **fstep** 设置频率步长;**mult.f** 意味着 fstep 用作频率的乘法器而非加法器.

语句 **models** 用来定义模拟过程中合适的模型.**fldmob** 为倚赖电场的迁移率.**conmob** 为倚赖浓度的迁移率.不同浓度的迁移率可以从查找表中看到,且仅适用于室温.

## Experiment 8. BJT Device

### Example1. NPN Gummel plot and $f_T$ Extraction

目标结构:多晶硅发射极 NPN 双极晶体管.

目标参数:最终从 IV 曲线中提取出包括  $f_T$  和 Gain 在内的设计参数.

在该例中将使用:

- √ 多晶硅发射双极器件的工艺模拟.
- √ 在 DEVEDIT 中对结构网格重新划分.
- √ 直流分析: $V_{ce}$  不变, $V_{be}$  变化.
- √  $V_{be}$  变化过程中同步交流模拟.
- √ 提取  $f_T$  和 peak gain.

第一次硼注入形成本征基区;第二次硼注入自对准(self-aligned)于多晶硅发射区以形成一个连接本征基区和  $p+$  基极接触的 connection.多晶发射极旁的侧墙(spacer-like)结构用来隔开  $p+$  基极接触和提供自对准.在模拟过程中,relax 语句是用来减小结构深处的网格密度,从而只需模拟器件的一半.

经常遇到这样一种情况:一个网格可用于工艺模拟,但如果用于器件模拟效果却不甚理想.在这种情况下,可以用网格产生工具 DEVEDIT 用来重建网格,从而以实现整个半导体区域内无钝角三角形.

ATLAS 中的解过程:

- ① 设置集电极偏压为 2V.
- ② 用 log 语句用来定义 Gummel plot 数据集文件.
- ③ solve 语句上的参数 ac 激活交流分析.信号频率设为 1MHz.
- ④ 用 extract 语句提取 BJT 的最大增益"maxgain"以及最大  $f_T$ ,"maxft".

Gummel plot:晶体管的集电极电流  $I_c$ 、基极电流  $I_b$  与基极-发射极电压  $V_{be}$  关系图(以半对数坐标的形式).

模拟程序

go athena

#TITLE: Polysilicon Emitter Bipolar Example - Ssuprem4->Devedit->Spisces2

# 网格定义 ATHENA Mesh Define (利用该菜单即可进行网格定义,下略)

line x loc=0.0 spacing=0.03

line x loc=0.2 spacing=0.02

line x loc=0.24 spacing=0.015

line x loc=0.3 spacing=0.015

line x loc=0.8 spacing=0.15

#

line y loc=0.0 spacing=0.01

```

line y loc=0.07      spacing=0.01
line y loc=0.1       spacing=0.01
line y loc=0.12      spacing=0.01
line y loc=0.3       spacing=0.02
line y loc=0.5       spacing=0.06
line y loc=1.0       spacing=0.35

```

```

# 衬底定义:n 型 ATHENA Mesh Initialize
init silicon c.arsenic=2e16 orientation=100

```

```

# 本征基区 ATHENA Implant & ATHENA Diffuse
implant boron energy=18 dose=2.5e13 crystal
diffuse time=60 temp=920 nitro

```

```

# 沉积多晶硅作发射极 ATHENA Deposit
deposit poly thick=0.3 divisions=6
# 多晶硅掺杂 ATHENA Implant
implant arsenic dose=7.5e15 energy=50 crystal
# 刻蚀多晶硅 ATHENA Etch
etch poly right p1.x=0.2

```

```

# 网格密度降低 (可选步骤)
relax y.min=.2 x.min=0.2
relax y.min=.2 x.min=0.2

```

```

# 多晶硅氧化,形成掩蔽层 ATHENA Diffuse
method compress fermi
diffuse time=25 temp=920 dryo2 press=1.00

```

```

# 退火处理 ATHENA Diffuse
diffuse time=50 temp=900 nitrogen press=1.00

```

```

# 离子注入,使本征基区与 p+基接触有一连接 ATHENA Implant
implant boron dose=2.5e13 energy=18      # 第二次硼注入

```

```

# 形成侧墙
# ATHENA Deposit
deposit oxide thick=0.4 divisions=10 min.space=0.1
# ATHENA Etch
etch oxide dry thick=0.5

```

# 运行到此时,得到的结构图如图 8.1

```

# 形成基极接触 ATHENA Diffuse & Diffuse
implant boron dose=1e15 energy=30 crystal      # 第三次硼注入,形成 p+基区

```




图 8.1 形成侧墙

diffuse time=60 temp=900 nitrogen press=1.00

# 镜像以得到完整结构 ATHENA Mirror  
structure reflect left

# 沉积铝和刻蚀铝以得到接触 ATHENA Deposit & Etch

deposit alum thick=0.05 div=2

# 沉积铝后的结构图和待刻蚀区域见图 8.2 及 8.3

etch alum start x=-0.16 y=-4

etch continue x=-0.16 y=0.2

etch continue x=-0.6 y=0.2

etch done x=-0.6 y=-4

etch alum right p1.x=0.15




图 8.2 在完整 BJT 结构上沉积铝




图 8.3 阴影部分 Al 为待刻蚀区域

```

# 电极定义 ATHENA Electrode
electrode x=0.0    name=emitter
electrode x=-0.65  name=base
electrode backside name=collector

# 保存最终结构 ATHENA File I/O
structure outfile=bjtex01_0.str

# 重新定义网格
go devedit
base.mesh height=0.25 width=0.25
bound.cond apply=false max.ratio=300
constr.mesh max.angle=90 max.ratio=300 max.height=1 max.width=1 \
    min.height=0.0001 min.width=0.0001
constr.mesh type=Semiconductor default
constr.mesh type=Insulator default max.angle=170
constr.mesh type=Metal default max.angle=180

# 定义网格最小间隙
imp.refine min.spacing=0.025
# Select a list of solution (impurity) gradients to refine upon...
imp.refine imp="Arsenic" sensitivity=0.5
imp.refine imp="Boron" sensitivity=0.5

#
mesh
struct outfile=bjtex01_0.str
#
##### Gummel Plot Test #####
go atlas
# 设置材料模型、接触性质
# ATLAS Commands—>Models—>Material...
material number=2 taun0=5e-6 taup0=5e-6      # 硅中电子、空穴 SRH 复合寿命均为  $5 \times 10^{-6}$ s
# ATLAS Commands—>Models—>Contacts...
contact name=emitter n.poly surf.rec          # 定义发射极接触为 n+多晶硅,且多晶硅/硅
                                              # 界面的表面复合速度有限

# ATLAS Commands—>Models—>Models...
models bipolar print

# 数值方法定义
# ATLAS Commands—>Solutions—>Method... 选全耦和方法(newton)即可
method newton itlimit=25 trap atrap=0.5 maxetrap=4 autonr nrccriterion=0.1 \
    Tol.time=0.005 dt.min=1e-25

```

# 图 8.4

```

# 提供初值 write
solve init

# 设置集电极偏压
solve vcollector=2 local

# 设置基极偏压
solve vbase=0.1

# 保存&绘画  $I_C$ - $V_B$  关系图
log outf=bjtex01_2.log master
solve vbase=0.2 vstep=0.05 vfinal=0.9 name=base ac freq=1e6 aname=base

# Now dump a structure file, for tonyplotting... but first decide what
# you want in it, on top of the default quantities.....
output e.field flowlines jx.el jx.ho jy.el jy.ho
save outf=bjtex01_3.str
tonyplot bjtex01_3.str

# 提取器件参数
extract name="peak collector current" max(curve(abs(v."base"),abs(i."collector"))))
extract name="peak gain" max(i."collector"/ i."base")
extract name="max fT" max(g."collector""base"/((2*3.1415*c."base""base")))
#
tonyplot bjtex01_2.log

# 获取能带结构
solve init
output qfn qfp con.band val.band
save outfile=bjtex01_band.str
tonyplot bjtex01_band.str

# 退出模拟
quit

```

运行结果

```

peak collector current=0.000397951 A (最大集电极电流  $I_{CM}$ )
peak gain=83.1365
max fT=7.69477e+09

```

特征频率:使集电极电流与基极电流之比下降到 1 的信号频率,也就是无法将输入信号放大时的频率.因此也称截至频率.

1. 结构图如图 8-4 所示






图 8.4 完整的 NPN 结构图

## 2. 净掺杂浓度分布

利用工具栏中 Cutline 工具(方法同前), 画出点(-0.7,0)到点(-0.7,5),以及点(0,-0.28)点到(0,1).最后结果如图. 可以看出:发射极、基极、集电极的净掺杂浓度分别为 10 的 19、17(接触处为 19)、16 次方量级.




图 8.5 NPN 的掺杂浓度分布图

### 3. $I_c$ - $V_B$ 关系图




图 8.6 NPN 的  $I_c$ - $V_B$  关系图

### 4. 能带图




图 8.7 NPN 的能带图

## Example 2: NPN-Gummel Plot and $I_c/V_{ce}$ Characterization

模拟程序

go atlas

TITLE Bipolar Gummel plot and  $I_C/V_{CE}$  with constant  $I_B$

# 网格定义

mesh

x.m l=0.0 spacing=0.15

x.m l=0.8 spacing=0.15

x.m l=1.0 spacing=0.03

x.m l=1.5 spacing=0.12

x.m l=2.0 spacing=0.15

y.m l=0.00 spacing=0.006

y.m l=0.04 spacing=0.006

y.m l=0.06 spacing=0.005

y.m l=0.15 spacing=0.02

y.m l=0.30 spacing=0.02

y.m l=1.00 spacing=0.12

# 材料定义

region num=1 silicon

# 电极定义

electrode num=1 name=emitter left length=0.8

electrode num=2 name=base right length=0.5 y.max=0

electrode num=3 name=collector bottom

# 定义 B、C、E 三极

doping reg=1 uniform n.type conc=5e15 # 衬底,集电极

doping reg=1 gauss n.type conc=1e18 peak=1.0 char=0.2

doping reg=1 gauss p.type conc=1e18 peak=0.05 junct=0.15 # 基极

doping reg=1 gauss n.type conc=5e19 peak=0.0 junct=0.05 x.right=0.8 # 发射极 (peak 指 y 轴)

doping reg=1 gauss p.type conc=5e19 peak=0.0 char=0.08 x.left=1.5 # 基极接触

# 模型定义

models conmob fldmob consrh auger print

# 定义接触

contact name=emitter n.poly surf.rec

# BJT 结构的保存&绘画

solve init

save outf=bjtex04\_0.str

tonyplot bjtex04\_0.str

# 数值方法定义

method newton autonr trap

# Gummel plot

#  $V_C$  从 0V 变化到 2V

solve vcollector=0.025

solve vcollector=0.1

solve vcollector=0.25 vstep=0.25 vfinal=2 name=collector #  $V_C=2V$

# 偏压  $V_b$  增加到 0.2V

solve vbase=0.025

solve vbase=0.1

solve vbase=0.2

log outf=bjtex04\_0.log #  $I_C$ - $V_B$  关系图 ( $V_B=0.3\sim 1V$ )

solve vbase=0.3 vstep=0.05 vfinal=1 name=base

tonyplot bjtex04\_0.log -set bjtex04\_0\_log.set

#  $I_C$ - $V_{ce}$  曲线族

#  $V_B=0\sim 0.7V$

log off

solve init

solve vbase=0.025

solve vbase=0.05

solve vbase=0.1 vstep=0.1 vfinal=0.7 name=base #  $V_B=0.7V$

# 设置电流边界条件:基极电流保持不变

contact name=base current

# 求解不同基极电流( $I_b=1、2、3、4、5\mu A$ )下的输出曲线

# MASTER 定义输出文件为标准结构形式.这种结构形式的文件可以在 TONYPLOT 中绘画.

solve ibase=1.e-6

save outf=bjtex04\_1.str master

solve ibase=2.e-6

save outf=bjtex04\_2.str master

solve ibase=3.e-6

save outf=bjtex04\_3.str master

solve ibase=4.e-6

save outf=bjtex04\_4.str master

solve ibase=5.e-6

save outf=bjtex04\_5.str master

# 输出曲线的保存

load inf=bjtex04\_1.str master

log outf=bjtex04\_1.log

solve vcollector=0.0 vstep=0.25 vfinal=5.0 name=collector

load inf=bjtex04\_2.str master

```

log outf=bjtex04_2.log
solve vcollector=0.0 vstep=0.25 vfinal=5.0 name=collector
load inf=bjtex04_3.str master
log outf=bjtex04_3.log
solve vcollector=0.0 vstep=0.25 vfinal=5.0 name=collector
load inf=bjtex04_4.str master
log outf=bjtex04_4.log
solve vcollector=0.0 vstep=0.25 vfinal=5.0 name=collector
load inf=bjtex04_5.str master
log outf=bjtex04_5.log
solve vcollector=0.0 vstep=0.25 vfinal=5.0 name=collector
# 将不同  $I_b$  时的输出曲线绘制在一张图上
tonyplot -overlay bjtex04_1.log bjtex04_2.log bjtex04_3.log bjtex04_4.log bjtex04_5.log -set
bjtex04_1_log.set
quit

```

运行结果

### 1. 浓度分布图

利用工具栏中 Cutline 工具(方法同前), 画出点(0.4,0)到点(0.4,1),点(1.2,0)到点(1.2,0.3) 以及点(1.8,0)到点(1.8,0.3).最后结果如图. 可以看出:发射极、基极、集电极的净掺杂浓度分别为 10 的 19、17(接触处为 19)、15 次方量级.




图 8.8 NPN 的浓度分布图

## 2. Gummel plot




图 8.9 Gummel Plot

## 3. $I_C$ - $V_C$ 关系图




图 8.10 NPN 的  $I_C$ - $V_C$  关系图

### Example 3: NPN 的集-发射击穿电压 BVCEO

集电极-发射极反向击穿电压:该电压是指当晶体管基极开路时,其集电极与发射极之间的最大允许反向电压,一般用 VCEO 或 BVCEO 表示.

结构如例二相同,不同是地方是网格设置.

模拟程序

```
go atlas
TITLE Bipolar BVCEO simulation
#
mesh
x.m l=0 spac=0.1
x.m l=2 spac=0.1
#
y.m l=0 spac=0.002
y.m l=1 spac=0.10
#####
region num=1 silicon
#
electrode num=1 name=emitter left length=0.8
electrode num=2 name=base right length=0.5 y.max=0
electrode num=3 name=collector bottom
#
doping reg=1 uniform n.type conc=5e15
doping reg=1 gauss n.type conc=1e18 peak=1.0 char=0.2
doping reg=1 gauss p.type conc=1e18 peak=0.05 junct=0.15
doping reg=1 gauss n.type conc=5e19 peak=0.0 junct=0.05 x.right=0.8
doping reg=1 gauss p.type conc=5e19 peak=0.0 char=0.08 x.left=1.5
#
save outf=bjtex05_0.str
tonyplot bjtex05_0.str -set bjtex05_0.set
#
contact name=emitter n.poly surf.rec
#####
# 定义电子/空穴的 SRH 寿命
material taun0=5e-6 taup0=5e-6

# 定义模型
models bipolar print
impact selb # 用于击穿分析的 Selberherr 碰撞电离模型.

# 定义计算方法
solve init
```

```

method newton trap    # 计算方法为牛顿迭代法.如果解过程分散,电极偏置降低为原来 1/2.
#
solve prev
solve vbase=0.025
solve vbase=0.05
solve vbase=0.2      #  $V_B=0.2V$ 
#
contact name=base current
#
method newton trap ir.tol=1.e-20 ix.tol=1.e-20

# 为了增加稳定性(稳定是该类模拟所必需的——采用零基极电流),解的第一步是基极电流
# 取一个小的值, $I_b=3.e-15 A/\mu m$ .
solve ibase=3.e-15

log outf=bjtex05.log master

# ramp collector voltage
solve vcollector=0.25
solve vcollector=0.5
solve vcollector=1
solve vcollector=3
solve vcollector=5

# 为了对集电极电流设限,采用参数"compl=<num> cname=collector".num 设置极限值,单位
#  $A/\mu m$ ,而 cname 表明测量的是哪个电极的电流.一旦集电极电流超过了设定的极限值,模拟
# 将跳过循环,而进行下一句模拟.
solve vstep=0.5 vfinal=10 name=collector
compl=5.e-11 e.comp=3    #  $V_C=0.25\sim 10V$ 

# plot results
tonyplot  bjtex05.log -set bjtex05_log.set
quit

```




图 8.11 NPN 的击穿特性曲线



## Experiment 9. Solar Cell

### 基础知识

不论是一般的化学电池还是太阳能电池,其输出特性一般都是用如下图所示的电流-电压曲线来表示.由光电池的伏安特性曲线,可以得到描述太阳能电池的四个输出参数.




图 9.1 太阳电池和锰干电池的 IV 曲线

#### ① 开路电压 $V_{oc}$

$R=\infty, I=0$ , 即:  $I_L=I_F$ . 将  $I=0$  代入光电池的电流电压方程, 得开路电压为:

$$V_{oc} = \frac{kT}{q} \ln\left(\frac{I_L}{I_S} + 1\right) \quad (9.1)$$

#### ② 短路电流 $I_{sc}$

如将 pn 结短路 ( $V=0$ ), 因而  $I_F=0$ , 这时所得的电流为短路电流  $I_{sc}$ . 显然, 短路电流等于光生电流, 即:  $I_{sc}=I_L$ .

#### ③ 填充因子 FF

它表示了最大输出功率点所对应的矩形面积在  $V_{oc}$  和  $I_{sc}$  所组成的矩形面积中所占的百分比. 特性好的太阳能电池就是能获得较大功率输出的太阳能电池, 也就是  $V_{oc}$ ,  $I_{sc}$  和 FF 乘积较大的电池. 对于有合适效率的电池, 该值应在 0.70-0.85 范围之内.




图 9-2 太阳电池的  $I-V$  特性和工作点

图中(●)表示最佳工作点, (○)表示实际工作点

其值越大表示太阳能电池的输出功率越大. FF 的值始终小于 1. FF 可由下列经验公式给出:

$$FF = \frac{V_{oc} - \ln(V_{oc} + 0.72)}{V_{oc} + 1} \quad (9.2)$$

式中  $V_{oc}$  是归一化的开路电压, 即  $U_{oc}/(nKT/q)$ . 当  $V_{oc} > 15$  时, 该公式的精度可达 4 位有效数字. 实际上, 由于受串联电阻和并联电阻的影响, 实际太阳能电池填充因子的值要低于上式所给出的理想值.

#### ④ 能量转化效率 $\eta$

它表示入射的太阳光能量有多少能转换为有效的电能.

$$\begin{aligned} \eta &= (\text{太阳能电池的输出功率} / \text{入射的太阳光功率}) \times 100\% \\ &= (V_{OP} I_{OP} / P_{in} S) \times 100\% \end{aligned}$$

$$= \frac{V_{oc} I_{sc} FF}{P_{in} S} \quad (9.3)$$

## Example 1: 单晶硅太阳能电池

在本例中, 将首先用 ATHENA 模拟的是晶体硅太阳能电池的结构(pn 结), 接着用 ATLAS 模拟太阳能电池的一些基本性质:  $V_{oc}$ 、 $I_{sc}$ 、光谱反应及考虑增加增透膜后对光谱反应的影响.

在太阳能电池的模拟实验中, 将考虑到外部光源的影响. 因此, 模拟中将用到 **BEAM** 语句.

例 1: 单色光

```
beam num=1 wavelength=0.6 x=0.5 y=-2.0 ang=90.0 min=-0.1 max=0.1 \
Min.power=0.05
```

该语句定义了一传输角度为 90 度、波长为  $0.6 \mu\text{m}$  的单色光, 其起始坐标为 (0.6, -0.5). 当光强降至初始光功率的 5% 时传输终止. (入射光线图如图 10.3)

例 2: 多光谱光线

```
beam num=2 x=0.0 y=-1.0 ang=45.0 power.file=source.spc wavel.start=0.4 \
wavel.end=0.6 wave.num=4
```

该语句定义一起始于 (0.0, -1.0)、入射角度为 45 度的多光谱光线. 该光线由光谱文件 source.spc 引入, 波长范围为  $0.4 \mu\text{m}$  至  $0.6 \mu\text{m}$ .




图 9.3 入射光线图

此外, 本实验还将用到增透膜(或称减反射膜, anti - reflection coating). INTERFACE 语句可用来

定义增透膜及其界面的光学性质. 参数 `ar.Chick` 定义膜厚; `ar.Index` 定义折射率. 绝大多数情况下, 单层增透膜厚度等于  $1/4$  波长(在膜里). 对于多层膜, 必须通过参数 `coating` 和 `layer` 来详细阐述. 如果膜由吸收物质制作而成, 还应该引入 `ar.Absorb` 参数进行考虑.

模拟程序

```
go athena
# 网格定义
line x loc=0.00 spac=1
line x loc=10 spac=1
#
line y loc=0.00 spac=0.05
line y loc=0.25 spac=0.02
line y loc=1 spac=0.1
line y loc=50 spac=10

# 衬底定义
init silicon c.boron=1.0e14 orientation=100

# 淀积氧化层
deposit oxide thickness=0.05 div=1

# 离子注入形成 n+层
implant phos dose=1e15 energy=30

# 退火处理
diffuse time=10 temp=900

# 提取结深
extract name="junc_depth" xj material="Silicon" mat.occno=1 x.val=0.1 junc.occno=1

# 形成接触
etch oxide right p1.x=8
deposit alum thickness=0.1 div=3
etch alum left p1.x=8

# 网格间距加大
# relax y.min=0.6 是指将 y 方向上从 0.6 到 y 的最大值范围内的网格减少一半
relax y.min=0.6
relax y.min=2.0
relax y.min=10

# 镜像以获得完整结构
structure mirror right

# 电极定义
```

```

electrode name=cathode x=10
electrode name=anode backside

# 保存结构
structure outf=optoex08_0.str

go atlas
# 设置接触材料铝不透明
material material=Aluminum imag.index=1000

# 设置硅的性质
# 衬底硅中电子、空穴的复合寿命为 1e-6s
material material=Silicon taun0=1e-6 taup0=1e-6

# 光照性质定义
# 定义光线是单色光,出发点为(10.0,-2.0),垂直入射,且太阳光谱文件来自外部文件（指文件
# optoex08.spec,它包含 AM0 太阳光谱数据.）
beam num=1 x.origin=10.0 y.origin=-2.0 angle=90.0 power.file=optoex08.spec

# 保存光强性质至 solution 文件
output opt.int
# 模型定义
models conmob fldmob srh print

solve
# 获取短路电流
log outf=optoex08_0.log
solve b1=1.0e-15 # 定义光功率
extract name="short_circuit_current" max(abs(i."cathode"))
save outf=optoex08_1.str

# 获取开路电压
solve init
contact name=cathode current # 为了设定电流为零,先设置电流边界条件
solve icathode=0 b1=1e-15
extract name="open_circuit_voltage" max(abs(vint."cathode"))
save outf=optoex08_2.str
tonyplot optoex08_2.str

go atlas
#
# 光谱反应(spectral response)求解
#
material material=Aluminum imag.index=1000

```

```

#
material material=Silicon  taun0=1e-6 taup0=1e-6
#
beam  num=1 x.origin=10.0 y.origin=-2.0 angle=90.0

#
output opt.int

models conmob fldmob srh print

# spectral response
solve init b1=0
log outf=optoex08_2.log
solve b1=1 lambda=0.3
solve b1=1 lambda=0.35
solve b1=1 lambda=0.4
solve b1=1 lambda=0.45
solve b1=1 lambda=0.5
solve b1=1 lambda=0.55
solve b1=1 lambda=0.6
solve b1=1 lambda=0.65
solve b1=1 lambda=0.7
solve b1=1 lambda=0.75
solve b1=1 lambda=0.8
solve b1=1 lambda=0.85
solve b1=1 lambda=0.9
solve b1=1 lambda=0.95
solve b1=1 lambda=1.00

tonyplot optoex08_2.log -set optoex08_3.set

go atlas
#
# 考虑增透膜的情形
#
# 网格重新定义
mesh space.mult=1.0
x.mesh loc=0.0  spacing=10.0
x.mesh loc=10.0 spacing=10.0
y.mesh loc=0.0  spacing=0.2
y.mesh loc=50.0 spacing=0.2

region num=1 material=Silicon

```

```

#
elec name=cathode bottom
#
doping uniform conc=1e14 n.type

# define a beam (be sure to include REFLECT and BACK parameters)
# 定义入射线数目、起始点、入射角度、考虑反射及窗口大小、最小光功率
beam num=1 x.origin=5.0 y.origin=-1.0 angle=90.0 back.refl front.refl reflect=5 min.w=-2
    \ max.w=2 min.power=0.001

solve init

log outf=optoex09_noarc.log
solve      b1=1 lambda=0.3 index.check
solve      b1=1 lambda=0.325 index.check
solve      b1=1 lambda=0.35 index.check
solve      b1=1 lambda=0.375 index.check
solve      b1=1 lambda=0.4 index.check
solve      b1=1 lambda=0.425 index.check
solve      b1=1 lambda=0.45 index.check
solve      b1=1 lambda=0.475 index.check
solve      b1=1 lambda=0.5 index.check
solve      b1=1 lambda=0.525 index.check
solve      b1=1 lambda=0.55 index.check
solve      b1=1 lambda=0.575 index.check
solve      b1=1 lambda=0.6 index.check
solve      b1=1 lambda=0.625 index.check
solve      b1=1 lambda=0.65 index.check
solve      b1=1 lambda=0.675 index.check
solve      b1=1 lambda=0.7 index.check
solve      b1=1 lambda=0.725 index.check
solve      b1=1 lambda=0.75 index.check
solve      b1=1 lambda=0.775 index.check
solve      b1=1 lambda=0.8 index.check
log off

go atlas
mesh space.mult=1.0
x.mesh loc=0.0  spacing=10.0
x.mesh loc=10.0 spacing=10.0
y.mesh loc=0.0  spacing=0.2
y.mesh loc=50.0 spacing=0.2

region num=1 material=Silicon

```

```

#
elec name=cathode bottom
#
doping uniform conc=1e14 n.type

# define a beam (be sure to include REFLECT and BACK parameters)
beam num=1 x.origin=5.0 y.origin=-1.0 angle=90.0 back.refl front.refl reflect=5 min.w=-2
\ max.w=2 min.power=0.001

# 定义增透膜（一层）
interface optical ar.index=2.05 ar.thick=0.07 p1.x=0.0 p1.y=0.0 p2.x=10.0 p2.y=0.0

solve init

log outf=optoex09_1arc.log

solve      b1=1 lambda=0.3 index.check
solve      b1=1 lambda=0.325 index.check
solve      b1=1 lambda=0.35 index.check
solve      b1=1 lambda=0.375 index.check
solve      b1=1 lambda=0.4 index.check
solve      b1=1 lambda=0.425 index.check
solve      b1=1 lambda=0.45 index.check
solve      b1=1 lambda=0.475 index.check
solve      b1=1 lambda=0.5 index.check
solve      b1=1 lambda=0.525 index.check
solve      b1=1 lambda=0.55 index.check
solve      b1=1 lambda=0.575 index.check
solve      b1=1 lambda=0.6 index.check
solve      b1=1 lambda=0.625 index.check
solve      b1=1 lambda=0.65 index.check
solve      b1=1 lambda=0.675 index.check
solve      b1=1 lambda=0.7 index.check
solve      b1=1 lambda=0.725 index.check
solve      b1=1 lambda=0.75 index.check
solve      b1=1 lambda=0.775 index.check
solve      b1=1 lambda=0.8 index.check
log off

go atlas

mesh space.mult=1.0
x.mesh loc=0.0 spacing=10.0
x.mesh loc=10.0 spacing=10.0

```

```

y.mesh loc=0.0    spacing=0.2
y.mesh loc=50.0   spacing=0.2

region num=1 material=Silicon
#
elec  name=cathode bottom
#
doping uniform conc=1e14 n.type

# define a beam (be sure to include REFLECT and BACK parameters)
beam num=1 x.origin=5.0 y.origin=-1.0 angle=90.0 back.refl front.refl reflect=5 min.w=-2
    \ max.w=2 min.power=0.001

# 定义增透膜（两层）
interface optical material=Oxide ar.thick=0.06 \
    x.min=0.0 y.min=0.0 x.max=10.0 y.max=0.0 z.min=0 z.max=10
interface optical material=Nitride ar.thick=0.05 coating=1 layer=2

solve init

log outf=optoex09_2arc.log

solve      b1=1 lambda=0.3 index.check
solve      b1=1 lambda=0.325 index.check
solve      b1=1 lambda=0.35 index.check
solve      b1=1 lambda=0.375 index.check
solve      b1=1 lambda=0.4 index.check
solve      b1=1 lambda=0.425 index.check
solve      b1=1 lambda=0.45 index.check
solve      b1=1 lambda=0.475 index.check
solve      b1=1 lambda=0.5 index.check
solve      b1=1 lambda=0.525 index.check
solve      b1=1 lambda=0.55 index.check
solve      b1=1 lambda=0.575 index.check
solve      b1=1 lambda=0.6 index.check
solve      b1=1 lambda=0.625 index.check
solve      b1=1 lambda=0.65 index.check
solve      b1=1 lambda=0.675 index.check
solve      b1=1 lambda=0.7 index.check
solve      b1=1 lambda=0.725 index.check
solve      b1=1 lambda=0.75 index.check
solve      b1=1 lambda=0.775 index.check
solve      b1=1 lambda=0.8 index.check
tonyplot -overlay optoex09_noarc.log optoex09_1arc.log optoex09_2arc.log -set optoex09.set

```



quit

模拟结果

1.结构图




图 9.4 单晶硅太阳能电池的结构

2. 光生率分布图

借助 SILVACO 的 TONYPLOT 模块,使研究在光照下太阳能电池内的载流子光生细节变得可能.这对模拟多结器件尤为重要.下图显示的是硅太阳能电池光生率的分布图.注意在图中,器件的中间部分有一个不透明的金属电极.一旦获得了光生率,如果再知道终端电流,那么就可以估计太阳能电池的量子效率.




图 9.5 硅太阳能电池的光生率分布图

3. 光谱反应图

上图画出了源光电流、可利用的光电流以及实际负极电流随波长变化的对比图.绿线即光源等价电流,红线为在光照下太阳能电池可利用的电流,而蓝线为实际负极电流.其中,前两者的差异由反射和传送(部分光穿过器件)引起.如果不考虑最短的波长部分,该原因是引起损耗的主要原因.后两者的损耗则是由复合引起.由本图还可以确定电池的外量子效率——为可利用光电流与源光电流的比值.




图 9.6 硅太阳能电池的光谱反应

4. 增透膜对负极电流的影响




图 9.7 有无增透膜对太阳能电池输出电流的影响

上图表明,对于一层 ARC,0.6μm 波长附近时负极就有电流通过.一个二层膜,可在很宽的波长范

围内降低折射率.

## Example 2: PIN 晶体硅/非晶硅太阳能电池

本例将模拟的是 pin 型 c-Si/a-Si 太阳能电池.对该软件而言,非晶硅是一个新物质.故需采用定义新物质的方法来定义非晶硅.定义新物质的第一步是在构建所需结构时用软件中已有物质代替(半导体用半导体代替,金属用金属代替,如是.).第二步是在进行器件模拟时必须定义物质相关参数,如带隙、亲和能、介电常数等等(半导体、绝缘体、导体所需参数略有不同).

模拟程序

```
go athena
#
line x loc=0.00 spac=10
line x loc=300 spac=10
#
line y loc=0.00 spac=0.001
line y loc=0.3 spac=0.001
line y loc=0.6 spac=0.02
line y loc=1 spac=0.2
line y loc=2 spac=1
line y loc=5 spac=15
line y loc=350 spac=15

# pc-Si
init silicon c.boron=1.0e16 orientation=100

# ia-Si
deposit polysilicon thick=0.005 div=5

# na-Si
deposit polysilicon thick=0.02 c.phosphor=1.0e18 div=10

# 淀积氧化层
deposit oxide thickness=0.02 div=1

# 形成接触
etch oxide start x=145.00 y=-0.045
etch cont x=155.00 y=-0.045
etch cont x=155.00 y=-0.025
etch done x=145.00 y=-0.025

# 沉积&刻蚀铝
deposit alum thickness=0.02 div=1
etch aluminum above p1.y=-0.045
```

```

# 电极定义
electrode name=cathode x=150.00
electrode name=anode backside
structure outf=0.str

go atlas
# 设置接触物质铝非透明
material material=Aluminum imag.index=1000

# 设置非晶硅的性质
material material=polysilicon taun0=1e-6 taup0=1e-2 affinity=3.86 \
    EG300=1.70 NC300=2.5e+20 NV300=2.5e+20 mun=2 mup=1e-3 \
    VSATN=1e+7 VSATP=1e+7

# 设置硅的性质
material material=silicon taun0=1e-6 taup0=1e-6

# 保存光强至 solution 文件
output opt.int

# 模型定义
models conmob fldmob srh print

# 接触定义
# while r=100,con.resist=100*10um*1um=1e-5 ohm*cm^2
contact name=cathode con.resist=1E-5          # 考虑外界负载

# 初始解
solve init

# 光照定义
beam num=1 x.origin=150.0 y.origin=-1 angle=90 wavelength=1.101

# 获取漏电流
log outf=0_Is.log
solve b1=0
extract name="0_leak-a" max(abs(i."anode"))
extract name="0_leak-c" max(abs(i."cathode"))

# 获取短路电流
log outf=0_Isc.log
solve b1=0.0609
extract name="short_circuit_current" max(abs(i."anode"))

```

```

save outf=0_Isc.str
tonyplot 0_Isc.str

# 获取开路电压
solve init
contact name=anode current
solve ianode=0 b1=0.0609
extract name="open_circuit_voltage" max(abs(vint."anode"))
save outf=0_Voc.str
tonyplot 0_Voc.str

```

运行结果:

```

0_leak-a=8.17859e-20
0_leak-c=3.86358e-24
short_circuit_current=3.23193e-08
open_circuit_voltage=0.522253

```

接下来,改变外界负载的大小(这次取 0ohm 和 100ohms),并利用 FF 经验公式及  $\eta$  计算公式,可得出如下结论:

R(Ω)	Leak_a(A)	Leak_c(A)	I <sub>sc</sub> (A)	U <sub>oc</sub> (V)	FF	$\eta$ (%)
0	8.18e-20	1.46e-24	3.232e-8	0.522	0.809	9.100
50		3.86e-24				
100		2.12e-23				

表 9.1 外界负载对太阳能电池的影响

初步结论: 外接电阻越大, 负极漏电流越大; 而对四个基本特性参数无影响.

## Experiment 10. TFT

### 基础知识

制作 TFT 的最常用材料:氢化非晶硅(a-Si:H)与多晶硅.

最常用结构:分为顶端栅极结构和底部栅极结构两种.(衬底:玻璃/石英或是覆盖薄  $\text{SiO}_2$  层的硅衬底等绝缘衬底)

由于工艺限制,氢化非晶硅 TFT 常采用底部栅极结构,多晶硅 TFT 多采用顶端栅极结构.

详见《半导体器件物理与工艺》P202-204

### Example1. 钝化的非晶硅 TFT 器件

该例将模拟钝化的 a-Si:H TFT 器件的  $I_d/V_{gs}$  特性.它包括:

- √ 用 ATLAS 模拟结构;
- √ 钝化 a-Si 的 material 和 model 设置;
- √ 正偏  $I_d/V_{gs}$  特性:漏压 5V,栅压-1~20V.

**TFT 模拟中的关键命令是 `defect` 语句.它将定义硅中一系列连续的陷阱态浓度和相关陷阱横截面.**

$I_d/V_{gs}$  特性模拟与模拟 MOS 器件阈值电压方法类似.结果可与非钝化 a-Si 器件作比较.

### 模拟程序

```
go atlas
```

```
TITLE    a-si TFT simulation
```

```
# 网格定义
```

```
mesh
```

```
x.m      l=0      spac=0.5
```

```
x.m      l=2      spac=0.5
```

```
x.m      l=3.5    spac=1.0
```

```
x.m      l=5      spac=0.25
```

```
x.m      l=10     spac=1.5
```

```
x.m      l=15     spac=0.25
```

```
x.m      l=16.5   spac=1.0
```

```
x.m      l=18     spac=0.5
```

```
x.m      l=20     spac=0.5
```

```
#
```

```
y.m      l=-0.1   spac=0.05
```

```
y.m      l=0      spac=0.0075
```

```
y.m      l=0.05   spac=0.01
```

```
y.m      l=0.1    spac=0.0075
```

```
y.m      l=0.5    spac=0.25
```

```
y.m      l=10     spac=5
```

### # 区域定义

```
# 1=oxide 2=silicon 3=oxide
region num=1 y.max=0. oxide
region num=2 y.min=0. y.max=0.1 silicon
region num=3 y.min=0.1 oxide
```

### # 电极定义

```
# 1=gate 2=substrate 3=source 4=drain
elec num=1 x.min=5 x.max=15 y.min=-0.1 y.max=-0.1 name=gate
elec num=2 substrate name=substrate
elec num=3 x.min=0. x.max=2. y.min=0. y.max=0. name=source
elec num=4 x.min=18.0 x.max=20.0 y.min=0. y.max=0. name=drain
```

### # 掺杂定义

```
doping reg=2 uniform conc=7.e14 n.type
doping reg=2 gauss conc=3.e18 n.type x.right=5 char=0.3
doping reg=2 gauss conc=3.e18 n.type x.left=15 char=0.3
```

### # 非晶硅参数设置

```
material region=2 mun=20 mup=1.5 nc300=2.5e20 nv300=2.5e20 eg300=1.9
```

#

```
defects nta=1.e21 ntd=1.e21 wta=0.033 wtd=0.049 \
nga=1.5e15 ngd=1.5e15 ega=0.62 egd=0.78 wga=0.15 wgd=0.15 \
sigtae=1.e-17 sigtah=1.e-15 sigtde=1.e-15 sigtdh=1.e-17 \
siggae=2.e-16 siggah=2.e-15 siggde=2.e-15 siggdh=2.e-16
```

### # 栅极接触定义

```
contact num=1 alum
```

### # 模型定义

```
models temp=300
```

### # 计算方法定义

```
method itlimit=30
```

### # 结构保存&绘画

```
solve init
save outf=tftex01_0.str
tonyplot tftex01_0.str -set tftex01_0.set
```

### # 计算方法定义

```
method gummel newton
```

```

solve vgate=-1.0
method newton
solve vdrain=0.1
solve vdrain=0.2
solve vdrain=0.5
solve vdrain=1 vfinal=5 vstep=1 name=drain
#  $V_d$ 

```

#### # Id-Vg 曲线保存&绘画

```

log outf=tfex01_1.log
solve vstep=0.5 vfinal=20. name=gate vdrain=5
#  $V_g$  ramp
tonyplot tfex01_1.log -set tfex01_log.set

```

#### # 提取阈值电压、漏电流、亚阈值斜率 (略)

运行结果:

$v_t=2.24776$  V

$leak=1.55261e-06$  A/um

$subvt=0.666109$  V/decade




图 10.1 顶端栅极 TFT 结构图

1. 结构图如图 10-1 所示
2. 净掺杂浓度分布图见图 10-2






图 10.2 顶端栅极 TFT 的净掺杂浓度分布图

### 3. $I_d$ - $V_g$ 关系曲线图




图 10.3 顶端栅极 TFT 的  $I_d$ - $V_g$  关系曲线图

### Example 2: 非钝化的非晶硅 TFT 器件

在例一中,我们模拟了钝化的非晶硅 TFT,接下来,将对非钝化的非晶硅 TFT 进行模拟.因为结构一致,工艺一致.不同的只是非晶硅的性质.我们只需对例一作稍微修改,即可达到目的:将例一中 DEFECT 语句中 **nga、ngd(类受主、施主能态高斯分布峰对应的能量)**数值由  $1.5e15$  改为  $4.5e15$ .

运行结果

$v_t=2.30924$  V

$leak=1.54646e-06$  A/um

$subv_t=0.513561$  V/decade

结论:非钝化的非晶硅  $V_t$  较大.

### Example 3: 正/负栅压特性

在前面两例中,我们模拟了顶端栅极结构 TFT 及其转移特性.在本例中将模拟 TFT 的另一种结构——底部栅极结构.在此基础上,我们将考虑正负栅压下漏极电流的变化.

衬底定义为氧化硅.该晶体管的金属栅在底部,栅极绝缘体由氧化层和氮化层组成.淀积的轻掺硅层作为沟道区域(**硅的上半部分为重掺,下半部分轻掺.采用的是分两步淀积**).接着定义源/漏接触.然后刻蚀形成源/漏(注意刻蚀方法:回刻 (etchback) 金属和重掺硅;同时一些轻掺硅也刻掉了.)

解的方法是:漏压+10V 保持不变,而栅压从-20V 变化到+20V.

在 TFT 的模拟中,最重要的必要条件是设置 **DEFECT** 语句来定义半导体带隙中的能态密度.缺陷态定义为类施主和类受主及尾和隙间高斯态(tail and mid-gap gaussian states).本例中的该值为 TFT 中非晶硅的典型值,但是每个用户制程不同.因为缺陷态密度敏感于器件的氢退火条件,所以用户可以自动调节.接下来定义每个半/绝界面存在一个界面电荷.用 **INTERFACE** 语句的 bounding box parameters 就有可能通过位置改变这个电荷.

TFT 模拟所需要的模型比较简单.

模拟程序

```
go athena
# ATHENA TFT formation -> ATLAS Id/Vg extraction example
# 网格定义
line x loc=0   spac=0.2
line x loc=1.0 spac=0.1
line x loc=3.0 spac=0.25

line y loc=0   spac=0.1
line y loc=0.2 spac=0.1

# 衬底定义
initialize oxide

# 沉积铝作为栅极
deposit aluminum thick=0.30 spaces=2

# 沉积 MIS 绝缘层
deposit oxide thick=0.20 spaces=4
deposit nitride thick=0.2 spaces=4

# 沉积非晶硅
deposit silicon thick=0.15 div=10 c.phos=1e14
deposit silicon thick=0.05 c.phos=1.0e20 div=4
# 沉积源/漏接触
deposit aluminum thick=0.2 div=2

# 沉积阻挡层
deposit barrier thick=0.1 div=1
etch barrier p1.x=1.0 right
etch aluminum dry thick=0.20
etch silicon dry thick=0.10
etch barrier all
```

```

# 沉积氧化层
deposit oxide thick=0.2 div=2

# 镜像以获得完整结构
structure mirror right

# 电极定义
electrode name=source x=0 y=-1.0
electrode name=gate x=3
electrode name=drain x=6 y=-1.0

# 结构保存&绘画
structure outf=tftex05_0.str
tonyplot tftex05_0.str -set tftex05_0.set

go atlas
# Title a-si TFT simulation
# 非晶硅参数设置
material material=silicon mun=20 mup=1.5 nc300=2.5e20 \
nv300=2.5e20 eg300=1.8 taun0=1e-8 taup0=1e-8
#
defects nta=1.e21 ntd=1.e21 wta=0.05 wtd=0.05 \
nga=1.0e16 ngd=1.0e16 ega=0.6 egd=0.6 wga=0.3 wgd=0.3 \
sigtae=1.e-17 sigtah=1.e-15 sigtde=1.e-15 sigtdh=1.e-17 \
siggae=2.e-16 siggah=2.e-15 siggde=2.e-15 siggdh=2.e-16
interface qf=3e10

# 栅极定义
contact name=gate alum

# 模型定义
# 常迁移率模型及模拟反向漏电流的带间隧穿模型("models bbt.std"),bb.b 是带间隧穿模型中
# 的指数参数,在这里,bb.b=7.5e6 是用来模拟非晶硅.
models srh temp=300 print bbt.kl bb.b=7.5e6

# 计算方法定义
method newton

# 求初始估计值
solve init

# 变化漏压至+10V.然后保存这时的结构
solve prev vdrain=0.1
solve prev vdrain=0.2

```

```

solve prev vdrain=0.5
solve vdrain=1 vfinal=10 vstep=1 name=drain      #  $V_d=10V$ 
#
output e.field
save outf=vds.str                                # 电场信息储存在 vds.str 文件中

# Solve  $I_d$  versus reverse  $V_g$  curve
# 反向栅压变化至-20V.在此期间,因为 TFT 的隧穿效应,漏极电流将增加(值的大小受带间隧穿
# 穿模型参数控制).最高反偏栅压的结构文件将进行保存.
log outf=txtex05_1.log
solve vgate=0 vstep=-0.5 vfinal=-20. name=gate    #  $V_g=-0.5\sim-20V$ 
save outf=txtex05_1.str
tonyplot txtex05_1.str -set txtex05_1.set
# 提取最大反向电流
extract name="max_rev_id" y.val from curve(v."gate",i."drain") where x.val=-20

# Solve  $I_d$  versus forward  $V_g$  curve
load inf=vds.str master
log outf=txtex05_2.log
solve vgate=0 vstep=0.5 vfinal=20. name=gate      #  $V_g=0\sim20V$ 
# 提取亚阈斜率 (subthreshold slope)
# 亚阈值 leakage slope,敏感于半导体内缺陷分布,其典型值为 0.5-1.5 V/decade.
extract name="subvtslope" 1.0/slope(maxslope(curve(abs(v."gate"),log10(abs(i."drain")))))

```

#### # $I_d$ - $V_g$ 曲线图

```

tonyplot -overlay txtex05_1.log
txtex05_2.log -set txtex05_log.set
quit

```

运行结果

1. 结构图如图 10-4 所示




图 10-4 底部栅极 TFT 结构图

## 2. 转移特性




图 10.5 TFT 的  $I_d$ - $V_g$  曲线图

## Appendix

### Appendix A: ATLAS 中常用 STATEMENTS 及其部分参数

#### BEAM 语句:

BEAM 语句用来定义以校准的光束形式输入的光信号.

**angle:** 定义光束传输的角度(参看图 10.3), 单位为度.  $\text{angle}=90$  表示从器件顶端垂直光照下来.

该参数的同义字是  $\phi$ .

**back.refl:** 表示考虑底部的反射. 当定义了该参数, 器件之外的区域假设为真空(即  $n=1.0, k=0.0$ ).

**b<n>:** 定义光线数目( $n$ )及光功率, 单位为  $\text{W}/\text{cm}^2$ . 其中, 光线数目必须是 1 到 10 之间的整数.

**front.refl:** 表示考虑前表面的反射. 当定义了该参数, 器件之外的区域假设为真空.

**reflect:** 定义跟踪的反射次数.

**max.window、:** 定义相对于光束坐标系统照明窗口光线的最大  $X$  值(参见图 10.3), 单位为  $\mu\text{m}$ .

该参数的同义字是  $x_{\text{max}}$ .

**min.window:** 定义相对于光束坐标系统照明窗口光线的最小  $X$  值. 该参数的同义字是  $x_{\text{min}}$ .

**min.power:** 定义相对于给定光源的最小光强, 单位为  $\text{W}/\text{cm}^2$ . 这对限制光照光线的数目非常有用.

**number:** 定义光线的数目(从 1 到 10). 这个数目用在 SOLVE 语句中, 用来定义不同光线的相对强度.

**power.file:** 定义一个光谱文件的文件名.

**wavel.start:** 定义当进行多光谱模拟时光源光线的最小波长(真空中).

**wavel.end:** 定义当进行多光谱模拟时光源光线的最大波长(真空中).

**wavel.num:** 定义多光谱模拟中的波长数.

#### CONTACT 语句:

与半导体材料相接的电极的接触类型默认为欧姆接触. 如果定义了功函数, 则电极看作肖特基接触. CONTACT 语句用来定义一个或多个电极的功函数.

**all:** 定义所有电极的性质相同.

**aluminum:** 定义电极材料为铝.

**barrier:** 开启肖特基势垒降低机制.

**con.resist:** 定义分布接触电阻, 单位  $\Omega\text{cm}^2$ . 不能同时定义 RESISTANCE 跟 CON.RESIST.

**name:** 定义电极的名字.

**number:** 将要定义电极的数目.

**resistance:** 定义集总电阻值, 单位  $\Omega\mu\text{m}$ .

**workfun:** 定义电极的功函数, 单位为  $\text{V}$ . 如果材料的功函数没有定义则默认为欧姆接触.

#### DEFECT 语句:

不规则材料的带隙中有大量缺陷能态. 采用连续的能态密度, 可以精确地模拟由多晶或非晶物质组成的器件. DEFECT 语句定义缺陷态密度 (density of defect states, DOS) 由指数衰减的带尾态和高斯分布的带隙态(mid-gap states)两部分组成. 除此之外, 必须把晶粒间边界界面作为热离场发射边界来模拟.

**ega:** 类受主能态高斯分布峰对应的能量. 可通过测量导带的边缘获得.

**egd**:类施主能态高斯分布峰对应的能量,可通过测量价带的边缘获得.

**f.tftacc**:在用 C 语言解释器定义 DEFECT 时,用来表示包含 C 语言函数的文件名,描述受主态密度分布作为能量的函数.

**f.tftdon**:在用 C 语言解释器定义 DEFECT 时,用来表示包含 C 语言函数的文件名,描述施主态密度分布作为能量的函数.

**nta**:导带边缘尾分布 (tail distribution) 处类受主的能态密度.

**ntd**:价带边缘尾分布处类施主的能态密度.

**nga**:高斯分布中总的受主态密度.

**ngd**:高斯分布中总的施主态密度.

**siggae**:类受主的高斯分布中电子的俘获截面.

**siggah**:类受主的高斯分布中空穴的俘获截面.

**siggde**:类施主的高斯分布中电子的俘获截面.

**siggdh**:类施主的高斯分布中空穴的俘获截面.

**sigtae**:类受主的尾分布中电子的俘获截面.

**sigtah**:类受主的尾分布中空穴的俘获截面.

**sigtde**:类施主的尾分布中电子的俘获截面.

**sigtdh**:类施主的尾分布中空穴的俘获截面.

**wga**:类受主能态的高斯分布所对应的特征衰变能 (characteristic decay energy) .

**wgd**:类施主能态的高斯分布所对应的特征衰变能.

**wta**:类受主能态的尾分布所对应的特征衰变能.

**wtd**:类施主能态的尾分布所对应的特征衰变能.

## DOPING 语句:

DOPING 语句用来定义掺杂分布.

**char**:注入的主要特征长度.如果该参数没定义,该值可以从极性参数、边界参数 (Polarity Parameters、Boundary Parameters) 求出.

**dose**:高斯分布中总的剂量.

**junct**:高斯分布中结位置 (即结深).当 JUNCT 定义了,特征长度可通过测量

**peak**:高斯分布峰值浓度的位置.

## EXTRACT 语句:

EXTRACT 语句可用来提取器件参数.它可以灵活使用,既可引用软件自带的计算公式,又可根据自己需要构建一个新的公式.EXTRACT 语句是基于已有的曲线或结构文件.默认情况下,EXTRACT 使用当前打开的 log 文件.如果想打开的是以前的 log 文件,使用下面的语句即可:

EXTRACT INIT INF="<filename>"

## INTERFACE 语句:

INTERFACE 语句用来定义半/绝界面电荷密度和表面复合速度.

**ar.absorb**:定义增透膜的吸收系数.默认值为  $0.0\text{cm}^{-1}$ .

**ar.thick**:定义一层增透膜的厚度.

**material**:定义增透膜材料.

**optical**:定义对象界面的光学性质将用矩阵转换方法(Transfer Matrix Method)模拟.具体见

ATLAS 手册第 10 章的 2.4 节"Matrix Method".

**qf**:定义半/绝界面处固定的氧化层电荷密度 ( $\text{cm}^{-2}$ ).

**s.n:**定义电子表面复合速度.

**s.p:**定义空穴表面复合速度.

**p1.x,p1.y,p2.x** 和 **p2.y** 定义一个束缚盒子(bounding box).在这个盒子内必须是代表增透膜的界面.**p1.x,p2.x,p1.y** 和 **p2.y** 分别为这个盒子的 x 方向左、右值和 y 方向上的下、上值.

## MATERIAL 语句:

MATERIAL 语句用来定义材料的物理参数.

**eg300:**300K 时的带隙.

**mun:**低电场下电子迁移率(该参数仅用于非浓度依赖迁移率模型).

**mup:**低电场下空穴迁移率(同上).

**nc300:**300K 时导带底有效态密度.

**nv300:**300K 时价带顶有效态密度.

**taumob.el:**依赖温度的迁移率模型中电子的弛豫时间.默认值为  $0.4 \times 10^{-12}$ S.

**taun0:**电子的 Shockley-Read-Hall(SRH)寿命.

**taup0:**空穴的 Shockley-Read-Hall(SRH)寿命.

**taurel.el:**能量平衡模型中电子的弛豫时间.默认值为  $0.4 \times 10^{-12}$ S.

**tmun:**依赖温度的电子格点迁移率的格点温度系数.

**tmup:**依赖温度的空穴格点迁移率的格点温度系数.单位为 cm/s.

**vsaturation:**定义依赖电场的迁移率模型中的饱和速度.

## MESH 语句:

**nx:**X 方向上的节点数.(ny/nz 含义类似)

## METHOD 语句: 用来设置解方程和与之相关参数的数值方法.

METHOD 语句用来设置随之而来解的数值方法.所有结构和模型定义都应该在 METHOD 语句之前,而所有的偏置都应该置于其后.

**block:** 定义在随后的 SOLVE 语句中,将把 BLOCK NEWTON 方法当作一个可能的解答方法(直到另有定义).该方法仅当在模拟中考虑格点加热或能量平衡时才有意义.对于等热漂移-扩散模拟,BLOCK 等价于 NEWTON.

**climit:**定义为浓度归一化因子.具体参见 18-8 这一页.

**gummel:**定义在随后的 SOLVE 语句中,将把 GUMMEL 方法当作解答方法(直到另有定义).如果同一个 METHOD 语句中定义了其他方法(BLOCK 或 NEWTON),各个解方法将依次使用,直到收敛.其顺序为先 GUMMEL 然后 BLOCK 再使用 NEWTON.

**gummel.newton:**定义使用 GUMMEL 解方程后还将用 NEWTON 方法,即使在使用 GUMMEL 的时候所有方程已经收敛.

**itlimit:**允许外部循环(Newton 迭代法或 Gummel continuity iterations) 的最大值,默认值为 25.

**newton:**定义在随后的 SOLVE 语句中,将把 NEWTON 方法当作解答方法(直到另有定义).

如果未定义解答方法,默认使用 NEWTON.

**quasi:**定义瞬态模拟的准静态近似.这对于每个时间步长内处于平衡态的器件的长时间尺度的瞬态模拟非常有效.

**trap:**定义如果解过程分散,电极偏置将在初始估计值基础上按倍增因子 ATRAP(默认值为 0.5)降低.

**maxtrap:**定义如果解过程分散,trap 过程的最大次数.该值的范围为 1 到 10,stack 是它的别名.



## MODEL 语句:

物理模型的定义是采用 **MODELS** 和 **IMPACT** 语句.这些模型参数语句包括:**MODELS**, **IMPACT**, **MOBILIT** 和 **MATERIAL**.物理模型可归为五类:迁移率、复合、载流子统计、碰撞电离和隧穿.

### 1. 载流子统计模型:

**bgn**:带隙变窄(Bandgap Narrowing)模型,对于重掺区域尤为重要.

**boltzmann**:玻尔兹曼统计模型(默认模型).

**fermi**:费米-迪拉克统计模型,用于重掺的简并半导体.

### 2. 迁移率模型:

**analytic**:依赖浓度和温度的迁移率模型.

**conmob**:依赖浓度的迁移率模型.

**cvt**:考虑  $N$ 、 $T$ 、 $E_{\parallel}$ 、 $E_{\perp}$  影响的完全模型.

**fldmob**:依赖平行场的迁移率模型.

### 3. 复合模型:

**auger**:即俄歇复合,表征三个载流子的直接跃迁.

**consrh**:寿命依赖于浓度的 **SRH** 复合模型.推荐使用于 **Si**.

**optr**:带间复合模型.仅用于直接带隙材料.

**s.n** 及 **s.p**:考虑半导体-绝缘体的界面复合.它们在 **INTERFACE** 语句中设置.

**srh**:Shockley-Read-Hall 模型.在模型的前提假设为少数载流子的寿命为一定值.该模型适用于大多数模拟.

### 4. 碰撞电离模型:

**impact selb**:Selberherr 碰撞电离模型.

### 5. 隧穿模型

**bbt.kl**:Klaassen 带间隧穿模型——在一个足够大的电场下,器件的局部能带发生弯曲,以至于价带电子隧穿至导带.其结果是,导带上有额外的电子,价带上有等量的空穴.该产生机制可以用一连续方程来表示:隧穿几率  $G_{\text{BBT}} = \text{BB.A} \times E^{\text{BB.GAMMA}} \exp(-\text{BB.B}/E)$ .式中,  $E$  为电场包括直接跃迁和间接跃迁. (**bb.a**, **bb.b** 和 **bb.gamma** 均为带间隧穿参数.详见 atlas\_users 之 P163.)

在 **ATLAS** 中, **MODEL** 语句中的 **bipolar** 参数是双极器件合理的默认的物理模型集.其中包括:依赖浓度的迁移率模型(**CONMOB**)、依赖场的迁移率模型(**FLDMOB**)、禁带变窄模型(**BGN**)、寿命依赖于浓度的 **SRH** 模型(**CONSRH**)和俄歇复合模型(**AUGER**).

其他参数:

**evsatmod**:定义依赖平行场的迁移率模型应该用于电子.

**print**:是指将模型参数打印出来.建议在所有 **ATLAS** 运行中使用该参数.

**ir.tol**:定义绝对电流收敛准则(absolute current convergence criteria).

**ix.tol**:定义相对电流收敛准则(relative current convergence criteria).

## PROBE 语句:

**PROBE** 语句用来输出 **log** 文件中一些分布量(distributed quantities)的最值或者某些特定位置的值.

**dir**:定义特定向量偏离  $X$  轴的角度,默认值为零度.特定向量包括 **field**, **n.mob**, **p.mob** 以及 **polarization**.

**max**、**min**:定义 **PROBE** 将找到网格上的最大、小值.

**name**:定义某种描述并在 **TONYPLOT** 上显示的字符串.

**n.mob:**定义将探测到电子迁移率.如果使用了参数 **n.mob**,还应定义 **dir** 参数.

## SOLVE 语句:

ATLAS 中的 SOLVE 语句用来求解一个或多个偏压下的电压/电流.每个 SOLVE 语句必须定义一个初始偏置.只有解决了直流情形,才可能进行瞬态和交流分析. Solve 这个语句是整个器件模拟中真正进行数值计算的语句,它将会调用前面所用到的 **material**, **interface**, **contact**, **model** 等等所设置的模型和参数,来联立求解泊松方程、连续性方程、传输方程等方程式.

**ac:**交流参数(一个或多个).交流参数用来定义交流分析数据.

**cname:**定义 **compliance** 参数中的使用到的电极的名称.

**compliance:**对由 **cname** 或 **e.compliance** 参数定义的电极的电流设置一个限定值.当达到了 **compliance** 值,任何变化偏置停止,程序将从下一行继续运行.一般情况下,**compliance** 参数的单位为 A;当定义了 **grad** 参数,**compliance** 的单位则定义为 A/V.

**dc:**一个或多个直流偏置参数.

**e.compliance:**定义 **compliance** 参数中所用的电极数.

**ep:**初值猜测(或估计)参数(一个或多个).该参数定义初值怎样获得.

**fp:**一个或多个文件参数.

**grad:**定义 **compliance** 值为一个电流/电压梯度值,而不是一个电流值.

**index.check:**定义传输光线折射指数的实部和虚部将显示于运行输出.该参数可以用来确认输入的用户自定义的折射指数或用来核对默认参数.

**initial:**设置所有偏置为零.对于一个既定的结构,如果第一个偏置点没有定义该参数,一个 SOLVE INIT 语句将自动添加其中.其目的是为后面解答提供一个好的初始估计值.

**ion:**电离积分参数的集合.

**local:**定义初始估计值应该使用准费米能级的定阈值( **local values**).详见手册章节 18.6.

**photo:**光生参数(一个或多个).该参数用来定义光照数据.

**previous:**把前面的解作为初始估计值.

**ramptime:**应用于任意偏压变化(线性),变化起始于  $t=t_0$  时刻结束于  $t=t_0+\text{ramptime}$ .

**relative:**定义 **tstop** 值是相对于当前时间值.

**tstep** 或 **dt:**定义使用的时间步长.对于自动的时间步长,dt 仅用来选择第一个时间步长.

**tstop:**定义时间间隔的结束.如果模拟开始于  $t=t_0$ ,并定义了 **relative**,模拟结束于  $t=t_{\text{stop}}+t_0$ .

**tp:**瞬态参数(一个或多个).这些参数用来定义瞬态分析时的数据.

## TRAP 语句:

TRAP 语句用来设置半导体带隙内处于分散能级水平的体陷阱及其参数值.

**acceptor:**定义一个受主型陷阱能级.

**degen.fac:**定义陷阱能级的简并度.

**density:**定义陷阱能级的最大态密度,其单位为  $\text{cm}^{-3}$ .

**e.level:**定义分散陷阱能级的能量,其单位为 eV. 对于受主,e.level 是相对导带边缘;对于施主,则依赖于价带边缘.

**fast:**定义陷阱密度是不变的,即使是进行瞬态模拟(即即刻到达平衡态).

**sign、sigp:**定义电子、空穴陷阱的俘获横截面积,其单位为  $\text{cm}^2$ .

## OUTPUT 语句:

OUTPUT 语句用来定义将以标准结构形式文件储存的数据.

**charge:**定义净电荷将包含于标准结构文件之中.

**e.field** 或 **efield**:定义总的电场将包含于标准结构文件之中.

**e.velcocity**:定义总的电子速度将包含于标准结构文件之中.

**flowlines**:定义电流线(current flowlines)将包含于标准结构文件之中.

**jx.electron**、**jx.hole**:定义电子、空穴电流密度 X 方向分量将包含于标准结构文件之中.

**jy.electron**、**jy.hole**:定义电子、空穴电流密度 Y 方向分量将包含于标准结构文件之中.

**qfn**、**qfp**:定义电子、空穴准费米能级将包含于标准结构文件之中.

**opt.intens**:定义光强将包含于标准结构文件之中.

## Appendix B: IC 制造工艺简述

### IC制造工艺主要工序

1. 图形转换: 将设计在掩模版(类似于照相底片)上的图形转移到半导体单晶片上;
2. 掺杂: 根据设计的需要,将各种杂质掺杂在需要的位置上,形成晶体管、接触等;
3. 制膜: 制作各种材料的薄膜.

### 光刻

光刻是加工集成电路微图形结构的关键工艺技术,通常,光刻次数越多,就意味着工艺越复杂.另一方面,光刻所能加工的线条越细,意味着工艺线水平越高.光刻工艺是完成在整个硅片上进行开窗的工作.

#### 光刻工艺步骤:

打底膜(HMDS粘附促进剂)→涂光刻胶→前烘→对版曝光→显影→坚膜→刻蚀:采用干法刻蚀(Dry Etching)→去胶:化学方法及干法去胶.

#### 光刻三要素:光刻胶、掩模版和光刻机.

(光刻胶受到特定波长光线的作用后,导致其化学结构发生变化,使光刻胶在某种特定溶液中的溶解特性改变.)

**正胶:**分辨率高,曝光后可溶,在VLSI工艺中通常使用正胶.

**负胶:**分辨率差,曝光后不可溶,适于加工线宽 $\geq 3\text{ }\mu\text{m}$ 的线条.

#### 几种常见的光刻方法:

**接触式光刻:**分辨率较高,但是容易造成掩模版和光刻胶膜的损伤.

**接近式曝光:**在硅片和掩模版之间有一个很小的间隙(10~25mm),可以大大减小掩模版的损伤,分辨率较低.

**投影式曝光:**利用透镜或反射镜将掩模版上的图形投影到衬底上的曝光方法,目前用的最多的曝光方式.

### 刻蚀

**湿法刻蚀:**利用液态化学试剂或溶液通过化学反应进行刻蚀的方法;

**干法刻蚀:**主要指利用低压放电产生的等离子体中的离子或游离基(处于激发态的分子、原子及各种原子基团等)与材料发生化学反应或通过轰击等物理作用而达到刻蚀的目的.

### 掺杂

**定义:**将需要的杂质掺入特定的半导体区域中,以达到改变半导体电学性质,形成PN结、电阻、欧姆接触.

#### 掺杂工艺分类:

**热扩散法掺杂:**利用原子在高温下的扩散运动,使杂质原子从浓度很高的杂质源向硅中扩散并形成一定的分布.

**离子注入法掺杂:**将具有很高能量的杂质离子射入半导体衬底中的掺杂技术,掺杂深度由注

入杂质离子的能量和质量决定,掺杂浓度由注入杂质离子的数目(剂量)决定.

#### 两大掺杂工艺优缺点:

**热扩散:**工艺相对简单,但掺杂浓度控制精确度差、位置准确度也差.

**离子注入:**离子注入技术以其掺杂浓度控制精确、位置准确等优点,正在取代热扩散掺杂技术,成为

VLSI工艺流程中掺杂的主要技术. 但需昂贵的设备和退火工艺.由于高能粒子的撞击,导致硅结构的晶格发生损伤.为恢复晶格损伤,在离子注入后要退火处理,根据注入的杂质数量不同,退火温度在450℃~950℃之间,掺杂浓度大则退火温度高,反之则低.在退火的同时,掺入的杂质同时向硅体内进行再分布,如果需要,还要进行后续的高温处理以获得所需的结深和分布.

## 退火

定义:也叫热处理,集成电路工艺中所有的在氮气等不活泼气氛中进行的热处理过程都可以称为退火作用 1.激活杂质:使不在晶格位置上的离子运动到晶格位置,以便具有电活性,产生自由载流子,起到杂质的作用; 2.消除损伤

退火方式

- 1.炉退火
- 2.快速退火:脉冲激光法、扫描电子束、连续波激光、非相干宽带频光源(如卤光灯、电弧灯、石墨加热器、红外设备等)

## 氧化

目的:制备  $\text{SiO}_2$  层. $\text{SiO}_2$  是一种十分理想的电绝缘材料,它的化学性质非常稳定,室温下它只与氢氟酸发生化学反应.

作用:在 MOS 电路中作为 MOS 器件的绝缘栅介质,器件的组成部分;  
扩散时的掩蔽层,离子注入的(有时与光刻胶、 $\text{Si}_3\text{N}_4$  层一起使用)阻挡层;  
作为集成电路的隔离介质材料;  
作为电容器的绝缘介质材料;  
作为多层金属互连层之间的介质材料;  
作为对器件和电路进行钝化的钝化层材料.

制备方法

- 1.热氧化法  
干氧化、水蒸汽氧化、湿氧化、干氧—湿氧—干氧(简称干湿干)氧化法、氢氧合成氧化
- 2.化学气相沉积法
- 3.热分解沉积法
- 4.溅射法

干氧:干法氧化是在纯氧中氧化,氧化速率慢,薄膜致密,固定电荷密度少.  
(标准干氧化:900℃—1050℃,数分钟至数小时.)

湿氧:湿法氧化是在氧和水的混合物中氧化氧化速率比干氧工艺大大提高,但氧化层的致密性不如干氧氧化的薄膜.

(湿氧化(水蒸汽,可掺 HCl):900℃—1050℃,数分钟至数小时.)

硅局部氧化 (LOCOS 技术)

通常的氧化是在圆片表面整个氧化成氧化物平面,然后进行刻蚀.刻蚀出来的窗口存在台阶,影响后续工艺.LOCOS 是在不需要氧化的地方涂上一层氮化硅,氮化硅层下面的硅层不被氧化.LOCOS 工艺在大约 1000℃ 高温下进行.

## 接触与互连

Al 是目前集成电路工艺中最常用的金属互连材料,但 Al 连线也存在一些比较严重的问题,如电迁移严重、电阻率偏高、浅结穿透等.Cu 连线工艺有望从根本上解决该问题.IBM、Motorola 等已经开发成功.目前,互连线已经占到芯片总面积的 70~80%;且连线的宽度越来越窄,电流密度迅速增

加.