

半导体仿真工具 Silvaco TCAD 学习资料

前言

接触Silvaco TCAD仿真软件已经有很长时间了，这期间还熟悉了一下ISE TCAD。学习的过程中，老师、师兄姐和同学给了我极大的帮助。这本书的主旨就是希望提供一些对于Silvaco仿真的可资借鉴的经验。学习和使用Silvaco的时候各人的视角会不一样，虽然这本书只是介绍一些常用的仿真，例句也有限，但因为流程控制和语法都有相通的地方，完全可以参照本书和手册的说明灵活地建立仿真以及仿真一些本书未提及的特性。

写本书的另一个动力是想体验一下 \LaTeX 在排版书籍上和word的不同。经过这次编排觉得各有优缺点。 \LaTeX 对排图片和表格这些浮动元素很费力。我在排此书的时候经常遇到图片和表格按照自动方式排的位置很差，和文字内容脱节很远，为此不得不处理图片，有时还将表格拆开等。如果各位遇到很脱节的地方还请见谅。 \LaTeX 的使用需要有很强的耐心和探索精神，也可以说是牺牲精神。不管是MS Office还是 \LaTeX ，总的目的是想内容通顺，阅读起来舒服，用什么工具就是“八仙过海，各显神通”了。

真诚地希望本书在大家探索仿真的旅途上能起到一定的指引作用。

感谢石老师给我这个机会，不然我也没有勇气下决心写这样一个心得。

本人所学有限，错误和不足之处在所难免。如蒙批评指正，我将感激不尽。

谨以此书献给我辛勤劳作的父母，正是他们不言的教诲使我一步步成长到今天，祝他们永远健康幸福和快乐。

化龙居士¹
2009年5月 于 望江

¹shaohuachn@126.com

目 录

第一章 仿真准备	1
1.1 认识Silvaco TCAD	1
1.1.1 数值计算	1
1.1.2 基于物理的计算	1
1.2 Silvaco TCAD	2
1.2.1 主要组件	3
1.2.2 目录结构	6
1.2.3 文件类型	7
1.3 DeckBuild命令	7
1.3.1 命令格式	8
1.3.2 Go	8
1.3.3 Set	9
1.3.4 Tonyplot	10
1.3.5 Extract	12
1.4 学习方法	18
第二章 二维工艺仿真	21
2.1 ATHENA 概述	21
2.2 仿真流程	25
2.2.1 定义网格	25
2.2.2 衬底初始化	27
2.2.3 工艺步骤	29
2.2.4 抽取特性	29
2.2.5 结构操作	29
2.2.6 Tonyplot显示	29
2.3 单项工艺	30
2.3.1 离子注入	30
2.3.2 扩散	33
2.3.3 淀积	35
2.3.4 刻蚀	37

2.3.5	外延	43
2.3.6	抛光	44
2.3.7	光刻	45
1.	MASK	46
2.	ILLUMINATION	47
3.	PROJECTION	47
4.	FILTER	48
5.	LAYOUT	49
6.	IMAGE	50
7.	EXPOSE	52
8.	BAKE	53
9.	DEVELOP	54
10.	完整的光刻流程	55
2.4	集成工艺	57
2.5	优化	60
2.5.1	优化设置	60
2.5.2	待优化参数	62
2.5.3	优化目标	62
2.5.4	优化结果	62
第三章	二维器件仿真	65
3.1	ATLAS概述	65
3.2	仿真流程	68
3.3	定义结构	68
3.3.1	命令语句生成结构	69
3.3.2	DevEdit编辑得到结构	72
3.4	材料参数及模型	74
3.4.1	接触特性	74
3.4.2	材料特性	75
3.4.3	界面特性	79
3.4.4	物理模型	79

3.5	数值计算方法	80
3.6	获取器件特性	82
3.6.1	直流特性	82
3.6.2	交流小信号特性	84
3.6.3	瞬态特性	85
3.6.4	高级特性	85
3.7	仿真结果分析	88
3.7.1	实时输出	88
3.7.2	日志文件	91
3.7.3	DeckBuild抽取	92
3.7.4	Tonyplot显示	92
第四章	高级的特性	93
4.1	三维仿真	93
4.1.1	三维工艺仿真	93
4.1.2	三维器件仿真	93
4.1.3	Tonyplot3D	96
4.2	C解释器	97
附录A	材料系统	99
A.1	材料类型	99
A.2	自定义材料	100
附录B	物理	103
B.1	基本半导体方程	103
B.2	基本的载流子统计理论	105
B.3	边界物理	108
B.4	物理模型	108

插图目录

1-1	Silvaco产品分布图	2
1-2	Silvaco仿真流程图	3
1-3	ATHENA输入输出	4
1-4	ATLAS输入输出	5
1-5	DeckBuild界面	7
1-6	Tonyplot显示结构文件	10
1-7	Tonyplot显示器件仿真的数据文件	10
1-8	Tonyplot Display(2D Mesh)	11
1-9	Tonyplot Coutline	11
1-10	Tonyplot显示的二维器件结构及Coutline得到的一维能带图	12
2-1	硼注入及退火后的结构	24
2-2	均匀网格	25
2-3	非均匀网格	25
2-4	淀积多晶硅层的网格分布	26
2-5	initlize的参数选择窗口	27
2-6	implant的参数选择窗口	30
2-7	不同离子注入角度(TILT)的注入结果	32
2-8	diffuse的参数选择窗口	33
2-9	gasflow的参数选择窗口	33
2-10	deposit的参数选择窗口	36
2-11	rate.deposit的参数选择窗口	36
2-12	etch的参数选择窗口	37
2-13	rate.etch的参数选择窗口	39
2-14	各种刻蚀效果的比较	42
2-15	polish的参数选择窗口	44
2-16	rate.polish的参数选择	45
2-17	Maskviews的界面	46
2-18	illumination的参数选择窗口	47
2-19	projection的参数选择窗口	48

2-20	filter的参数选择窗口	48
2-21	layout的参数选择窗口	49
2-22	image的参数选择窗口	51
2-23	expose的参数选择窗口	52
2-24	bake的参数选择窗口	53
2-25	develop的参数选择窗口	54
2-26	完整光刻流程的仿真结果	56
2-27	优化方案的设置	61
2-28	待优化的参数设置	62
2-29	优化的目标设置	62
2-30	优化得到结果	63
3-1	ATLAS命令组及其主要的状态	68
3-2	eliminate使纵向的网格线变稀疏的效果	69
3-3	ATLAS命令生成的器件结构	72
3-4	DevEdit生成的器件结构	72
3-5	材料能带结构参数示意图	78
3-6	log文件导出成csv格式以查看仿真数据	92
4-1	DevEdit3D界面	94
4-2	DevEdit3D导入的原二维结构以及编辑得到三维结构的二维剖面图	95
4-3	Tonyplot显示器件结构	96
4-4	Tonyplot3D的Cutplane及其预览	96
A-1	自定义材料GaMnAs	101

表格目录

2-1	Athena Features and Capabilities	21
2-2	initialize的主要参数及其说明	28
2-3	implant的主要参数及其说明	31
2-4	diffuse的主要参数及其说明	34
2-5	deposit的主要参数及其说明	35
2-6	etch的主要参数及其说明	38
2-7	RIE etch和wet etch的主要参数及其说明	38
2-8	plasma etch的主要参数及其说明	39
2-9	Monte Carlo plasma etch的主要参数及其说明	40
2-10	epitaxy的主要参数及其说明	43
2-11	polish的主要参数及其说明	44
2-12	rate.polish的主要参数及其说明	45
2-13	mask的主要参数及其说明	46
2-14	illumination的主要参数及其说明	47
2-15	pupil.filter的主要参数及其说明	49
2-16	layout的主要参数及其说明	50
2-17	image的主要参数及其说明	51
2-18	expose的主要参数及其说明	52
2-19	bake的主要参数及其说明	53
2-20	develop的主要参数及其说明	54
2-21	rate.develop的主要参数及其说明	55
3-1	材料特性的常用参数及其说明	76
A-1	ATLAS材料	99
B-1	状态密度的参数	106
B-2	常规能带参数	107
B-3	禁带变窄模型及其参数	107
B-4	低场迁移率模型参数	108
B-5	SRH复合模型参数	109
B-6	俄歇复合模型参数	109

第一章 仿真准备

这一章主要介绍TCAD的基本概念，DeckBuild的主要组件以及DeckBuild的命令(extract, go,set和tonyplot)，Silvaco文档的分布和学习方法等。

1.1 认识Silvaco TCAD

TCAD就是Technology Computer Aided Design，指半导体工艺模拟以及器件模拟工具，世界上商用的TCAD工具有Silvaco公司的Athena和Atlas，Synopsys公司的TSuprem和Medici以及ISE公司（已经被Synopsys公司收购）的Dios和Dessis。Synopsys 公司最新发布的TCAD工具命名为Sentaurus。

1.1.1 数值计算

计算机仿真必须基于数值计算。Silvaco中的数值计算是基于一系列的物理模型及其方程的，而方程中的某些量又用其他的方程来描述。然而实际的物理系统非常复杂，信息量也非常巨大，必须将其离散化，那么自然地半导体仿真应用上就基于网格计算。

网格计算：将半导体仿真区域划分成网格，在网格点处计算出希望得到的特性（如：电学性质，光学性质，工艺步骤的速率等等）。

数值计算必须考虑的问题：

- 精确性（和网格密度、计算的步长疏密、算法和物理模型的选择等有关）
- 计算速度（和网格密度、计算步长的疏密以及算法等有关）
- 收敛性（和计算步长的疏密、初始值以及算法有关）

仿真计算时参数设置上需要在精确性、计算速度和收敛性之间取得折衷。网格点的密度和计算的精确度有关，但Silvaco中的网格点总数¹是有限制的。请注意，网格点太多导致信息量太大，在仿真时会报错。

1.1.2 基于物理的计算

仿真的精度与正确性和选择的物理模型相关。基于物理的计算是指在仿真计算时采用的方程是有物理意义的，在不同的应用场合使用不同的物理模型。通常Silvaco的仿真思路和采用的模型是基于成熟的成果，这些成果通常是发表在IEEE上。Silvaco采用这些成果并打造成Silvaco library。

物理模型的详细说明请参见ATLAS手册中物理部分或附录B。

¹如果Silvaco能将网格总数上限加大，则精确性必然提高很多，虽然将付出更多的时间和机器耗费，但我想这是众多模拟者所乐见的。

1.2 Silvaco TCAD

Silvaco名称是由三部分组成的，即“Sil”，“va”和“co”，从字面意思上不难理解到是“硅”，“谷”和“公司”英文单词的前几个字母的组合。Silvaco的中文名称叫矽谷科技公司。

来自美国的矽谷科技公司(Silvaco)经过20多年来的成长与发展，现已成为众多领域卓有建树的EDA公司，包括TCAD工艺和器件模拟、Spice参数提取、高速精确电路仿真、全定制IC设计与验证等等。Silvaco拥有包括芯片厂、晶圆厂、IC设计企业、IC材料业者、ASIC业者、大学和研究中心等在内的庞大的国内外客户群。如今，Silvaco已在全球设立了12间分公司以提供更好的客户服务和合作机会。

Silvaco 是现今市场上唯一能够提供给Foundry 最完整的解决方案和IC 软件厂商。提供TCAD, Modelling 以及EDA 前端和后端的支持，也能提供完整的Analog Design Flow 给IC 设计业者。产品SmartSpice 是当今公认的模拟软件的黄金标准，因为支持多集成CPU 的SmartSpice 的仿真速度比起同类型软件更好，它是国外模拟设计师的最爱；SmartSpice的收敛性也被公认为仿真器最好的。Silvaco还有其他整套流程包括版图工具以及验证工具。许多世界知名Foundry包括台积电、联电、Jazz和X-FAB都跟Silvaco有PDK的合作。

Silvaco公司在2006年正式进入中国市场，希望以在国外超过20年的经验提供给国内Foundry最佳的解决方案。Silvaco公司是现今市场上唯一能够提供整套包括建模，TCAD，模拟软件以及PDK方案的EDA公司。

从图1-1中也可看出Silvaco的产品覆盖了半导体产业相当广泛的领域。这也是本书将标题定为半导体仿真工具的原因。




图 1-1: Silvaco产品分布图

1.2.1 主要组件

下面介绍一下Silvaco TCAD的一些组件的特性，这些组件包括交互式工具DeckBuild和Tonyplot，工艺仿真工具ATHENA，器件仿真工具ATLAS和器件编辑器DevEdit，还有一些内部的模块。

1、DeckBuild

各TCAD仿真组建均可在DeckBuild界面调用，例如先由ATHENA或DevEdit生成器件结构，再由ATLAS对器件特性进行仿真，最后由Tonyplot2D或Tonyplot3D显示输出。

DeckBuild特性功能如下：

- 输入和编辑仿真文件
- 查看仿真输出并对其进行控制
- 提供仿真器组件间的自动转换
- 提供工艺优化以快速而准确地获得仿真参数
- 内建的抽取功能对仿真得到的特性进行抽取
- 内建的显示提供对结构的图像输出
- 可从器件仿真的结果中提取对应SPICE模型的参数

各仿真器都是通过集成环境DeckBuild组织的。Silvaco仿真流程如图1-2中所示。




图 1-2: Silvaco仿真流程图

2、TonyPlot可视化工具

Tonyplot可视化工具对结构进行显示，可显示的结构包括一维、二维和三维结构（三维结构显示需要使用Tonyplot3d），可显示的信息包括几何结构，材料特性，器件仿真得到的电学、光学和热学信息等。如图1-10为Tonyplot显示的二维器件结构及Cutline得到的一维能带图，图1-7为Tonyplot显示的器件仿真的特性曲线。

Tonyplot可以将显示结果导出图片，也可将结构中的信息导出，这样就能清楚地获取仿真的数据结果，以便进行处理。

Tonyplot还提供动画制作等功能，可以将各步工艺的图像结果制作成动画以观察各工艺的效果。

1.3.4节将详细介绍Tonyplot2D，Tonyplot3D将在4.1.3节进行介绍。

3、ATHENA

工艺模拟软件ATHENA能帮助工艺开发和优化半导体制造工艺。ATHENA提供一个易于使用，模块化的，可扩展的平台。ATHENA能对所有关键制造步骤（离子注入，扩散，刻蚀，淀积，光刻以及氧化等）进行快速精确的模拟。仿真能得到包括CMOS，Bipolar，SiGe，SOI，III-V，光电子以及功率器件等器件结构，并精确预测器件结构中的几何参数，掺杂剂量分布和应力。优化设计参数使速度、产量、击穿、泄漏电流和可靠性达到最佳结合。它通过模拟取代了耗费成本的硅片实验，可缩短开发周期和提高成品率。

ATHENA工艺仿真软件的主要模块有：SSuprem4，二维硅工艺仿真器，蒙特卡洛注入仿真器，硅化物模块的功能，精英淀积和刻蚀仿真器，蒙特卡洛淀积和刻蚀仿真器，先进的闪存材料工艺仿真器，光电印刷仿真器。相应的工艺仿真步骤将在第2章详细介绍。

图1-3为ATHENA的输入输出框架。




图 1-3: ATHENA输入输出

4、ATLAS

ATLAS器件仿真系统可以模拟半导体器件的电学、光学和热学行为。ATLAS提供一个基于物理的，使用简便的模块化的可扩展的平台，用以分析所有二维和三维模式下半导体器件的直流、交流和时域响应。

ATLAS可以仿真硅化物、III-V、II-VI、IV-IV或聚合/有机物等各种材料。可以仿真的器件类型很多，如CMOS、双极、高压功率器件、VCSEL、TFT、光电子、激光、LED、CCD、传感器、熔丝、铁电材料、NVM、SOI、HEMT、Fin和HBT等。

ATLAS器件仿真软件的主要模块有S-Pisces(二维硅器件模拟器)，Device3D(三维硅器件模拟器)，Blaze2D/3D(高级材料的二维/三维器件模拟器)，TFT2D/3D(无定型和多晶体二维/三维模拟器)，VCSELS模拟器，Laser(半导体激光/二极管模拟器)，Luminous2D/3D(光电子器件模块)，Ferro(铁电场相关的介电常数模拟器)，Quantum(二维/三维量子限制效应模拟模块)，Giga2D/3D(二维/三维非等温器件模拟模块)，NOISE(半导体噪声模拟模块)，ATLAS C解释器模块和MixedMode(二维/三维组合器件和电路仿真模块)等。

5、DevEdit2D/3D结构和Mesh编辑器

器件编辑器DevEdit2D/3D可以编辑得到器件结构。器件编辑器有很多优点，如器件编辑器的结构区域是由一系列特定位置的点构成的，所以结构的边界可以很灵活，而且还可以从工艺仿真得到的结构基础上进行编辑。

器件编辑器的使用将在器件特性仿真的结构定义部分进行介绍。

图1-4为ATLAS的输入输出框架。




图 1-4： ATLAS输入输出

6、掩膜输出编辑器

MaskViews Layout Editor可以编辑掩膜结构，以便光刻等后续工艺中采用。MaskViews有图形化界面。当然掩膜结构也可以用命令生成。

掩膜编辑器将在工艺仿真的光刻部分进行介绍。

在仿真之前须先熟悉Silvaco的架构。之前介绍的只是TCAD 的一些概念以及Silvaco各仿真模块的特性。下面将介绍软件使用的信息，如文件分布、文件类型等。

稍后介绍DeckBuild集成环境的命令，以便在后续的工艺仿真和器件仿真中集中关注相应的仿真语法。

1.2.2 目录结构

以windows下的2008-09版本为例，目录结构如下：

X:\sedatools

- |—Doc （程序安装以及sflm² 说明文档）
- |—exe （可执行程序的快捷方式）
- |—Shortcuts （程序控制及主要仿真环境的快捷方式）
- |—lib （组件库）
 - |—ATHENA （工艺仿真器）
 - |—5.16.3.R （版本号）
 - |—common （包含模型文件、模板、材料参数和默认仿真参数等）
 - |—docs （用户手册及组件更新的说明文档文件夹）
 - |—notes （各版本的新特性说明文档）
 - |—x86-nt （Windows平台下的可执行程序）
 - |—Atlas
 - |—deckbuild
 - |—Tonyplot
 - |—ssuprem3
 - |—devedit
 - ...
- |—example （示例库）
 - |—athena （工艺仿真的子示例库）
 - |—atlas
 - |—Mocasim
 - |—ssuprem3
 - |—tonyplot
 - ...
- ...

²Standard Floating License Manager, license管理器。

1.2.3 文件类型

- 输入文件“*.in”（DeckBuild界面的仿真输入文件）
- 结构文件“*.str”（工艺仿真或器件编辑器得到的器件结构）
- 器件仿真结果文件“*.log”（器件仿真时存储仿真结果）
- 设置文件“*.set”（Tonyplot的显示设置）
- 掩膜结构文件“*.lay”（光刻的掩膜信息）
- 抽取的结果文件“*.dat”（抽取得到的数据）
- 其他文件类型:“*.sepc”，“*.opt”，“*.lib”（C-编译器编写的函数文件）...

1.3 DeckBuild命令

DeckBuild是一个交互式、图形化的实时运行环境，在工艺和器件仿真中作为仿真平台。DeckBuild有仿真输入和编辑的窗口，也有仿真输出和控制的窗口。DeckBuild有很强的灵活的工具，也提供很多自动的特性，如仿真器的切换、内建的抽取（extract）等。




图 1-5: DeckBuild界面

图1-5为DeckBuild界面。顶部是工具栏、编辑菜单和命令条按钮。工具栏菜单有File、Command、Execution、Examples等。Command子菜单提供简单的抽取语句、显示当前结构和工艺优化的功能。工具栏下是新建、打开、保存、剪切、复制、粘贴和撤销等常用编辑操作。第三栏是命令条按钮。命令条按钮及意义如下：Run—从上致下执行或执行到“断点”（即Stop at line的“line”）时结束；Stop—执行完当前行就停止，Next—运行下一行就停止，Continue—在之前停止运行的行开始往下运行，Initialize—从仿真历史初始化，Kill—立即结束，Line—重置当前行为选择的行，Stop at line—设置断点。

上部窗口是命令输入和编辑窗口。下部窗口是实时输出窗口，输出当前的仿真状态。底部是状态栏。工作路径可由Edit>Preferences>Working Directory进行查看或更改。

DeckBuild具有很多方便的特性：

- 可由内建的显示工具显示当前结构
- 对仿真器进行全面的控制
- 允许备份或调用历史记录
- 显示当前执行的行以及在输出窗口显示实时的仿真状态

三个仿真器：SSuprem3，ATHENA和ATLAS在DeckBuild库中可直接使用。

DeckBuild内建的抽取命令可以对工艺和器件的仿真结果进行抽取，抽取以“函数计算器”的形式，由标准的描述（或经修改后）抽取得到的数据尽可能满足用户的需要。Extract内建的一维器件仿真器QUICKMOS和QUICKBIP可以方便快捷地对MOS和Bipolar器件的电学特性进行测试。Extract的结果可以以datafile(*.dat)的形式保存，这样就可以将数据取出并进行后续处理，如用Origin绘制数据图形等等。

1.3.1 命令格式

Silvaco的命令由command³和parameter两部分组成。通用格式如下：

command parameter1=<n> parameter2=<c>

其中“n”代表数值，如“30”；“c”代表字符串，如“silicon”。

① 命令可以简写，以不与其他简写相冲突为原则，如“deposit”可以用“depo”取代。不区分大小写。

② 命令和参数之间以空格分开。

③ 一行写不完的在该行的末尾加“\”（注意“\”前需留有空格），则下一行和该行将被视为同一个“command”。

④ “#”号后面是注释，仿真时不运行该行的内容。空行也不运行。

DeckBuild命令有go、set、tonyplot和extract，以下内容将对这四个命令的语法进行介绍。

1.3.2 Go

仿真之前需要启动相应的仿真器。command “go”的作用是仿真器启用或切换，仿真器可以是ATHENA，ATLAS，Ssuprem3等等。语法如下：

go <simulator> | simflags=<simulator>

simulator为仿真器名称，simflags指出仿真参数和程序版本。提供几个例句如下⁴：

```
go atlas
```

```
go atlas simflags="-V 5.0.8.R"
```

```
go devedit "-3d"
```

³工艺仿真和器件仿真时的statement同此command。

⁴以后的书写当中，可在DeckBuild中输入的命令将以一个密闭的黑边框框起来（除表示省略的点外，其他都可以用英文输入法在DeckBuild输入）。为了配合说明，有的行设置为灰底，有的语句进行了加粗。敬请留意。

在启动仿真器时如果注意DeckBuild的实时输出窗口就会发现，窗口内会显示当前仿真器的版本及可用的组件。如启动ATHENA工艺仿真器时，输出窗口里含有以下信息，表明这些仿真特性是license所允许的。还有提示导入的模型文件“athenamod”。

```
=====
ATHENA                      : Enabled
SSUPREM4                    : Enabled
Parallel Monte Carlo Implant : Enabled
ELITE                       : Enabled
Monte Carlo Deposit/Etch    : Enabled
OPTOLITH                    : Enabled
ATHENA multicore            : Enabled
=====
```

Loading model file 'C:\sedatools\lib\athena\5.16.3.R\common\athenamod'... done.

1.3.3 Set

全局变量进行设置或设置Tonyplot的显示方式（显示方式可以保存在*.set文件中，这样在显示时可直接调用）。

语法如下：

```
set <variable>=[ <value> | <expr> ][nominal]
```

<variable>为任意的变量设置，在后续语句中使用变量则需写为\$<variable>，即在前加一美元符号。“<EXPR>”表示整数，“|”号表示左右两边的参数选其一。

变量设置例句：

```
set temp=1000
set gaspress=1
diffuse time=30 temp=$temp press=$gaspress
```

此例中变量为“temp”和“gaspress”，值分别为“1000”和“1”。这样在后续仿真语句中声明“\$temp”和“\$gaspress”时，将会把设置的值自动代替这些变量。

这种全局变量设置的方法提供了很方便的特性，像变参数仿真中，可使得仿真文件更易更改。value可以是数字，也可是有某些变量经运算后得到的结果。

```
extract name="oxide thickness" thickness oxide
set etch_thickness= ($"oxide thickness" * 10000) + 0.05
etch oxide dry thickness=$etch_thickness
```

上面的例句中，第一句先抽取（extract）氧化层厚度，将其名称设为“oxide thickness”。第二句设置（set）变量“etch_thickness”值为之前抽取得到的氧化层厚度经一个运算（乘以10000再加上0.05）得到的结果。第三句干法刻蚀（etch）二氧化硅，刻蚀的厚度就是之前运算的值，即“\$etch_thickness”。如果语句读起来暂时觉得有困难，可以在以后熟悉了抽取和工艺仿真时再回到此处体会一下。

显示设置例句：

```
tonyplot structure.str -set show.set
```

“-set”表示Tonyplot显示将使用set设置，“show.set”表示采用的设置文件。Tonyplot设置将会在第1.3.4节Tonyplot中详细说明。

1.3.4 Tonyplot

Tonyplot是可视化工具。可将仿真时生成的临时文件（结构）、工艺仿真中保存的结构文件、器件编辑器生成的结构文件、器件仿真保存的log文件和抽取得到的dat文件显示出来。Tonyplot也是交互式工具，其内建的计算器可以得到很多信息。

语法：

```
tonyplot -args
```

Tonyplot有三种方式打开要显示的文件，“Add Plot”，“Overlay Plot”和“Replace Plot”，默认的是“Add Plot”。可以显示的文件类型有*.str，*.log，*.rsm，*.sta和*.dat等。

重叠显示

```
tonyplot -overlay cv.log iv.log -set show.set
```

上面重叠显示的例句表示将器件仿真得到的cv.log和iv.log按照设置文件show.set的设置，以重叠(Overlay)的方式显示在一个窗口中。图1-7就是将三个log文件叠放到一起进行显示的。

显示当前的结构

```
tonyplot
```




图 1-6: Tonyplot显示结构文件




图 1-7: Tonyplot显示器件仿真的数据文件

当Tonyplot窗口显示的是结构文件时，点右键会出现“display”选项，选择“display”后，弹出如图1-8的显示方式选择窗口。

下面从左致右介绍Tonyplot:Display(2D Mesh)的功能。




图 1-8: Tonyplot Display(2D Mesh)

显示选项	特性
Mesh	显示结构中的网格分布
Edges	显示结构的边界
Regions	显示材料区域
Contours	显示器件结构内部的物理量（有的需要在“output”中列出，如：con.band，val.band为导带和价带位置的信息）
Vectors	显示器件内部的矢量信息（如电场分布，电流分布...）
Light	显示光线在器件内部的特性（光强、波长和反射率...）
Junction	显示器件结构中结的边界
Electrodes	显示电极的名称
3D	显示三维信息
Lines	按照颜色来显示




图 1-9: Tonyplot Cutline

Tonyplot可显示二维结构文件，执行Tonyplot>Tool>Cutline，就会弹出一个Cutline的窗口，如图1-9所示。划线的方式可以是竖线、横线、斜线、折线或是由起始点和终止点连成的直线等多种方式。Tonyplot的Cutline可以得到一维的特性，Tonyplot3d的Cutplane可以得到某二维剖面的信息。

图1-10的右图为在原结构图中用Cutline划一条横线（位于沟道的高度）得到的一维特性。显示的量有导带，价带以及准费米能级的位置。从图中可看出电子和空穴的准费米能级重合，而且从准费米能级相对导带和价带的位置可以判断此结构为NMOS。




图 1-10: Tonyplot显示的二维器件结构及Cutline得到的一维能带图

由File>Export，将Format选为Comma Separated Values，再将路径及文件名设置好就可以将数据（ATLAS仿真得到的log文件）导出成csv格式，这样可以用Excel和Origin打开和编辑。结构文件可由Cutline显示一维特性，将其导出成“Tonyplot user data”后就可以查看一维信息了。也可由下一节将介绍的Extract抽取出一维特性。

1.3.5 Extract

简单语法：

```
extract extract - parameters
```

功能：抽取仿真中得到的信息

详细的语法：（语句的读法：“<QSTRING>”表示字符串，“<EXPR>”表示整数，竖线“|”表示左右的参数中选其一，中括号“[]”内的参数为可选参数。）

第1.3.1小节介绍过Silvaco的命令格式，即由命令command和一些参数parameter组成。有些参数只有一部分，只用一个简单的单词就可以描述，如name,location等。有些参数是由多部分组成的，一般是左边的描述右边。这里列出几个：c.boron，硼的浓度（浓度是硼

杂质的)；min.val，最小值；2d.max.conc，某二维范围内的最大浓度。在以后看语法的时候，敬请留意。

```
extract init infile=<QSTRING>
extract <value_type>|<curve_type> [name=<QSTRING>][outfile=<QSTRING>]
    [datafile=<QSTRING>][hide] [min.val=<EXPR>] [max.val=<EXPR>]
extract start<test_setup1>
    [extract cont <test_setup n>]
    extract done <value_multi_line> | <curve_multi_line> [name=<QSTRING>]
    [outfile=<QSTRING>] [datafile=<QSTRING>] [hide] [min.val=<EXPR>]
    [max.val=<EXPR>]
```

```
<value_type> = <value_sigle_line> | <value_multil_line>
<value_sigle_line> = thickness,xj,surf.conc,1dvt, 2d.max.conc | 2d.min.conc,
    max.conc | min.conc, x.pos | y.pos 2d.conc.file max.bound | min.bound(1D)
    max.bound | min.bound(2D) 2d.area...
<value_multi_line> = sheet.res | p.sheet.res | n.sheet.res conduct | p.conduct |
    n.conduct...
<test_setup> = [<material>][mat.occno=<EXPR>][region.occno=<EXPR>]
    [bias=<EXPR> | y.val=<EXPR> | region=<QSTRING>] [qss=<EXPR>]
    [interface.occno=<EXPR>]
```

抽取曲线定义：

```
curve_def=(<x_axis>,<y_axis> [x.min=<EXPR> x.max=<EXPR>])
```

```
<curve_def>=<curve_single_line> | <curve_multi_line>
```

各举一例如下：

single_line:

```
curve(depth,<impurity> [<material>][mat.occno=<EXPR>] [y.val=<EXPR>
    | x.val=<EXPR> | region=<QSTRING>])
```

multi_line:

```
curve(bias,n.conc | p.conc | n.qfl | p.qfl | intrinsic | potential | n.mobility | p.mobility
    | efield | econductivity [material="silicon" | "polysilicon"][region.occno=<EXPR>]
    [mat.occno=<EXPR>][y.val=<EXPR>|x.val=<EXPR>|region=<QSTRING>]
    [workfunc=<EXPR>][soi][semi.poly] [temp.val=<EXPR>]
<test_setup>=[<material>][mat.occno=<EXPR>] [region.occno=<EXPR>]
    [bias=<EXPR>][bias.step=<EXPR>] [bias.stop=<EXPR>][y.val=<EXPR>
    | x.val=<EXPR> | region=<QSTRING>]
    OR
    [interface.occno=<EXPR>] [qss=<EXPR>]
```

impurity可以是boron, phosphorous, arsenic, “net doping”等, 也可是<QSTRING>。Tonyplot中可以显示的量都可以作为<QSTRING>,如impurity=“electron conc”。

extract可以对轴或数据进行“+”, “-”, “*”, “/”, “^”等基本计算, 也支持类似C语言的描述如abs(), log(), dydt(), atan()的运算。

此外还支持数据的统计和拟合功能。min()返回最小值, ave()返回平均值, slope(LINE)返回线性拟合 $y = ax + b$ 的 a 值, yintercept(LINE)返回线性拟合 $y = ax + b$ 的 b 值, xintercept(LINE)返回线性拟合 $y = ax + b$ 的 $-b/a$ 值。

curve中的value:

v. “<electrode>”, 电极上的电压
i. “<electrode>”, 流过电极的电流
c. “<electrode1>” “<electrode2>”, 电极1和2间的电容
g. “<electrode1>” “<electrode2>”, 电极1和2间的电导
time, 时间
temperature, 温度
frequency, 频率
beam. “<beam no>”, 第no束光的光强
ie. “<electrode>”, 电子电流
rl. “<electrode>”, 集总电阻
cl. “<electrode>”, 集总电容
il. “<electrode>”, 集总电导
q. “<electrode>”, 电荷
id. “<electrode>”, 位移电流
rho. “<layer>”, 对应layer的总方块电阻
probe. “<probe name>”, ATLAS中probe的值
elect. “<parameter>”, 特定的电学参数...

抽取的默认参数:

material=“silicon”
impurity=“net doping”
x.val | y.val | region (x.val为距器件左边界5%的位置)
*.occno=1
datafile=“results.final”
1dvt type=ntype
2d.area (x.step为器件尺寸的10%)
temp.val=300
bias=0
1dvt (bias.stop=5, bias.step=0.25, vb=0)

```

1dcapacitance (bias.stop=5,bias.step=0.25,vb=0,vg=0,bias.ramp=vg)
soi=false
semi.ploy=false
incomplete=false

```

抽取结果的默认单位:

```

Material thickness (angstroms)
Junction depth (microns)
Impurity concentrations (impurity units, typically atoms/cm3)
Junction capacitance (Farads/cm2)
QUICMOS capacitance (Farads/cm2)
QUICKMOS 1D Vt (Volts)
Sheet resistance (Ohm/square)
Sheet conductance (square/Ohm)
Electrode voltage (Volts)
Electrode internal voltage (Volts)
Electrode current (Amps)
Capacitance (Farads/micron) Conductance (1/Ohms)
Transient time (Seconds)
Frequency (Hertz)
Temperature (Kelvin)
Luminescent power (Watts/micron)
Luminescent wavelength (Microns)
Available photo current (Amps/micron)
Source photo current (Amps/micron)
Optical wavelength (Microns)
Optical source frequency (Hertz)
Current gain (dB)
Unilateral power gain frequency (dB)
Max transducer power gain (dB)

```

下面通过例句，进一步了解extract的命令格式，例句尽量选择有代表性的。通过一个例句，务求能掌握这一类的抽取命令。

抽取工艺仿真特性的例句。

抽取栅氧化层厚度:

```
extract name="gateox" thickness oxide mat.occno=1 x.val=0.49
```

抽取结深:

```
extract name="nxj" xj silicon mat.occno=1 x.val=0.1 junc.occno=1
```

抽取结深的另一种办法:

```
extract name="Junction Depth" x.val from curve(depth,(impurity="Gallium" \
material="Silicon" mat.occno=1) -(impurity="Phosphorus" \
material="Silicon" mat.occno=1)) where y.val=0.0
```

抽取表面浓度:

```
extract name="chan surf conc" surf.conc impurity="Net Doping" \
material="Silicon" mat.occno=1 x.val=0.45
```

抽取 $x=0.1\mu\text{m}$ 处的硼浓度分布:

```
extract name="bcurve" curve(depth,boron silicon mat.occno=1 x.val=0.1) \
outfile="extract.dat"
```

抽取激活了的砷的总浓度:

```
extract name="Active_Arsenic" 1.0e-04 * (area from curve (depth,impurity= \
"Active Arsenic" material="Silicon" mat.occno=1))
```

抽取方块电阻:

```
extract name="n++ sheet rho" sheet.res material="Silicon" mat.occno=1 \
x.val=0.05 region.occno=1
```

extrat有内建的1D QUICKMOS和QUICKBIP仿真器，可以在工艺仿真器中就抽取其器件特性，下面举例抽取电容、电导、1dvt等电学特性。

抽取结电容:

```
extract start material="Silicon" mat.occno=1 bias=0.0 bias.step=0.25 \
bias.stop=5.0 region.occno=1
extract done name="cjcurve" curve(bias,1djunc.cap material="Silicon" \
mat.occno=1 region.occno=1 junc.occno=2) outfile="cj.dat"
```

抽取电导:

```
extract start material="Polysilicon" mat.occno=1 bias=0.0 bias.step=0.2 \
bias.stop=2 x.val=0.45
extract done name="cond v bias" curve(bias,1dn.conduct material="Silicon" \
mat.occno=1 region.occno=1)
```

抽取结的击穿电压:

```
extract start material="Silicon" mat.occno=1 bias=0.0 bias.step=0.25 \
bias.stop=30.0 region.occno=1
extract done name="jbv" x.val from curve(bias, n.ion material="Silicon" \
mat.occno=1 region.occno=1) where y.val=1.0
```

抽取 V_t :

```
extract name="n1dvt" 1dvt ntype vb=0.0 qss=1e10 x.val=0.49
```

也可用下面的办法抽取 V_t :

```
extract start poly mat.occno=1 bias=0.0 bias.step=0.20 bias.stop=3.0
extract done name="vt_from_cond_curve" xintercept(maxslope(curve(bias, \
1dn.conduct silicon mat.occno=1 region.occno=1)))
```

抽取器件仿真的特性:

器件仿真的方法是对器件的电极施加电流、电压、磁场或是光照等，对器件的端电流电压和器件内部的电学量进行仿真计算（solve）。仿真时solve得到的器件信息（含电学信息）可以保存在结构文件（*.str）或日志文件（*.log）中。

在抽取前需要导入抽取的来源文件,如:

```
extract init inf="device_test_*.log"
```

从结构文件中抽取solve后的特性:

```
extract name="test" 2d.max.conc impurity="E Field" material="Silicon" \
x.val=0.5
```

I - V特性抽取:

```
extract name="iv" curve(v."emitter1", i."base2")
```

瞬态特性的抽取:

```
extract name="It curve" curve(time, i."anode")
```

随温度变化的特性:

```
extract name="VdT" curve(v."drain", temperature)
```

随频率变化的特性:

```
extract name="Idf" curve(i."drain", frequency)
```

抽取电容:

```
extract name="cv" curve(c."electrode1" "electrode2", v."electrode3")
```

抽取电导:

```
extract name="gv" curve(g."electrode1" "electrode2", v."electrode3")
```

抽取其他的电学参数的曲线:

```
extract name="IdT" curve(elect."parameter", v." drain")
```

轴处理在器件特性抽取中显得尤为重。如 β , V_t 等都需要由输出的电流电压特性中进行计算得到，下面提供一些对轴操作的例句。由于不是很难看懂，在此也不做特别说明。

```
extract name="big iv" curve(v."gate"/50, 10*i."drain")
```

```
extract name="combine" curve(i."collector", i."collector"/i."base")
```

```
extract name="dydx" deriv(v."gate", i."drain") outfile="dydx.dat"
```

```
extract name="dydx2" deriv(v."gate", i."drain", 2) outfile="dydx2.dat"
```

```
extract name="limit" max(curve(v."gate", i."drain", x.min=0.5 x.max=2.5)) \
  outf="limit.dat"
```

```
extract name="max beta" max(curve(i."collector", i."collector"/i."base"))
```

```
extract name="vt" xintercept(maxslope(curve(abs(v."gate", abs(i."drain")))))
```

```
extract name="Ic[Vb=2.3]" y.val from curve(abs(v."base"), abs(i."collector")) \
  where x.val = 2.3
```

在介绍extract特性时已经简要的提到了轴操作，包括min(),ave()等，下面列出这些轴的操作，以便抽取的时候做参考。

```
min(curve)
max(curve)
ave(curve)
minslope(curve)
maxslope(curve)
slope(line)
xintercept(line)
yintercept(line)
area from curve
area from curve where x.min=X1 and x.max=X2
x.val from curve where y.val=k
y.val from curve where x.val=k
x.val from curve where y.val=k and val.occno=n
y.val from curve where x.val=k and val.occno=n
grad from curve where y.val=k
grad from curve where x.val=k
```

1.4 学习方法

Silvaco给我的体会是功能全面、易学易用、运算速度快，自定义材料和C解释器等提供很好的扩展功能。所以还是很值得学习一下的。

就我个人的经验而言，Silvaco TCAD的示例是很好的教材。示例提供了丰富的语法使用实例、仿真流程、仿真方案等，是学习Silvaco TCAD的绝佳资料。在学习时可以一边参考example的语法和方案一边查阅用户手册。官方网站上的资源也相当丰富，是学习者千万不可错过的。

我曾经戏称参考示例或直接对示例中的语句使用“Ctrl+C”和“Ctrl+V”为“拿来主义”，我想此“拿来主义”实为学习方法的上上之选，相信看到此书的人会有同感。

学习资料:

- 程序自带的EXAMPLE
EXAMPLE路径, X:\sedatools\examples
- 用户手册:
如ATHENA手册, 位置 X:\sedatools\Athena\X.XX.0R\docs***_user.pdf.
- SILVACO官方网站 <http://www.silvaco.com>
- SIMUCAD官方网站 <http://www.simucad.com>
- SILVACO中国 <http://www.silvaco.cn>
- 有关EDA的专题网站:
EDA中心 <http://www.eda.ac.cn>
EDA中国门户网站 <http://www.edacn.net>
中国EDA技术网 <http://www.51eda.com>
EDA爱好者 <http://www.edafans.com>
EDA教学与研究 <http://www.edateach.com>

在此给仿真程序的学习提几点建议, 谨以共勉。

- 1、电脑以及仿真程序始终只是工具, 真正的发现还是来自于实践, 两者不可偏废;
- 2、电脑只能解决问题而不能提出问题, 所以电脑的功能永远有限;
- 3、仿真是否就意味着不真呢, 尽信软件不如没有软件;
- 4、在使用中学习而不是在记忆中学习。

总 结

这一章可看作是仿真前的准备。介绍了一些TCAD的基本概念, Silvaco软件的目录结构和文件位置等。详细讲解了DeckBuild的命令Go, Set, Tonyplot和Extract。最后对Silvaco TCAD的学习提了一些建议。从下章开始介绍Silvaco的TCAD部分。第二章介绍二维工艺仿真, 第三章介绍二维器件仿真, 第四章介绍Silvaco的一些高级特性。

第二章 二维工艺仿真

2.1 ATHENA 概述

工艺仿真组件ATHENA有很强的工艺仿真能力，包括单项工艺：注入、扩散、淀积、刻蚀、光刻，氧化等等。集成工艺从根本上说是由单项工艺的组合而来，ATHENA的示例库中含有很多单项工艺和集成工艺的例子。

ATHENA仿真时使用了大量的默认参数，参数存储在一些仿真程序的文件目录中，这些文件有：

- *athenamod*文件含有大量关于物理模型、扩散、氧化系数、数值计算方法、淀积和刻蚀机器特性、材料和光刻胶的光学特性等的默认参数；
- *implant.table*目录下的*std.table*和一些*svdp****文件含有离子注入的参数；
- *pls*和*models*目录下的****.mod*文件内含有高级的扩散模型的参数；
- *athenares*文件含有电阻率随掺杂浓度变化的数据。

ATHENA生成的标准结构文件¹可以在其他的仿真器如ATLAS或DevEdit中调用。SSF内含有mesh信息，solution信息，model信息和其他相关的参数。

表2-1是ATHENA对各工艺步骤的仿真能力：

表 2-1: Athena Features and Capabilities

Features	Capabilities
Bake	<ul style="list-style-type: none"> • Time and temperature bake specification. • Models photoresist material flow. • Models photo-active compound diffusion.
C-Interpreter	<ul style="list-style-type: none"> • Allows user-defined models for implant damage, Monte Carlo plasma etching and diffusion in SiGeC.
CMP	<ul style="list-style-type: none"> • Models Chemical Mechanical Polishing. • Hard and soft models or a combination of both. • Includes isotropical etch component.
Deposition	<ul style="list-style-type: none"> • Deposition • Hemispherical, planetary, and conical metallization models. • Unidirectional or dual directional deposition models. • CVD model. • Surface diffusion/migration effects. • Ballistic deposition models including atomistic positioning effects. • User-definable models. • Default deposition machine definitions.
Development	<ul style="list-style-type: none"> • Five different photoresist development models.

¹Standard Structure File, 以下简称SSF。

表2-1: Athena Features and Capabilities (续)

Features	Capabilities
Diffusion	<ul style="list-style-type: none"> • Impurity diffusion in general 2D structures including diffusion in all material layers. • Fully coupled point defect diffusion model. • Oxidation enhanced/retarded diffusion effects. • Rapid thermal annealing. • Models simultaneous material reflow and impurity diffusion. • Impurity diffusion in polysilicon accounting for grain and grain boundary components.
Etch	<ul style="list-style-type: none"> • Extensive geometric etch capability. • Wet etching with isotropic profile advance. • RIE model that combines isotropic and directional etch components. • Microloading effects. • Angle dependence of etchant source. • Default etch machine definitions. • Monte Carlo plasma etching. • Dopant enhanced etching.
Epitaxy	<ul style="list-style-type: none"> • Epitaxy
Exposure	<ul style="list-style-type: none"> • Model is based on the Beam Propagation Method simulating reflections and diffraction effects in non-planar structures with capability to take into account local modification of material optical properties the absorbed dose. • Defocus and large numerical aperture effects.
Imaging	<ul style="list-style-type: none"> • Experimentally verified Pearson and dual Pearson analytical models. • Experimentally verified Pearson and dual Pearson analytical models. • Binary Collision Approximation Monte Carlo calculations for crystalline and amorphous materials. • Universal tilt and rotation capability for both analytic and Monte Carlo calculations.
Oxidation	<ul style="list-style-type: none"> • Compressible and viscous stress dependent models. • Separate rate coefficients for silicon and polysilicon materials. • HCL and pressure-enhanced oxidation models. • Impurity concentration dependent effects. • Ability to simulate the oxidation of structures with deep trenches, undercuts, and ONO layers. • Accurate models for the simultaneous oxidation and lifting of polysilicon regions.
Silicidation	<ul style="list-style-type: none"> • Models for titanium, tungsten, cobalt, and platinum silicides. • Experimentally verified growth rates. • Reactions and boundary motion on silicide/metal and silicide/silicon interfaces. • Reactions and boundary motion on silicide/metal and silicide/silicon interfaces.

第4页的图1-3是ATHENA的输入输出框架，输入侧重工艺条件的控制，输出侧重工艺仿真的仿真能力。但是具体的一个工艺仿真怎么组织呢？

下面的步骤可建立ATHENA仿真流程：

- 1、建立仿真网格；
- 2、仿真初始化；
- 3、工艺步骤（如淀积，光刻，氧化，刻蚀，扩散...）；
- 4、抽取特性（如结深、材料厚度，浓度分布...）；
- 5、结构操作（导入结构，对结构旋转，做镜像和保存等）；
- 6、Tonyplot显示。

仿真的时候上面的步骤不一定都需要，顺序也可以灵活一些。如导入现成结构的话就可以直接从工艺步骤的仿真开始，要仔细观察变化可以在每一步工艺后存一个结构再用Tonyplot显示出来。

下面就从一个硼离子注入和退火的示例来看ATHENA仿真流程的建立，语法如下：

```
go athena
#TITLE: Simple Boron Anneal
#the x dimension definition
line x loc = 0.0 spacing=0.1
line x loc = 0.1 spacing=0.1
#the vertical definition
line y loc = 0 spacing = 0.02
line y loc = 2.0 spacing = 0.20
#initialize the mesh
init silicon c.phos=1.0e14
#perform uniform boron implant
implant boron dose=1e13 energy=70
#perform diffusion
diffuse time=30 temperature=1000
extract name="xj" xj silicon mat.occno=1 x.val=0.0 junc.occno=1
#plot the final profile
tonyplot
#save the structure
structure outfile=boron_implant.str
quit
```

各语句的意义如下（注意#号后面的是注释语句）：

启动工艺仿真器ATHENA：

```
go athena
```

网格定义（loc是location的简称）：

```
line x loc = 0.0 spacing=0.1
line x loc = 0.1 spacing=0.1
line y loc = 0 spacing = 0.02
line y loc = 2.0 spacing = 0.20
```

仿真初始化，亦即衬底定义：

```
init silicon c.phos=1.0e14
```

工艺步骤（硼离子注入和退火）：

```
implant boron dose=1e13 energy=70
diffuse time=30 temperature=1000
```

抽取仿真特性（结深）：

```
extract name="xj" xj silicon mat.occno=1 x.val=0.0 junc.occno=1
```

显示当前结构：

```
tonyplot
```

保存结构到文件：

```
structure outfile=boron_implant.str
```

退出仿真器：

```
quit
```




图 2-1： 硼注入及退火后的结构

图2-1是仿真后Tonyplot显示得到的结构。由于衬底杂质是磷，注入的是硼，所以就会形成结，抽取得到结深“xj”为 $0.700554\mu\text{m}$ （显示在实时输出窗口）。

2.2 仿真流程

上一节提供了一个简单工艺的仿真示例，从这一节开始将深入地介绍这些流程。

2.2.1 定义网格

在工艺仿真之前需要先定义衬底，在衬底的基础上再经过一系列工艺步骤来生成结构。在本书的最开始也已经提到了Silvaco TCAD是基于网格计算的仿真工具，也就是在网格点处计算其特性，而且网格点（node）的总数 N_p 不能超过20,000个。网格点的多少决定了仿真的精确程度和快慢，所以合理的定义网格分布很重要。

定义网格线的命令line，参数有x、y、location（通常简写成loc）和spacing。x和y参数设定网格线垂直于x轴或y轴，loc设定网格线在轴上的坐标，spacing设定网格线间的间距（loc和spacing的默认单位都是 μm ）。如果在几个loc处的spacing都是一样大小，那么网格线就是均匀分布的。如果spacing不一样，那么Silvaco会自动调整，尽量使loc处的spacing和设定的值一样，这时网格线就不是均匀的了。

均匀网格的例子，结果如图2-2所示：

```
line x loc = 0.0 spacing=0.1
line x loc = 1 spacing=0.1
line y loc = 0 spacing = 0.20
line y loc = 2.0 spacing = 0.20
```

非均匀网格的例子，结果如图2-3所示：

```
line x loc = 0.0 spacing=0.02
line x loc = 1 spacing=0.1
line y loc = 0 spacing = 0.02
line y loc = 2.0 spacing = 0.20
```




图 2-2： 均匀网格




图 2-3： 非均匀网格

对于图2-2和图2-3，衬底范围大小是一样的，除网格线的分布不一样外，网格点数也有差别（图2-2为121个点，图2-3为546个点）。

以上只是定义衬底的网格分布，但后续工艺如氧化，淀积等新引入的材料它的网格该如何定义呢？Silvaco对此也给予了充分考虑，这些工艺都有参数来指定网格的分布，下面就从淀积的示例来看怎样实现网格定义，对于相关工艺的其他工艺参数会在稍后的工艺仿真部分做详细介绍，此处略去不提。

淀积多晶硅以及对新增加的多晶硅层的网格进行定义（衬底采用图2-3的非均匀网格的衬底），语法及参数说明如下：

```
deposit polysilicon thick=0.50 divisions=10 dy=0.1 ydy=0.02 min.dy=0.01 \
min.space=0.01
```

网格控制的主要参数	含义
divisions	Total number of grid layers
dy	Nominal grid spacing(μm)
ydy	Grid spacing location(μm)
min.dy	Minimum grid spacing(μm)
min.space	Minimum edge spacing(μm)




图 2-4： 淀积多晶硅层的网格分布

从图2-4可以看出新淀积的多晶硅层（表面的一层，和硅衬底的界面以一条粗线隔开）的x方向的网格分布和衬底一样，纵向由参数divisions（通常简写为div）=10将其分成10条网格线。由于有dy,ydy,min.dy,min.space等参数，而且本身的thickness/div后得到的大小跟硅衬底表面的网格间距有差异，所以纵向网格也是非均匀的。

ATHENA的网格（“grid”）是由连接着三角形的一些点（“point”）组成的，每一个点有一个或多个节点（“node”）与之对应。如果point在一个材料或区域内部，则对应一

个node,如果在几种材料的交界处,则对应多个node。每一个node有代表特定区域特定材料在该point处的一些状态值(solution)如掺杂浓度,迁移率,电场强度等等。所以网格定义对仿真结果非常关键。通常是在最需要精确计算其特性的地方就定义得密一些,如样品表面要比衬底底部定义得密一些,异质结界面处,材料厚度很薄的层等等。

2.2.2 衬底初始化

网格定义了之后就是对衬底进行初始化,初始化的命令是initialize(可简写为init)。初始化参数选择如图2-5。语法如下:

INITIALIZE

```
[MATERIAL] [ORIENTATION=<n>][ROT.SUB=<n>][C.FRACTION=<n>]
[C.IMPURITIES=<n> | RESISTIVITY=<n>][C.INTERST=<n>][C.VACANCY=<n>]
[BORON | PHOSPHORUS | ARSENIC | ANTIMONY] [NO.IMPURITY]
[ONE.D | TWO.D | AUTO] [X.LOCAT=<n>] [CYLINDRICAL]
[INFILE=<c>] [STRUCTURE | INTENSITY]
[SPACE.MULT=<n>] [INTERVAL.R=<n>][LINE.DATA][SCALE=<n>][FLIP.Y]
[DEPTH.STR=<n>] [WIDTH.STR=<n>]
```




图 2-5: initialize的参数选择窗口

表 2-2: initialize的主要参数及其说明

和材料相关参数	说明
MATERIAL	衬底材料
ORIENTATION	衬底晶向，只有100,110和111,默认值是100
ROT.SUB	在BCA注入模型中指明衬底方向，单位为度，默认值是-45即表示剖面为[101]面
C.FRACTION	三元化合物中第一种材料的组分，如InGaAs的In
和掺杂相关的参数	说明
C.IMPURITIES	衬底所含杂质的种类及含量且为均匀掺杂($atom/cm^3$)
RESISTIVITY	衬底的电阻率。此参数指定时会忽略C.IMPURITIES，只对硼，磷，砷和锑杂质有效，($\Omega\cdot cm$)
C.INTERST	衬底材料的空隙浓度(cm^{-3})
C.VACANCY	衬底材料的空位浓度(cm^{-3})
BORON,PHOSPHORUS, ARSENIC,ANTIMONY	指定采用RESISTIVITY参数时的杂质种类
NO.IMPURITY	衬底不进行掺杂
和仿真维度相关的参数	说明
ONE.D, TWO.D, AUTO	仿真的维度，如果是ONE.D则需设定X.LOCAT参数。默认是AUTO，即一开始采用一维计算直到需要采用二维计算（通常从etch开始）
X.LOCAT	2D Mesh结构中但要采用一维计算时的位置
CYLINDRICAL	圆柱形对称结构（对称轴为X=0.0）的边界线
从文件定义初始化	说明
INFILE	导入文件结构，必须包含结构或强度信息
STRUCTURE,INTENSITY	导入文件的类型，默认为结构
网格和结构相关参数	说明
SPACE.MULT	设定全局的spacing的乘数
INTERVAL.R	临近网格线间距的最大比值，默认是1.5
LINE.DATA	指定在仿真时显示网格线的位置
SCALE	输入网格线的放大比率，默认为1.0
FLIP.Y	Boolean参数，设定mesh对x轴做镜像
DEPTH.STR,WIDTH.STR	初始衬底的深度和宽度(μm)

下面从几个例子看初始化命令的语法。

硅衬底，含硼浓度 $3.0 \times 10^{15} cm^{-3}$ ，晶向100。two.d为二维仿真初始化，默认是one.d（图2-1和图2-7的效果，X轴表示深度，Y轴为结构中的物理量）。

```
init silicon c.boron=3.0e15 orientation=100 two.d
```

硅衬底，杂质为磷，电阻率 $10\Omega\cdot cm$ ，晶向111：

```
init silicon phosphor resistivity=10 orientation=111
```

InGaAs衬底，In组分0.5：

```
init ingaas c.fraction=0.5
```

init命令除了衬底初始化外，之前也提到用init导入结构文件。

2.2.3 工艺步骤

ATHENA工艺仿真器可以对很多工艺进行仿真。表2-1中列出这些工艺仿真的能力，这些工艺包括Epitaxy, Diffusion, Oxidation, Etch, CMP, Deposition, Bake, Exposure, Development, Imaging, Silicidation等。

工艺仿真有ATHENA二维仿真器和SSUPREM3一维仿真器。

工艺仿真将在下一节详细介绍。

2.2.4 抽取特性

工艺仿真得到结果的形式有仿真得到的结构文件*.str，或是抽取的特性如材料厚度、结深、表面浓度、浓度分布、某杂质的总浓度、方块电阻等等。extrat有内建的一维QUICKMOS和QUICKBIP，可以在工艺仿真器中抽取得到器件结构的信息，如一维结电容(1djunc.cap)，一维电导(1dn.conduct)，阈值电压等。

抽取在1.3.5节已经详细提到了，如果需要请回到那里再仔细阅读或是查看手册。

2.2.5 结构操作

结构操作是用structure命令。structure可以保存和导入结构，也可以对结构做镜像或上下翻转。做镜像用mirror，参数有left, right, top, bottom等。上下翻转的参数是flip.y。

保存当前结构：

```
structure outfile=filename.str
```

对结构做镜像：

```
structure mirror left
```

结构上下翻转，将衬底置于表面（这样可以对衬底进行操作）：

```
structure flip.y
```

导入结构文件：

```
structure infile=filename.str
```

导入结构也可用init²，即在启动仿真器ATHENA时由初始化导入结构。

2.2.6 Tonyplot显示

Tonyplot可视化工具用来显示当前的结构或是已经保存的结构文件的结构或掺杂等信息，也可以按照某些设置(set文件)进行显示。工艺仿真中各个步骤得到的结构可用Tonyplot的动画功能做成动画（格式为*.gif的文件）。Tonyplot提供简单的函数计算功能如积分，还有Poisson Solver的功能可以在工艺生成的结构中计算一些电学特性。Tonyplot已经在第1.3.4小节做了详细的介绍。

以上的各个步骤即是ATHENA工艺仿真的流程，从下节开始详细介绍各工艺仿真模块的语句及其特性。

²在器件仿真中如果不关心结构怎么生成，只需要得到它的器件特性时由init导入结构会很方便。

2.3 单项工艺

本章主讲Silvaco TCAD的工艺仿真，这些工艺包括离子注入、扩散、淀积、刻蚀、光刻、外延等。除了简单的模型之外，也会讲到高级的仿真如Monte Carlo仿真，等离子体刻蚀等。语法及其主要参数进行说明的同时也提供了参数选择窗口（图2-5也是）。语法对Windows版本和Linux版本都有效，参数选择窗口是在Linux版本才有的。给出参数选择窗口可以帮助理解每一步工艺的主要参数有哪些。曾经一度喜欢Linux的图形化界面，个人觉得初学习时容易上手些，但对命令的理解和对仿真流程及控制的理解没有写语法来得真切。所以一直以来都建议使用语法来描述仿真，有条件的话也可以去体会体会。

2.3.1 离子注入




图 2-6: implant的参数选择窗口

语法:

IMPLANT

```
[GAUSS | PEARSON | FULL.LAT | MONTECARLO | BCA] [CRYSTAL | AMORPHOUS]
IMPURITY ENERGY=<n> DOSE=<n> [FULL.DOSE]
[TILT=<n>] [ROTATION=<n>] [FULLROTATION] [PLUS.ONE] [DAM.FACTOR=<n>]
[DAM.MOD=<c>] [PRINT.MOM] [X.DISCN=<n>] [LAT.RATIO1] [LAT.RATIO2]
[S.OXIDE=<n>][MATCH.DOSE|RP.SCALE|MAX.SCALE][SCALE.MOM][ANY.PEARSON]
```

[N.ION=<n>] [MCSEED=<n>] [TEMPERATURE=<n>] [DIVERGENCE=<n>]
 [IONBEAMWIDTH=<n>] [IMPACT.POINT=<n>] [SMOOTH=<n>]
 [SAMPLING] [DAMAGE] [MISCUT.TH] [MISCUT.PH] [TRAJ.FILE=<n>]
 [N.TRAJ=<n>] [Z1=<n>] [M1=<n>]

表 2-3: implant的主要参数及其说明

模型选择参数	说明
GAUSS,BCA,PEARSON, FULL.LAT,和MONTECARLO CRYSTAL和AMORPHOUS	离子注入模型选择, BCA (Binary Collision Approximation) 设定注入步骤中硅晶格结构, 默认是CRYSTAL
对所有注入模型有效的参数	说明
IMPURITY	注入的杂质种类
ENERGY	注入能量(KeV)。注入能量和注入杂质分布的峰值浓度位置有关
DOSE	注入剂量, 单位 cm^{-2} 。剂量是对注入方向而言的, 其他方向会不同
TILT	离子束注入的纵向角度, 单位度。默认值是 7°
ROTATION	注入离子束和仿真面(器件剖面)的角度, 单位度, 默认值是 30°
FULLROTATION	注入在所有的ROTATION角度都进行, 可理解为注入时在旋转wafer
FULL.DOSE	指定TILT角度时对注入剂量进行调整。 调整后 $DOSE = DOSE / \cos(TILT)$
PLUS.ONE和DAM.FACTOR	注入损伤计算, PLUS.ONE (同UNIT.DAMAGE)
PRINT.MOM	输出所有离子或材料的动量
Analytical注入模型参数	说明
X.DISCR	平均离子束宽度扩展因子, 默认值在0.1和0.2之间
S.OXIDE	SVDP模型的屏氧参数, 默认值 $0.001\mu m$
MATCH.DOSE,RP.SCALE, MAX.SCALE	多材料结构的计算方法, 默认是MATH.DOSE
Monte Carlo/BCA注入参数	说明
N.ION	设定计算的离子轨道数。SAMPLING参数未指定时一维结构默认是1,000, 二维结构默认是10,000
TEMPERATURE	注入时衬底的温度
DIVERGENCE	此参数设定时TILT变为 $TILT + / - DIVERGENCE/2.0$, 离子束发散角度, 默认值 1°
IMPACT.POINT	离子束注入到表面的位置, $X = (left + IMPACT.POINT * L)$, left为左边界, L是总长度
IONBEAMWIDTH	离子束束宽
SMOOTH	采用Gaussian方法平滑处理
SAMPLING	设定Monte Carlo方法统计样本
DAMAGE	损伤计算

提供几个离子注入的示例语句以供参考。

Analytical注入:

```
implant phosph dose=1e14 energy=50 tilt=10
```

SVDP硼注入：

```
implant boron dose=1e14 energy=40 tilt=0 s.oxide=0.005
```

Monte Carlo注入：

```
implant phosph dose=1e13 energy=100 bca tilt=0 rotation=0 temperature=300
```

下面的示例为改变注入角度观察注入后硼的分布。总共四个的注入角度（tilt）分别为1,2,7和10°。每一个条件都从定义网格开始到保存结构，差别只是注入角度的不同，图2-7为注入的结果。

```
implant boron energy=35 dose=1.e13 tilt=1 rotation=0 print.mom
```

```
implant boron energy=35 dose=1.e13 tilt=2 rotation=0 print.mom
```

```
implant boron energy=35 dose=1.e13 tilt=7 rotation=0 print.mom
```

```
implant boron energy=35 dose=1.e13 tilt=10 rotation=0 print.mom
```




图 2-7： 不同离子注入角度(TILT)的注入结果

由于注入的参数里有print.mom，此时留意一下输出窗口就会发现下面的信息：
The following implant parameters are found in the ** SVDP (UT at Austin) tables **
and used for: MATERIAL: silicon; ION: Boron; S.OXIDE.THICKNESS=0.001um. en-
ergy=35 dose=1e+013 tilt=10 rotation=0 range=0.1319 std.dev=0.0562 gamma=0.295714
kurtosis=3.08571 srangle=0.216829 sstd.dev=0.0687286 sgamma=0.396857 skurtosis=2.79571
dratio=0.808243 lstdev=0.056 kurtt=3 lsstd.dev=0.0032 skurtt=1.82

2.3.2 扩散

语法:
 DIFFUSE
 TIME=<n>[HOURS|MINUTES|SECONDS]
 TEMPERATURE=<n>[T.FINAL=<n>|T.RATE=<n>]
 [DRYO2|WETO2|NITROGEN|INERT][HCL.PC=<n>][PRESSURE=<n>]
 [F.O2=<n>|F.H2=<n>|F.H2O=<n>|F.N2=<n>|F.HCL=<n>]
 [C.IMPURITIES=<n>][NO.DIFF][REFLOW]
 [DUMP][DUMP.PREFIX=<n>][TSAVE=<n>][TSAVE.MULT=<n>]
 [B.MOD=<c>][p.MOD=<c>][AS.MOD=<c>][IC.MOD=<c>][VI.MOD=<c>]




图 2-8: diffuse的参数选择窗口

当扩散的氛围选择“Gas Flow”时会出现气体选择的窗口，如图2-9所示。




图 2-9: gasflow的参数选择窗口

表 2-4: diffuse的主要参数及其说明

扩散步骤的参数	说明
TIME	扩散的总时间
HOURS,MINUTES,SECONDS	扩散时间的单位，默认是MINUTES
TEMPERATURE	氛围的温度(°C)，恒温，超出700到1200 °C范围时扩散系数将会不够精确
T.FINAL	温度是变温时，设定最终的温度
T.RATE	温度是变温时，设定温度的变化率
扩散氛围的参数	说明
DRYO2,WETO2,NITROGEN,INERT	扩散的气体氛围，NITROGEN作用同INERT
HCL.PC	氧化剂气流中HCl的百分比
PRESSURE	指定气氛的分压，单位是atm，默认值为1
F.O2,F.H2,F.H2O,F.N2, F.HCL	气体的流速，此时不需定义DRYO2, WETO2,NITROGEN,HCL.PC
C.IMPURITIES	气体氛围中所含杂质($atoms/cm^3$)，可仿真预沉积
模型文件选择参数	说明
B.MOD,P.MOD,AS.MOD,IC.MOD,VI.MOD	模型文件boron.mod,phosphorus.mod,arsenic.mod,i.mod和defect.mod等的路径，默认位置X: sedatool\lib\athena\<version.number>\common\pls
混杂参数	说明
NO.DIFF	氧化和硅化时忽略杂质扩散
REFLOW	扩散时表面回流

扩散的示例语句。

预沉积示例语句：

```
diffuse time=30 temp=1000 c.boron=1.0e20
```

氧化示例语句（氧化层也可以由淀积和外延来得到）：

```
diffuse time=30 temp=1000 dryo2
```

气流示例语句：

```
diffuse time=10 temp=1000 f.o2=10 f.h2=10 f.hcl=.1
```

变温扩散语句：

```
diffus time=5 temp=800 t.final=1000 nitro press=1.5
```

```
diffus time=20 temp=1000 nitro press=1.5
```

```
diffus time=5 temp=1000 t.final=800 nitro press=1.5
```

高级扩散模型示例语句：

```
method pls
```

```
diffuse time=1 hour temp=950 c.boron=1e20 tsave=1 tsave.mult=10
```

```
dump.prefix=predep
```

2.3.3 淀积

语法:

DEPOSIT

MATERIAL [NAME.RESIST=<c>] THICKNESS=<n>

[SL.TO.POLY][TEMPERATURE=<n>]

[DIVISIONS=<n>] [DY=<n>][YDY=<n>][MIN.DY=<n>][MIN.SPACE=<n>]

[C.IMPURITIES=<n>][F.IMPURITIES=<n>][C.INTERST=<n>][F.INTERST=<n>]

[C.VACANCY=<n>][F.VACANCY=<n>][C.FRACTION=<n>] [F.FRACTION=<n>]

[MACHINE=<c>][TIME=<n>][HOURS | MINUTES | SECONDS]

[N.PARTICLE=<n>][OUTFILE=<c>][SUBSTEPS=<n>][VOID]

表 2-5: deposit的主要参数及其说明

通用参数	说明
MATERIAL	淀积的材料
NAME.RESIST	淀积光刻胶的类型
THICKNESS	淀积的厚度(μm)
TEMPERATURE	采用STRESS.HIST模型时淀积的温度
网格控制参数	在2.2.1节网格定义中已详细提到, 这些参数是DIVISIONS, DY, YDY, MIN.DY和MIN.SPACE。
淀积的掺杂参数	说明
C.IMPURITIES	淀积层的杂质浓度(cm^{-3})
F.IMPURITIES	只能和C.IMPURITIES一起设置, C.IMPURITIES为淀积层底部的杂质浓度, F.IMPURITIES为淀积层顶部的杂质浓度
C.INTERST	淀积层空穴的浓度
F.INTERST	作用和F.IMPURITIES类似, 只是是空穴浓度
C.VACANCY	淀积层的空位浓度
F.VACANCY	作用和F.VACANCY类似, 只是是空位浓度
C.FRACTION	淀积材料为三元化合物时指定第一种元素的组分, 如AlGaAs中的Al
F.FRACTION	作用和F.VACANCY类似, 只是是元素组分
GR.SIZE	淀积多晶硅的晶粒尺寸, 只有POLY.DIFF模型时有效
F.GR.SIZE	作用和F.VACANCY类似, 只是是多晶硅的晶粒尺寸
ELITE淀积模型参数	说明
MACHINE	淀积使用的机器名, 在RATE.ETCH状态参数中设定
TIME	淀积的时间
HOURS,MINUTES,SECONDS	TIME的单位, 默认是MINUTES
N.PARTICLE	Monte Carlo淀积时计算弹道颗粒数
OUTFILE	将Monte Carlo颗粒位置的信息存入文件
SUBSTEPS	ELITE模型的分步淀积的每一步
VOID	指定淀积材料未填充时空隙的形成




图 2-10: deposit的参数选择窗口




图 2-11: rate.deposit的参数选择窗口

下面提供几个示例，从示例中来看deposit的使用。

保角淀积示例，淀积二氧化硅 $0.1\mu m$ ，纵向含4个网格点：

```
deposit oxide thick=0.1 division=4
```

淀积用户自定义的材料BPSG，且含有杂质硼和磷：

```
deposit material=BPSG thickness=0.1 div=6 c.boron=1e20 c.phos=1e20
```

网格控制示例：

```
deposit nitride thick=0.3 dy=0.1 ydy=0.3 div=10
```

ELITE淀积示例：

```
rate.depo machine=MOCVD dep.rate=0.1 u.m step.cov=0.75 tungsten
deposit machine=MOCVD time=1 minute
```

2.3.4 刻蚀

语法：

ETCH [MATERIAL][NAME.RESIST]

[ALL | DRY][THICKNESS=<n>][ANGLE=<n>] [UNDERCUT=<n>]

[LEFT|RIGHT|ABOVE|BELOW][P1.X=<n>][P1.Y=<n>][P2.X=<n>][P2.Y=<n>]

[START | CONTINUE | DONE][X=<n>][Y=<n>]

[INFILE=<c>][TOP.LAYER][NOEXPOSE][DT.FACT=<n>][DT.MAX=<n>]

[DX.MULT=<n>][MACHINE=<c>][TIME=<n>][HOURS | MINUTES | SECONDS]

[MC.REDEPO][MC.SMOOTH=<n>][MC.DT.FACT=<n>][MC.MODFNAME=<c>]




图 2-12: etch的参数选择窗口

表 2-6: etch的主要参数及其说明

几何刻蚀参数	说明
MATERIAL	被刻蚀的材料
NAME.RESIST	刻蚀光刻胶的类型
ALL	指定的材料完全被刻蚀
DRY	刻蚀表面形貌不变，整体下降THICKNESS大小，如果ANGLE和UNDERCUT设定了，则表面形貌会受其影响
THICKNESS	干法刻蚀的刻蚀厚度(μm)
ANGLE	侧墙倾斜角度(度)，默认为90°，即垂直
UNDERCUT	掩膜下刻蚀延伸的距离(μm)，默认值为0
LEFT,RIGHT,ABOVE,BELOW P1.X, P1.Y, P2.X, and P2.Y	梯形刻蚀，刻蚀区域需由相应的P1.X,P1.Y和P2.X,P2.Y指定
START,CONTINUE,DONE	指定left/right/above/below刻蚀的位置(μm)，p1参数不可省
INFILE	任意的刻蚀，由一些点（参数X,Y指明位置）连成的线组成的区域
TOP.LAYER	由输入文件说明刻蚀剖面
NOEXPOSE	只有表层材料被刻蚀
ELITE组件的物理刻蚀参数	新得到的表面不会导致并发的氧化或淀积
MACHINE	说明
TIME	刻蚀机器
HOURS,MINUTES,SECONDS	刻蚀的时间
MC.PLASMA模型参数	刻蚀时间的单位
MC.REDEPO	说明
MC.SMOOTH	仿真时考虑重淀积，默认值ture
MC.DT.FACT	表面平滑的水平
	Monte Carlo刻蚀和重淀积的时间步骤控制

表 2-7: RIE etch和wet etch的主要参数及其说明

RIE和湿法刻蚀的主要参数	说明
A,H,A.M,A.S,U.H,U.M,U.S和N.M	刻蚀速率单位，A,U,N和H,M,S分别是长度和时间单位的缩写
DIRECTIONAL	RIE模型中定义刻蚀速率的方向组成（各向异性）
ISOTROPIC	RIE和Wet Etch中各向同性的刻蚀速率（这会导致掩膜的underetching）
DIVERGENCE	指定RIE模型中离子束的发散，到达硅片时采用高斯分布
CHEMICAL	DIVERGENCE不为0时RIE模型的刻蚀速率

当需要更高级的模型时可以采用ELITE组件。仿真特性有Wet Etch,RIE,Plasma Etch和Monte Carlo Plasma Etch。命令rate.etch中的machine参数的命名没有任何实质的意义，machine只是定义工艺设备、试剂和气体等工艺的共同参数。在后续工艺中可以采用这些共同的参数对不同材料，不同工艺条件进行仿真。如下面的plasma刻蚀例子刻蚀腔参数pressure,freq等，离子参数Tgas,Mion,Qio等，这些属性都描述Machine=PEMach。而在刻蚀Silicon材料时就采用了此Machine，这时Machine的属性就会自动复制过来。请回想一下还有没有类似的参数呢？个中意味还请认真体会。




图 2-13: rate.etch的参数选择窗口

表 2-8: plasma etch的主要参数及其说明

等离子体刻蚀主要参数	说明
PRESSURE	等离子体刻蚀机反应腔的压强(mTorr)，默认值50mTorr
TGAS	等离子体刻蚀机反应腔中气体的温度(K)，默认为300K
TION	等离子体刻蚀机反应腔中离子的温度(K)，默认值300K
VPDC	等离子体鞘层的DC偏压(V)，默认值32.5V
VPAC	等离子体鞘层和电珠之间的AC电压(V)，默认是32.5V
LSHDC	鞘层的平均厚度(mm)，默认值0.005mm
LSHAC	鞘层厚度的AC组成(mm)，默认值是0.0
FREQ	AC电流的频率(Mhz)，默认值13.6MHz
MGAS	气体原子的原子量，默认值为40
MION	等离子体离子的原子量默认值40
CHILD.LANGM,COLLISION, LINEAR,CONSTANT	只计算等离子体鞘层的电压降的模型，默认为CONSTANT
IONS.ONLY	中性颗粒在等离子体仿真中将忽略，默认为false
NPARTICLES	用蒙特卡诺计算来自等离子体的离子流的颗粒数，默认值10,000
ENERGY.DIV	等离子体离子流按照能量大小进行计算，能量总份数，默认为50
K.I	等离子体刻蚀速率的线性系数
K.F	化学流相关的等离子体刻蚀速率
K.D	淀积流量相关的等离子体刻蚀速率

留意一下图2-13就会发现刻蚀类型里只有Wet Etch和RIE，而Plasma Etch和Monte Carlo Plasma Etch是没有的。这也是我为什么建议使用直接写命令的方法来学习Silvaco TCAD的原因，因为有很多命令及参数在图形化界面下是没有列出来的。一旦对图形化界面形成依赖，那么语法的学习就会觉得吃力。

表 2-9: Monte Carlo plasma etch的主要参数及其说明

Monte Carlo等离子体刻蚀参数	说明
ION.TYPES	刻蚀的离子种类数（注意下面有些参数后面的数字1）
MC.POLYMP	喷射的聚合物颗粒数按体积规格化（用于MC仿真）
MC.RFLCTDIF	反射类型（1对应漫反射，0对应理想镜面反射）
MC.ETCH1	第一种离子的刻蚀速率参数，无单位
MC.ALB1	第一种离子的反射系数，（0为没有反射，1为完全反射）
MC.PLM.ALB	聚合体颗粒的反射系数，（0为没有反射，1为完全反射）
MC.NORM.T1	第一种离子的标准温度
MC.LAT.T1	第一种离子的横向的温度
MC.ION.CU1	第一种等离子体离子流密度，单位 $ions/second/cm^2$
MC.PARTS1	用于蒙特卡诺计算的第一种类型离子颗粒数
MC.ANGLE1	第一种离子的入射角，默认为0

刻蚀例句。

简单的几何刻蚀语句：

etch silicon left p1.x=0.5
etch aluminium right p1.x=1.5
etch oxide all

任意几何形状刻蚀语句（图2-14中掩膜即由此得到）：

etch oxide start x=1.5 y=0.0
etch continue x=1.5 y=-0.6
etch continue x=3.5 y=-0.6
etch done x=3.5 y=0.0

对材料刻蚀一定的厚度（效果如图2-14中(a)）：

etch silicon thick=0.5

各项异性刻蚀语句（效果如图2-14中(b)）：

etch dry thick=0.3

湿法刻蚀的语句（效果如图2-14中(c)）：

rate.etch machine=wet silicon wet.etch isotropic=0.5 u.m
etch machine=wet time=2 minutes

物理刻蚀语句（效果如图2-14中(d)）：

rate.etch machine=plasma1 silicon u.m rie isotropic=0.1 direct=0.9
etch machine=plasma1 time=2 minutes

等离子体刻蚀例句（效果如图2-14中(e)）：

```
Rate.Etch Machine=PEMach \
  Plasma \
  Pressure = 3.75 \
  Tgas = 300.0 \
  Tion = 3000.0 \
  Vpdc = 32.5 \
  Vpac = 32.5 \
  Lshdc = 0.005 \
  Lshac = 0.0 \
  Freq = 13.56 \
  Nparticles = 4000 \
  Mgas = 40.0 \
  Mion = 40.0 \
  Constant \
  Energy.Div = 50 \
  Qio = 1.7e-19 \
  Qcht = 2.1e-19
# Define the plasma etch parameters for Silicon
Rate.Etch Machine=PEMach \
  Plasma \
  Material=silicon \
  k.i = 1.1
```

Monte Carlo等离子体刻蚀例句（效果如图2-14中(f)）：

```
rate.etch machine=MCETCH \
  silicon mc.plasma ion.types=2 \
  mc.parts1=10000 mc.norm.t1=10.0 mc.lat.t1=1.0 \
  mc.ion.cu1=10 mc.etch1=1e-05 mc.alb1=0.2 \
  mc.parts2=10000 mc.norm.t2=6.0 mc.lat.t2=1.0 \
  mc.ion.cu2=5 mc.etch2=1e05 mc.alb2=0.2 \
  mc.polympt=3000 mc.plm.alb=0.9 mc.rflctdif=0.5
# Define the Monte Carlo plasma etch parameters for Silicon
etch machine=MCETCH \
  time=3 minutes mc.sm=0.001 \
  mc.redepo=f mc.dt.fact=2
```

图2-14为各种刻蚀效果的比较。掩膜为二氧化硅，厚度 $0.5\mu\text{m}$ 。被刻蚀材料为硅。图中沙漏标注的线（(1.5,0) 到(3.5,0)）为原来掩膜开口，叉号标记的曲线为刻蚀后的轮廓。




图 2-14: 各种刻蚀效果的比较

2.3.5 外延

语法

EPITAXY

TIME=<n> [HOURS | MINUTES | SECONDS]

TEMPERATURE=<n> [T.FINAL=<n> | T.RATE=<n>]

[THICKNESS=<n> | GROWTH.RATE=<n>]

[C.IMPURITIES=<n>][F.IMPURITIES=<n>][C.INTERST=<n>][F.INTERST=<n>]

[C.VACANCY=<n>][F.VACANCY=<n>]

[DIVISIONS=<n>][DY=<n>][MIN.DY=<n>][YDY=<n>][SI_TO_POLY]

表 2-10: epitaxy的主要参数及其说明

外延步骤参数	说明
TIME	淀积的时间数值
HOURS,MINUTES,SECONDS	淀积时间的单位，默认为minutes
TEMPERATURE	淀积氛围的温度（ $^{\circ}\text{C}$ ）超出700到1200 $^{\circ}\text{C}$ 时扩散系数会不精确
T.FINAL	变温时指定最终的温度
T.RATE	变温时的温度变化速率（ $^{\circ}\text{C}/\text{minute}$ ）
THICKNESS	淀积层的厚度（ μm ）
GROWTH.RATE	当淀积厚度未指定时，定义淀积速率（ $\mu\text{m}/\text{minute}$ ）
掺杂相关的参数	说明
C.IMPURITIES	外延时外延层的掺杂（ cm^{-3} ）
F.IMPURITIES	和C.IMPURITIES一起表征外延的非均匀掺杂。C.IMPURITIES为底部的掺杂，F.IMPURITIES为顶部的掺杂
C.INTERST	外延时外延层空穴的浓度（ cm^{-3} ）
F.INTERST	作用和F.IMPURITIES类似
C.VACANCY	外延时外延层空位的浓度（ cm^{-3} ）
F.VACANCY	作用和F.IMPURITIES类似
网格参数	说明
DIVISIONS	外延层纵向网格点数，默认为10
DY	外延层的名义间隔（ μm ）
YDY	DY的位置，由新生长的外延层的顶部计算间隔
MIN.DY	最小的间隔（0.001 μm ）
SI_TO_POLY	晶体态硅只在硅上淀积

现列出一些外延的示例语句。淀积速率示例：

```
epitaxy time=10 temp=1150 c.boron=5e14 growth.rate=0.5
```

时间和温度示例：

```
epitaxy thick=6 time=10 temp=1050 c.phos=5e15 divisions=20
```

网格控制示例：

```
epitaxy thick=10 time=10 temp=1200 dy=0.5 ydy=5.0 divisions=40
```

2.3.6 抛光

Polish为化学机制的抛光(CMP)。

语法

POLISH

MACHINE=<c>[TIME=<n>][HOURS | MINUTES | SECONDS]
[DX.MULT=<n>][DT.FACT=<n>][DT.MAX=<n>]




图 2-15: polish的参数选择窗口

表 2-11: polish的主要参数及其说明

化学机械抛光的主要参数	说明
MACHINE	抛光机器的名称
TIME	抛光时间
HOURS,MINUTES,SECONDS	抛光时间的单位,
DX.MULT	ELITE抛光的乘数器, 在抛光计算中离散化的尺寸将乘以该值
DT.FACT	时间步长尺寸, 不得超过0.5, 越小越精确
DT.MAX	时间步长的上限,

抛光示例语句:

polish machine=Polish1 time=1.0 minutes dx.mult=1.0 dt.fact=0.25 dt.max=0.25

RATE.POLISH

语法:

RATE.POLISH

MACHINE = <c> MATERIAL | NAME.RESIST=<n>
[A.H|A.M|A.S|U.S|U.M|U.H|N.M]
[SOFT.RATE][HEIGHT.FAC=<n>][LENGTH.FAC=<n>][KINETIC.FAC=<n>]
[MAX.HARD=<n>] [MIN.HARD=<n>] [ISOTROPIC=<n>]




图 2-16: rate.polish的参数选择

表 2-12: rate.polish的主要参数及其说明

rate.polish参数	说明
MATERIAL	抛光材料
A.H,A.M,A.S,U.H,U.M,U.S,N.M	抛光速率
SOFT.RATE	软抛光速率
HEIGHT.FAC	软抛光模型的纵向形变尺度(μm)
LENGTH.FAC	软抛光模型的横向形变尺度(μm)
KINETIC.FAC	软抛光的Kinetic因子
MAX.HARD	硬抛光的最大速率，同样的有MIN.HARD
ISOTROPIC	POLISH模型的各向同性刻蚀速率

rate.polish示例语句:

```
rate.polish machine=CMP silicon a.s soft=10 height.fac=0.0 \
  lenght.fac=0.0 kinetic.fac=0.0 max.hard=0.0 min.hard= 0.0 \
  isotropical=0.0
```

2.3.7 光刻

Photo (OPTOLITH) 模块有MASK、ILLUMINATION、PROJECTION、FILTER、LAYOUT、IMAGE、BAKE、EXPOSE和DEVELOP等工艺。

各个工艺的语法、参数说明及例句在下面将一一介绍。由于我对光刻不甚了解，在解释的时候肯定有很多问题，请大家看的时候要留心，错误的地方还请及时指正，我会重新整理此文档的。

1. MASK

语法:

MASK

NAME=<c> [REVERSE] [DELTA=<n>]

表 2-13: mask的主要参数及其说明

mask参数	说明
NAME	导入掩膜的文件名，需加双引号
REVERSE	负极性光刻胶
DELTA	掩膜尺寸的偏移，相应会改变mask的CD（critical dimension）

光刻胶的掩膜版，由Maskviews组件绘制。mask状态设定了，athena就可以淀积和刻蚀光刻胶。Maskviews是掩膜编辑器，界面如图2-17。可以导入的文件类型有GDS2 Stream Format(*.gds)，GDS2 Technology Format(*.tech)，CIF Format(*.cif)，Optolith Format(*.sec)。可以保存的项目Save Grid(*.grid)，Save Objects(*.user)，Save Biases(*.bias)。保存的其他文件类型可以是*.lay、*.gds和*.cif。




图 2-17: Maskviews的界面

个人觉得Maskviews很不好用，LEdit使用起来则要方便得多。我想可以用LEdit编辑掩膜图形，然后由Maskviews导入后再存为*.lay的格式。这样掩膜编辑起来就要省事很多。可能也是我对Maskviews不太习惯的缘故吧。

2. ILLUMINATION

ILLUMINATION, 设置OPTOLITH的照明参数。

语法:

ILLUMINATION

[I.LINE|G.LINE|H.LINE|KRF.LASER|DUV.LINE|ARF.LASER|F2.LASER|LAMBDA=<n>]
[X.TILT=<n>][Z.TILT=<n>] [INTENSITY=<n>]




图 2-18: illumination的参数选择窗口

表 2-14: illumination的主要参数及其说明

照明系统参数	说明
I.LINE,G.LINE,H.LINE, KRF.LASER(alias DUV.LINE), ARF.LASER,F2.LASER	照明系统采用的波长，这些波长相应是0.365,0.436,0.407,0.268, 0.193,0.157 μm
LAMBDA	定义和改变光源的波长(μm), 以单色光源对待
X.TILT,Z.TILT	照明系统和光轴的角度(°)
INTENSITY	定义和改变振幅的绝对值即掩膜或网线面的强度

i线例句:

illumination i.line x.tilt=0.2 z.tilt=0.5

用户定义的波长例句:

illumination lambda=0.365 x.tilt=0.25 z.tilt=0.1 intensity=2.0

3. PROJECTION

语法:

PROJECTION [NA=<n>][FLARE=<n>]

projection的参数很少，只有两个。“NA”参数设置光学投影系统的孔隙数，“FLARE”设定成像时出现的耀斑数，以百分比描述。




图 2-19: projection的参数选择窗口

projection例句:

```
projection na=0.50 flare=1
```

4. FILTER

设定OPTOLITH的发射孔(pupil)类型和光源形状及其滤波特性。有四种不同的发射孔类型并允许傅立叶转换平面空间滤波。

pipil.filter语法:

```
PUPIL.FILTER    CIRCLE | SQUARE | GAUSSIAN | ANTIGAUSS
[GAMMA=<n>][IN.RADIUS=<n>][OUT.RADIUS=<n>][PHASE=<n>]
[TRANSMIT=<n>][CLEAR.FIL]
```

illum.filter语法:

```
ILLUM.FILTER    [CIRCLE | SQUARE | GAUSSIAN | ANTIGAUSS | SHRINC]
[GAMMA=<n>] [RADIUS=<n>][ANGLE=<n>] [SIGMA=<n>]
[IN.RADIUS=<n>][OUT.RADIUS=<n>][PHASE=<n>][TRANSMIT=<n>][CLEAR.FIL]
```




图 2-20: filter的参数选择窗口

表 2-15: pupil.filter的主要参数及其说明

pupil.filter参数	说明
CIRCLE,SQUARE,GAUSSIAN, ANTIGAUSS	发射孔形状
GAMMA	GAUSSIAN和ANTIGAUSS发射孔透明度
IN.RADIUS,OUT.RADIUS	发射孔或照明系统等环状区域的强度透射率和相透射率,
PHASE	相变 $-180^{\circ} \leq PHASE \leq 180^{\circ}$
TRANSMIT	透射率
CLEAR.FIL	重置滤波列表

illum.filter的参数说明请参考表2-15关于pupil.filter的说明，两者大体类似。请注意图2-20中两个复选框实际上是二选一的，下面的例句也有体现。

pupil.filter例句:

pupil.filter square

pupil.filter in.radius=0.1 out.radius=0.2 phase=0 transmit=0

illum.filter例句:

illum.filter cirle in.radius=0.08 out.angle=0.15

illum.filter gaussian radius=0.05 angle=0.0 gamma=1.00 sigma=0.3 clear.fil

5. LAYOUT




图 2-21: layout的参数选择窗口

LAYOUT描述光刻时手动输入掩膜的特征。

语法:

LAYOUT

[LAY.CLEAR=<n>]
[X.LOW=<n>][Z.LOW=<n>][X.HIGH=<n>][Z.HIGH=<n>][X.TRI=<n>][Z.TRI=<n>]
[HEIGHT=<n>][WIDTH=<n>][ROT.ANGLE=<n>][X.CIRCLE=<n>][Z.CIRCLE=<n>]
[RADIUS=<n>][RINGWIDTH=<n>][MULTIRING][PHASE=<n>][TRANSMIT=<n>]
[INFILE=<c>][MASK=<c>][SHIFT.MASK=<c>]

表 2-16: layout的主要参数及其说明

版图参数	说明
LAY.CLEAR	清除以前的版图
X.LOW,X.HIGH	X边界值
Z.LOW,Z.HIGH	Z边界值, X和Z的边界组成一个矩形区域
X.TRI,Z.TRI	三角形(triangular)右角的X、Y坐标
HEIGHT	三角形右角的高度
WIDTH	三角形右角的底部宽度
ROT.ANGLE	对X轴旋转的角度, $-180^\circ \leq ROT.ANGLE \leq 180^\circ$, 默认为 0°
X.CIRCLE,Z.CIRCLE	圆或环中心的X和Z坐标
RADIUS	圆的半径或圆环的外径
RINGWIDTH	掩膜中圆环的宽度
MULTIRING	多圆环, 环的宽度及圆环间的距离由RINGWIDTH描述, 圆环数N满足 $2N * RINGWIDTH < RADIUS$
PHASE	相变, $-180^\circ \leq PHASE \leq 180^\circ$, 默认为 0°
TRANSMIT	光强透射率, $0 \leq TRANSMIT \leq 1$, 默认为整体透过
INFILE	Maskviews编辑的版图文件名
MASK	成像计算时的掩膜名
SHIFT.MASK	额外的掩膜层(通常是相变层)的名称

Layout示例:

```
layout x.lo=-2 x.hi=2 z.lo=-0.5 z.hi=0.5 rot.angle=45 transmit=1
```

```
layout clear
```

```
layout infile=my.lay mask=new
```

```
layout x.cir=-10 z.cir=10 radius=8
```

6. IMAGE

计算一维或二维的成像。

语法:

IMAGE [INFILE=<c>] [DEMAG=<n>] [GAP=<n>]
[OPAQUE | CLEAR] [DEFOCUS=<n>][CENTER]
[WIN.X.LOW=<n>][WIN.X.HIGH=<n>][WIN.Z.LOW=<n>][WIN.Z.HIGH=<n>]
[DX=<n>][DZ=<n>][X.POINTS=<n>][Z.POINTS=<n>][N.PUPIL=<n>]
[MULT.IMAGE][X.CROSS | Z.CROSS][ONE.DIM]




图 2-22: image的参数选择窗口

表 2-17: image的主要参数及其说明

成像的参数	说明
INFILE	导入Maskviews得到的掩膜数据文件，包含透射率和相的信息，文件名通常以extension.sec结尾
DEMAG	损伤因子，这样版图上的所有元素及成像窗口和网格等都将减小
GAP	定义接近试曝光中掩膜到wafer的间隙(μm)
OPAQUE,CLEAR	掩膜类型，OPAQUE为不透明，CLEAR为清除掩膜
DEFOCUS	散焦参数，如果小于0，在光阻上部，如果大于0则在光阻下部
CENTER	INFILE参数导入版图时的位置，中心点是(0,0)
WIN.X.LOW,WIN.X.HIGH, WIN.Z.LOW,WIN.Z.HIGH	成像窗口的最大最小的X和Z值
DX	镜像窗口的中X值
DZ	镜像窗口的中Z值。如果DX和DZ未指定，则有参数X.POINT和Z.POINT
X.POINTS,Z.POINTS	镜像窗口中心对应的X值和Z值
N.PUPIL	定义和改变成像仿真时出发射孔projector的网格点数
MULT.IMAGE	之前和当前的像将会添加进来
X.CROSS,Z.CROSS	一维像平行于X轴和Z轴
ONE.DIM	使用一维成像模型

image例句。

从文件中导入信息:

```
image infile=my_mask.sec dx=0.05
```

```
image opaque dx=0.1 win.x.h=12 win.x.lo=0 win.z.h=10 win.z.l=2
```

```
image clear win.x.lo=-5 win.z.lo=-5 win.x.hi=5 win.z.hi=5 x.p=1 z.p=1
```

接近试曝光例句：

```
image win.x.lo=-45 win.z.lo=-45 win.x.hi=45 win.z.hi=45 dx=.5 gap=10
```

7. EXPOSE

OPTOLITH的曝光模块。

语法：

EXPOSE [INFILE=<c>] [PERPENDICUL | PARALLEL] [X.CROSS | Z.CROSS]
[CROSS.VALUE=<n>] [DOSE=<n>] [X.ORIGIN=<n>]
[FLATNESS=<n>] [NUM.REFL=<n>] [MULT.EXPOSE][POWER.MIN=<n>]
[FRONT.REFL=<n>] [BACK.REFL=<n>] [ALL.MATS=<n>]




图 2-23: expose的参数选择窗口

表 2-18: expose的主要参数及其说明

曝光的参数	说明
INFILE	由输入文件定义光强分布
PERPENDICUL,PARALLEL	TE波或TM波，默认为PERPENDICUL
X.CROSS,Z.CROSS	指定剖面平行于X轴(Z为常数)，Z.CROSS同理。X.CROSS为默认值
CROSS.VALUE	剖面对应的X或Z的值(μm)，默认为image窗口的中心位置
DOSE	曝光剂量(mJ/cm^2)
X.ORIGIN	beam照射的位置(μm)，默认为0.0
FLATNESS	为了精确性确定表面角度，单位度，默认0.25, $0 \leq FLATNESS \leq 1$
NUM.REFL	光反射的次数
FRONT.REFL	前表面反射，默认为false
BACK.REFL	后表面反射，默认为false
ALL.MATS	显示光强在所有材料中的分布，默认只显示photoresist
MULT.EXPOSE	多次曝光，可以再加expose命令来实现多次曝光
POWER.MIN	多次反射时能量减小到此值就不进行计算

曝光的一些例句:

expose infile=cross.sect num.refl=3

expose z.cross cross.val=0.1

expose dose=240.0 num.refl=5

8. BAKE

描述对光阻的后曝光和后坚膜时的烘烤。

语法:

BAKE

[DIFF.LENGTH=<n>] [TEMPERATURE=<n>] [REFLOW]

[TIME] [SECONDS | MINUTES | HOURS]

[DUMP=<n>] [DUMP.PREFIX=<c>]




图 2-24: bake的参数选择窗口

表 2-19: bake的主要参数及其说明

烘烤的参数	说明
DIFF.LENGTH	后烘的扩散长度，默认值0.05 μm
TEMPERATURE	烘烤温度（ $^{\circ}C$ ）
REFLOW	烘烤时考虑回流，默认为false
TIME	烘烤的时间长度数值
HOURS,MINUTES,SECONDS	烘烤的时间单位，默认为minutes
DUMP,DUMP.PREFIX	每一个DUMPth时间步长时保存结构文件

烘烤的一些例句:

bake diff.lenght=0.03

bake time=30 temp=100

考虑回流:

bake reflow time=25 temp=150

9. DEVELOP

OPTOLITH的显影模块。

语法:

DEVELOP

[MACK | DILL | TREFONAS | HIRAI | KIM | EIB]

[TIME=<n>] [STEPS=<n>] [SUBSTEPS=<n>]

[DUMP=<n>] [DUMP.PREFIX=<c>]




图 2-25: develop的参数选择窗口

表 2-20: develop的主要参数及其说明

显影的参数	说明
MACK,DILL,TREFONAS, HIRAI,KIM,EIB	显影采用的模型
TIME,STEPS,SUBSTEPS	显影控制。TIME为显影总时间(s)，STEPS设定ETCH将进行的次数，每一个SUBSTEP的时间长度为 $TIME/STEP * SUBSTEPS$
DUMP	决定每一步(step)显影完成之后是否保存结构
DUMP.PREFIX	DUM所保存文件的前缀，最后文件名将是DUMP.PREFIX* * *. * * *.str，其中* * *是当时的时间

显影的示例:

```
develop kim dump=1 time=50 steps=5
```

```
develop dill time=100 steps=3 substeps=25
```

RATE.DEVELOP

语法:

RATE.DEVELOP

[NAME.RESIST=<c>] [G.LINE|H.LINE|I.LINE|DUV.LINE|LAMBDA=<n>]

[A.DILL=<n>] [B.DILL=<n>] [C.DILL=<n>]

[E1.DILL=<n>] [E2.DILL=<n>] [E3.DILL=<n>]

[RMAX.MACK=<n>] [RMIN.MACK=<n>] [MTH.MACK=<n>] [N.MACK=<n>]

```
[RO.TREFONAS=<n>] [Q.TREFONAS=<n>]
[RO.HIRAI=<n>] [RC.HIRAI=<n>] [ALPHA.HIRAI=<n>]
[R1.KIM=<n>] [R2.KIM=<n>] [R3.KIM=<n>] [R4.KIM=<n>] [R5.KIM=<n>]
[R6.KIM=<n>] [R7.KIM=<n>] [R8.KIM=<n>] [R9.KIM=<n>] [R10.KIM=<n>]
[C0.EIB=<n>] [C1.EIB=<n>] [C2.EIB=<n>] [C3.EIB=<n>]
[DIX.0=<n>] [DIX.E=<n>]
```

表 2-21: rate.develop 的主要参数及其说明

rate.develop 的参数	说明
NAME.RESIST	光刻胶名称
G.LINE,H.LINE,I.LINE, DUV.LINE,LAMDBA	对应光刻胶的波长(μm)
A.DILL,B.DILL,C.DILL	Dill曝光模型的A、B和C常数
E1.DILL,E2.DILL,E3.DILL	Dill显影速率函数的E1、E2和E3常数
RMAX.MACK,RMIN.MACK, MTH.MACK,N.MACK	Mack显影模型的常数
R0.TREFONAS, Q.TREFONAS	Trefonas显影模型的常数
R0.HIRAI,RC.HIRAI, ALPHA.HIRAI	Hirai显影模型的常数
R1.KIM,R2.KIM,... R10.KIM	Kim显影模型的常数
CO.EIB,C1.EIB,C2.EIB,C3.EIB	Eib显影模型的参数
DIX.0,DIX.E	后曝光烘烤时光敏化合物扩散的指数前的常数(cm^2/sec)和激活能(eV)

rate.develop 设置例句:

```
rate.develop name.resist=test e1.dill=1 e2.dill=0.5 e3.dill=0.003
```

10. 完整的光刻流程

光刻的工艺有这么多, 该如何将这些工艺串起来呢? 下面给出一个例子。

```
go athena
set lay_left=-0.5
set lay_right=0.5
#
illumination g.line
illum.filter clear.fil circle sigma=0.38
#
projection na=.54
pupil.filter clear.fil circle
```

```

layout lay.clear x.lo=-2 z.lo=-3 x.hi=$lay_left z.hi=3
layout x.lo=$lay_right z.lo=-3 x.hi=2 z.hi=3
image clear win.x.lo=-1 win.z.lo=-0.5 win.x.hi=1 win.z.hi=0.5 dx=0.05 one.d
structure outfile=mask.str intensity mask
tonyplot mask.str
line x loc=-2 spac=0.05
line x loc=0 spac=0.05
line x loc=2 spac=0.05
line y loc=0 spac=0.05
line y loc=2 spac=0.2
init silicon orient=100 c.boron=1e15 two.d
deposit nitride thick=0.035 div=5
deposit name.resist=AZ1350J thick=.8 divisions=30
rate.dev name.resist=AZ1350J i.line c.dill=0.018
structure outfile=preoptolith.str
# tonyplot preoptolith.str
expose dose=240.0 num.refl=10
bake time=30 temp=100
develop kim time=60 steps=6 substeps=24
structure outfile=optolith.str
tonyplot optolith.str
quit

```




图 2-26: 完整光刻流程的仿真结果

图2-26是仿真得到的掩膜及光强分布和对光刻胶曝光及显影后的刻蚀结果。

2.4 集成工艺

上一节讲到了一些单项工艺，然而这只是单项工艺而已。对于BJT、MOS、CCD、LED、ESD、Laser、射频用电容、电感等等器件或由这些器件组成的小规模电路的工艺无非也是由这些基本的工艺经过组合而来。这自然会引发两个疑问。一、由哪些单项工艺组合得到；二、这些工艺的参数又是怎样得到的。这一节试图去回答第一个问题，很复杂，而且很多问题³不是软件能解决的，本书只能在仿真软件的使用上提供一些细微的帮助。下一节提供了第二个问题的部分答案，或是参数验证手段。

Silvaco 提供PDK(process design kits)服务，而且Silvaco和很多代工厂商合作开发PDK，这些厂商包括：



之前给出了两个完整工艺仿真例子。一个是介绍仿真流程时提供了一个离子注入及退火的例子，另一个是在介绍光刻工艺时给了一个光刻的完整示例。

下面提供MOS的工艺仿真实例。从其中可以得到很多有用的东西。如仿真流程的组织，仿真注释及适当的空行以增强可读性，适时保存、显示结构以及抽取特性可以提供更详尽的各阶段的仿真结果。相信经过之前详细地讲解，看懂下面的MOS示例中的语法应该不成问题。

```
go athena
#
line x loc=0 spac=0.1
line x loc=0.2 spac=0.006
line x loc=0.4 spac=0.006
line x loc=0.5 spac=0.01
#
```

³在第1章的末尾提了一些建议，这里还是要不厌其烦地提醒大家软件的学习应该只是解决问题的辅助而不是全部。

```
#
line y loc=0.00 spac=0.002
line y loc=0.2 spac=0.005
line y loc=0.5 spac=0.05
line y loc=0.8 spac=0.15
#
init orientation=100 c.phos=1e14 space.mul=2
# pwell formation including masking off of the nwell
#
diffus time=30 temp=1000 dryo2 press=1.00 hcl=3
#
etch oxide thick=0.02
#
#P-well Implant
implant boron dose=8e12 energy=100 pears
#
diffus temp=950 time=100 weto2 hcl=3
#
# N-well implant not shown
# welldrive starts here
diffus time=50 temp=1000 t.rate=4.000 dryo2 press=0.10 hcl=3
#
diffus time=220 temp=1200 nitro press=1
#
diffus time=90 temp=1200 t.rate=-4.444 nitro press=1
#
etch oxide all
#
# sacrificial "cleaning" oxide
diffus time=20 temp=1000 dryo2 press=1 hcl=3
#
etch oxide all
# gate oxide grown here
diffus time=11 temp=925 dryo2 press=1.00 hcl=3
#
# Extract a design parameter
extract name="gateox" thickness oxide mat.ocno=1 x.val=0.5
#
```

```
#vt adjust implant
implant boron dose=9.5e11 energy=10 pearson
#
depo poly thick=0.2 divi=10
#
#from now on the situation is 2-D
#
etch poly left p1.x=0.35
#
method fermi compress
diffuse time=3 temp=900 weto2 press=1.0
#
implant phosphor dose=3.0e13 energy=20 pearson
#
depo oxide thick=0.120 divisions=8
#
etch oxide dry thick=0.120
#
implant arsenic dose=5.0e15 energy=50 pearson
#
method fermi compress
diffuse time=1 temp=900 nitro press=1.0
# pattern s/d contact metal
etch oxide left p1.x=0.2
deposit alumin thick=0.03 divi=2
etch alumin right p1.x=0.18

# Extract design parameters
# extract final S/D Xj
extract name="nxj" xj silicon mat.occno=1 x.val=0.1 junc.occno=1

# extract the N++ regions sheet resistance
extract name="n++ sheet rho" sheet.res material="Silicon" \
    mat.occno=1 x.val=0.05 region.occno=1

# extract the sheet rho under the spacer, of the LDD region
extract name="ldd sheet rho" sheet.res material="Silicon" \
    mat.occno=1 x.val=0.3 region.occno=1
```

```

# extract the surface conc under the channel.
extract name="chan surf conc" surf.conc impurity="Net Doping" \
  material="Silicon" mat.occno=1 x.val=0.45

# extract a curve of conductance versus bias.
extract start material="Polysilicon" mat.occno=1 \
  bias=0.0 bias.step=0.2 bias.stop=2 x.val=0.45
extract done name="sheet cond v bias" \
  curve(bias,1dn.conduct material="Silicon" mat.occno=1 region.occno=1) \
  outfile="extract.dat"

# extract the long chan Vt
extract name="n1dvt" 1dvt ntype vb=0.0 qss=1e10 x.val=0.49

structure mirror right

electrode name=gate x=0.5 y=0.1
electrode name=source x=0.1
electrode name=drain x=0.9
electrode name=substrate backside

structure outfile=mos0.str

# plot the structure
tonyplot mos0.str -set mos0.set

```

上面示例中唯一没有讲到的就是“electrode”的定义，且留待讲器件仿真时进行介绍。其他器件就不一一举例了，整体流程都差不多。

图1-6是最终得到的结构，图1-7是该MOS结构仿真得到的输出特性曲线。

2.5 优化

在工艺仿真时如何得到想要的结果（如希望栅氧长到多厚，离子注入后方块电阻为多少等），这就需要对参数进行优化。Silvaco提供了优化工具Optimizer。

启动Optimizer的方式：Commands >Optimizer...

2.5.1 优化设置

优化前需对优化方案进行设置，包括误差范围、寻找次数等等，如图2-27最大误差设置为1%。




图 2-27: 优化方案的设置

工艺仿真的语句如下，下面就从这个扩散的工艺着手讲解优化工具的使用。

```
go athena
line x loc=0.0 spacing=0.02
line x loc=1.0 spacing=0.02
line y loc=0.0 spacing=0.02
line y loc=1.0 spacing=0.02
init silicon orient=111 c.boron=1e14
diffuse time=20 temp=1000 dryo2
#
extract name="Tox" thickness oxide mat.ocno=1 x.val=0
structure outfile=dry_thick.str
tonyplot
quit
```

2.5.2 待优化参数

选择优化参数，在DeckBuild的命令行中选中需要优化的工艺步骤，通常是一行（上一页灰底的扩散语句那一行）。当选中后在Optimizer界面MODE 框中选择Parameters然后执行Edit>Add，再将“time”和“temp”的复选框勾上。即出现图2-28所示的参数设置窗口。参数设置中可设置参数扫描的范围。




图 2-28： 待优化的参数设置

2.5.3 优化目标

优化的结果通常是抽取的工艺结果，如结深、材料厚度、浓度等等。在DeckBuild命令行中选中抽取的行（上一页灰底的抽取氧化层厚度的那一行），选中后在Optimizer界面MODE 框中选择Targets然后执行Edit>Add，再设置“Target Value”到目标值。如图2-29所示。




图 2-29： 优化的目标设置

2.5.4 优化结果

优化设置、参数和目标值设置好后在Optimizer界面MODE框中选择Results然后执行“Optimizer”，DeckBuild就会按照设置来回扫描工艺参数，直到抽取的结果和目标值在

设定的误差范围内。如图2-30为几次扫描的参数值及目标值“Tox”。因为目标设置为500，所以扫描到502.544的时候Silvaco认为已经满意就停了下来。




图 2-30： 优化得到结果

工艺优化是很好很方便的工具。至此，上一节集成工艺刚开始时提出的两个问题初步给出了解决方法。

总 结

这一章从一步简单的工艺开始介绍Silvaco TCAD工艺仿真的流程，然后以这个流程为主轴，对各个部分进行详细讲解。各个单项工艺介绍完后又提供了一个MOS的仿真示例以期了解单项工艺如何整合的同时加深对仿真流程控制的理解。最后介绍了优化工具Optimizer的使用，这是从结果到参数的反向设计的绝佳工具。

第三章 二维器件仿真

3.1 ATLAS概述

ATLAS器件仿真框架可以仿真半导体器件的电学、光学和热学行为。ATLAS提供基于物理的模块化的易用的可扩展的平台，在此基础上分析2维或三维器件的DC、AC和时域响应，光-电，电-光转换等特性。

ATLAS主要模块有:S-Pisces、Devices3D、Blaze2D/3D、TFT2D/3D、VCSELS、Laser、Luminous2D/3D、Ferro、Quantum、Giga2D/3D、NOISE、C-Interpreter、MixedMode和DevEdit2D/3D。具体各模块的特性，可查看器件仿真的手册或是查询官方网站的说明。

ATLAS的特性功能很丰富：

ATLAS可以在DeckBuild交互式的运行环境中运行，允许ATHENA、SSUPREM3工艺仿真器输入，DevEdit器件编辑器输入，和UTMOST III¹ 接口可以对器件参数进行提取和对器件建模等。

全面的物理模型：

- DC, AC small-signal, and full time-dependency
- Drift-diffusion transport models
- Energy balance and Hydrodynamic transport models
- Lattice heating and heatsinks
- Graded and abrupt heterojunctions
- Optoelectronic interactions with general ray tracing
- Amorphous and polycrystalline materials
- General circuit environments
- Stimulated emission and radiation
- Fermi-Dirac and Boltzmann statistics
- Advanced mobility models
- Heavy doping effects
- Full acceptor and donor trap dynamics
- Ohmic, Schottky, and insulating contacts
- SRH, radiative, Auger, and surface recombination
- Impact ionization (local and non-local)
- Floating gates
- Band-to-band and Fowler-Nordheim tunneling
- Hot carrier injection

¹对UTMOST III我现在还只是粗略了解，如果收到很多人反馈，觉得必要讲解的话，我会考虑在本书第4章Silvaco高级特性部分再添加一节。

- Quantum transport models
- Thermionic emission currents

先进的数值方法:

- 准确和强大的离散化技术
- Gummel、Newton和block-Newton非线性迭代策略
- 对线性子问题的有效的求解，直接和迭代
- 强大的初步猜测技术
- 小信号计算技术，包括所有的频率
- 稳定和精确的时间集成

ATLAS基于物理的器件仿真问题可如此定义（这也可作为ATLAS仿真的大的框架）：

- 仿真采用物理结构
- 仿真采用物理模型
- 仿真电学特性、偏置状态

ATLAS可以由用户输入²材料的参数（如迁移率参数、寿命参数、能带参数、介电常数等等）及自定义材料，这给仿真验证提供了很好的途径。而且材料的界面，结等特性也是可以方便地进行编辑的。Silvaco的这些很方便的扩展功能是非常具有吸引力的。

开始运行ATLAS时，注意一下输出窗口中就会发现，DeckBuild的当前可用的组件会在仿真前显示出来。样式如下：

```
=====
ATLAS                : enabled
S-PISCES             : enabled
BLAZE                : enabled
GIGA                 : enabled
LUMINOUS             : enabled
LED                  : enabled
TFT                  : enabled
OTFT                 : enabled
OLED                 : enabled
MIXEDMODE            : enabled
LASER                : enabled
VCSEL                : enabled
FERRO                : enabled
QUANTUM              : enabled
NOISE                : enabled
...                  ...
=====
```

²输入参数可以是直接指定参数大小，也可以由C解释器提供的函数编辑功能来描述。

仿真前有必要了解内部的文件分布信息：

手册的路径：X:\sedatools\lib\Atlas\x.xx.0.R\doc\atlas_user1.pdf。

器件模型参数文件路径：X:\sedatools\lib\Atlas\x.xx.0.R\common\atlasmod

光学参数文件 (*.n, *.k) 路径：X:\sedatools\lib\Atlas\x.xx.0.R\common

C解释器的模板、数学符号等文件路径，X:\sedatools\lib\Atlas\x.xx.0.R\common\SCI

ATLAS语法和ATHENA类似，是由一系列状态及其参数组成的（DeckBuild语法是由命令和参数组成的），语法结构如下：

<STATEMENT> <PARAMETER>=<VALUE>

参数的VALUE有四种类型“Character”、“Integer”、“Logical”和“Real”。每一种类型参数的描述如下：

Parameter	Description	Value Required	Example
Character	Any character string	Yes	material=silicon
Integer	Any whole number	Yes	region=1
Logical	A true or false condition	No	gaussian
Real	Any real number	Yes	x.min=0.1

语法描述的一些例子：

doping uniform n.type concentration=1e16 region=1

electrode name=cathode bottom

interface x.min=-4 x.max=4 y.min=-0.5 y.max=4 qf=1e10 s.n=1e4 s.p=1e4

models k.p fermi incomplete consrh auger optr print

method gummel newton trap itlimit=20 maxtrap=6

material material=AlGaAs mun=2170 mup=350 tmun=1 tmup=2.1

output val.band con.band e.velocity

solve vbase=0.5

由于statement非常的多，本书也不可能做到面面俱到，如果遇到哪里不懂的一定要及时查询手册。而且我也力图避免将此书写成手册的中文版。

学习的另外一个很好的资源就是程序自带的例子。Silvaco的ATLAS部分的示例相当丰富，是按照器件进行分类的，这些示例库有bjt,ccd,diode,eprom,esd,ferro,hbt,hemt,isolation,laser,latchup,led,magnetic,mercury,mesfet,mos1,mos2,noise,optoelectronics,organic,quantum,power,seu,soi,tft,thermal和vcsel。对于这些器件又有不同的特性获取的示例。内容很多，我们只需要对自己关心的那一些深入下去就行了。

这一章主要提供仿真流程控制方面的学习经验，还有对一些常用器件(如diode,BJT,MOS,MESFET,HBT,HEMT等)的一些特性获取进行详细讲解，其他则只会提醒需要注意的地方。手册有1062页，我想再怎么样我也无法写到手册的细致，不足之处，还请见谅。

3.2 仿真流程




图 3-1: ATLAS命令组及其主要的状态

如图3-1所示，ATLAS的仿真是通过对一系列Statement的描述来进行组织的，而这些状态又可以分成一些组，大体是结构生成、设定材料模型、计算方法、器件特性获取和结果分析等五组状态。

这一章的思路就是按照这个流程来展开的。希望能起到既见树木又见森林的效果。

从下一节开始就分别从这些状态组及其状态来讲解ATLAS仿真。

3.3 定义结构

ATLAS在器件仿真时的结构可以直接导入现成的结构，可以由ATHENA生成，也可以由ATLAS命令来生成，还可以由DevEdit生成。第二章讲到了直接导入结构文件和由工艺仿真得到结构文件。

直接导入现成的结构文件：

```
go atlas
init infile=my.str
```

由工艺仿真得到结构文件：

```
go athena
:
structure outfile=process.str
```


3.3.1 命令语句生成结构

ATLAS生成结构也是从网格定义开始，主要的步骤有四个：1.初始化网格，2.定义区域和材料，3.定义电极，4.描述掺杂。

1、Initial Mesh

记得工艺仿真时的网格建立是用状态line，和参数location，spacing。和ATHENA的定义方式相近，ATLAS是由状态x.mesh，y.mesh及其对应的参数location和spacing。

space.mult定义由x.mesh和y.mesh定义的网格线的比例因子，默认值是1。大于1会使网格变粗糙，但仿真也更快。

```
mesh space.mult=<value>
```

```
x.mesh location=0.0 spacing=0.05
```

```
y.mesh loc=0.0 spac=0.05
```

同ATHENA的情形一样，如果各个location处的spacing一样的话，网格就是均匀的，如果spacing不一样的话，则网格就不均匀。

网格要考虑疏密分布。材料界面、很浅的材料层都需要定义得密一些。我认为密的原则就是在某个location周围参数变化显著，分段线性化时应该考虑将分的段的长度小一些。这样才能得到更精确的结果。ATLAS二维仿真的网格点数不得超过20,000个。

有的地方可以定义得稀疏一些，如衬底。命令eliminate可以在已有的网格基础上删除掉指定矩形区域一些网格线，删除的方式是这个矩形区域内纵向网格线隔一条删除一条。

eliminate的语句示例，如图3-2中(a)和(b)分别为使用前后的效果。

```
eliminate columns x.min=0.2 x.max=1.4 y.min=0.2 y.max=0.7
```




图 3-2: eliminate使纵向的网格线变稀疏的效果

如果定义三维器件则需有z.mesh参数。

2、Region and Materials

ATHENA的网格定义后，初始化即得到衬底，在衬底的基础上在经过一系列的工艺生成结构。ATLAS的网格定义是对整个器件的二维剖面（二维器件仿真）进行描述，且剖面一定是矩形的。

ATLAS的器件剖面可以按需要分成不同的区域(region)及对应的材料。定义区域需要指明位置（X和Y的范围），材料名称和序号。在后续定义掺杂时就可以按区域定义。

region的语法：

region number=<n> <material> [<position>]

区域定义示例：

```
region num=1 y.max=0.5 silicon
region num=2 y.min=0.5 y.max=1.0 x.min=0 x.max=1.0 oxide
region num=3 y.min=1.0 y.max=2.0 x.min=0 x.max=1.0 GaAs
```

3、Electrodes

electrode定义电极的名称和位置，位置可以是x.min,x.max,y.min和y.max组成的矩形框，也可以是按照剖面的特定位置如top,bottom,left,right和substrate，还可以是指定一点和电极的X方向的长度来定义。

electrode的语法：

electrode name=<en> [number=<n>][substrate]<position> <region>

```
elec name=emitter x.min=1.75 x.max=2.0 y.min=-0.05 y.max=0.05
```

```
elec name=gate x.min=0.25 length=0.5
```

```
elec substrate
```

```
elec num=1 name=source y.min=0 left length=0.25
```

```
elec num=2 name=drain y.min=0 right length=0.25
```

```
elec name=anode top
```

```
elec name=cathode bottom
```

4、Doping

杂质分布由doping状态设置，doping语法格式如下：

doping <distribution_type> <dopant_type> <position_parameters>

杂质分布类型可以是均匀(uniform)，高斯(gaussian)和误差函数(error function)分布。

均匀分布常用的参数有杂质类型(n.type, p.type)浓度(concentration)和区域。

```
doping uniform conc=1e16 n.type region=1
```

高斯分布常用的参数可以分成类型和分布参数两类。类型即n.type和p.type。分布参数又可以分成三组，三组参数选其中一组。第一组的参数有concentration和junction，第二组参数有dose和characteristic，第三组参数有concentration和characteristic。

```
doping region=1 gaussian conc=1e18 peak=0.1 characteristic=0.05 p.type \
  x.left=0.0 x.right=1.0
doping region=1 gauss conc=1e18 peak=0.2 junct=0.15
```

误差函数分布和高斯分布的参数定义是一致的。

```
doping erfc n.type peak=0.5 junction=1.0 conc=1e19 x.min=0.25 x.max=0.75 \
  ratio.lat=0.3 erfc.lat
doping p.type conc=1e18 uniform
```

上面的语句使用误差函数分布，峰值浓度 $1e19$ ，峰值位置在 $0.5\mu m$ 。char决定浓度随深度的变化速率，因为定义了结深是 $1\mu m$ ，所以 $1\mu m$ 处施主浓度也必须达到 $1e18$ （因为稍后定义有所有区域均为p型均匀掺杂 $1e18$ ），用下面的方程就可以计算出char为 $0.43\mu m$ 。

$$erfc([junction - peak]/char) = 0.1$$

也可以从文件导入掺杂信息。下面的例句为杂质分布信息由文件“concddata”导入，文件为ASCII编码。文件“concddata”内部信息由两列数字表示，左边的列表示深度(μm)，右边的列表示掺杂浓度(cm^{-3})。

```
doping x.min=0.0 x.max=1.0 y.min=0.0 y.max=1.0 n.type ascii infile=concddata
```

经过这几步就可以定义一个器件结构了，将结果保存成结构文件(*.str)即可。

下面是一个完整的Diode结构定义的例子，结构如图3-3所示。

```
go atlas
mesh
x.mesh loc=0.00 spac=0.05
x.mesh loc=0.10 spac=0.05
y.mesh loc=0.00 spac=0.20
y.mesh loc=1.00 spac=0.01
y.mesh loc=2.00 spac=0.20
#
region number=1 x.min=0.0 x.max=0.1 y.min=0.0 y.max=1.0 material=silicon
region number=2 x.min=0.0 x.max=0.1 y.min=1.0 y.max=2.0 material=silicon
#
electrode name=anode top
electrode name=cathode bottom
#
doping uniform conc=1e18 n.type region=1
doping uniform conc=1e18 p.type region=2
#
save outfile=diode_0.str
tonyplot diode_0.str
quit
```




图 3-3: ATLAS命令生成的器件结构




图 3-4: DevEdit生成的器件结构

3.3.2 DevEdit编辑得到结构

图3-4是由器件编辑器DevEdit生成的二极管结构。对比图3-3和图3-4就会发现器件编辑器得到的区域边界可以多样化。器件编辑器的区域是由一些“point”连起来的，而ATLAS命令中的区域是定义X和Y边界，所以只能是矩形。这样使DevEdit在结构的定义上就有很大的灵活性。

在器件编辑之前可以先定义工作面积，由x1,x2,y1,y2来定义。

DevEdit的结构是分区域(region)定义的，在不同区域内再指定材料和掺杂。“杂质”是以“impurity”定义的。

region语法：

```
REGION {DELETE {ID=<n> | ID=<c> | NAME=<STRING>} | \
{{ID=<n>}[NAME=<c>]} | ID=<c>} [MATERIAL=<c>] \
[COLOR=<n>] [PATTERN=<n>] POINTS=<point2d.list> \
[WORK.FUNCTION=<n>] [ELECTRODE.ID=<n>] [Z1=<n>] [Z2=<n>]}
```

电极也是由region定义的，包括电极的位置、功函数等。

impurity的参数非常多，经常用到的参数是region.id,id,impurity,reference.value,x1,x2,y1,y2,peak.value,resistivity等。

对这些参数做简要的说明。region.id定义impurity所属的区域。id是此区域的掺杂序号。impurity定义采用的杂质。x1,x2,y1,y2是杂质边界位置(μm)（如果是编辑三维器件除了之前的“point”要有三个坐标之外，impurity也需要z1和z2指明Z方向的范围）。杂质的分布由peak.value,reference.value或由resistivity指定电阻率。

区域和掺杂定义好后，就需对网格进行定义。网格分布由base.mesh,constr.mesh来定义。网格的参数有最大、最小的高度、宽度和角度等，定义完后由mesh生成网格。

关于器件编辑的详细设置可以参考DevEdit的手册。

状态和参数非常多，下面给出一个二极管编辑的例子，结果如图3-4所示。

```

go devedit
#
region reg=1 mat=silicon color=0xffb2 pattern=0x9 \
  points="0,0 0.1,0 0.1,1 0,1 0,0"
impurity id=1 region.id=1 imp=Donors \
  x1=0 x2=0.1 y1=0 y2=1 \
  peak.value=1e+18 ref.value=1000000000000 comb.func=Multiply \
  rolloff.y=both conc.func.y=Constant \
  rolloff.x=both conc.func.x=Constant
region reg=2 mat=silicon color=0xffb2 pattern=0x9 \
  points="-0.01,1 0.11,1 0.11,2 -0.01,2 -0.01,1"
impurity id=1 region.id=2 imp=acceptors \
  x1=-0.01 x2=0.11 y1=1 y2=2 \
  peak.value=1e+18 ref.value=1000000000000 comb.func=Multiply \
  rolloff.y=both conc.func.y=Constant \
  rolloff.x=both conc.func.x=Constant
#electrode
region reg=3 name=anode mat=contact elec.id=1 work.func=0 \
  points="0,0 0.1,0 0.1,-0.01 0,-0.01 0,0"
region reg=4 name=cathode mat=contact elec.id=2 work.func=0 \
  points="-0.01,2 0.11,2 0.11,2.01 -0.01,2.01 -0.01,2"
# Set Meshing Parameters
base.mesh height=0.1 width=0.125
#
bound.cond max.slope=28 max.ratio=300 rnd.unit=0.001 line.straightening=1 \
  align.points when=automatic
#
constr.mesh mat.type=semiconductor max.angle=180 max.ratio=200 \
  max.height=1 max.width=1 min.height=0.001 min.width=0.001
#
constr.mesh x1=0 x2=0.1 y1=0.0 y2=1 max.height=0.08 min.width=0.01
constr.mesh x1=-0.01 x2=0.11 y1=1 y2=2 max.height=0.1 min.width=0.01
mesh

structure outf=diode_1.str
tonyplot diode_1.str

```

3.4 材料参数及模型

网格，几何结构，掺杂分布等定义好后，就可以定义器件仿真时的电极参数、材料特性和物理模型等。以下分别介绍接触特性，材料特性，界面特性和物理模型。

3.4.1 接触特性

接触的状态由contact定义，参数有功函数参数，边界情形，寄生参数，电极连接参数，浮栅电容参数等等。参数很多，下面从一些经常用到的接触类型的示例来看contact的语法。

1、Gate的功函数和肖特基接触

电极和半导体材料的接触默认是欧姆接触，如果定义了功函数参数则是肖特基接触。

```
contact name=gate workfunction=4.8
```

功函数可以在接触材料是aluminum,n.polysilicon,p.polysilicon,tungsten和tu.disilicide时定义。

```
contact name=gate n.polysilicon
```

```
contact name=drain aluminum
```

铝和重掺杂硅的接触通常是欧姆接触，这种情况就不要指定功函数。

当指定功函数时还可以用barrier和alpha参数设置接触的势垒。

```
contact name=anode workfunction=4.9 barrier alpha=1e-7
```

2、设置电流边界状态

contact定义的电极可以是电压控制，也可定义成电流控制（这时计算方法solution需要设置成block或newton）。电流的初始值是0A。

```
contact name=drain current
```

3、定义外电阻、电容或电导

下面的例句为定义接触时外电阻、电容和电导的，单位分别是 Ω ，F和H。

```
contact name=source resistance=50.0 capacitance=20e-12 inductance=1e-6
```

分布电阻可以用con.resist定义，下面的例句为定义分布电阻 $0.01\Omega \cdot cm^2$

```
contact name=source con.resistance=0.01
```

4、浮动接触

contact的参数floating可定义EEPROM中的浮栅电极，也可以在功率器件仿真中定义浮场极板(floating field plates)。

```
contact name=fgate floating
```

浮动接触也可以以很高的外接触电阻来替代。这在击穿仿真时很有用，如果耐压为1V而外电阻为 $100M\Omega \cdot \mu m$ ，则该接触的最大电流就是 $10^{-8}A/\mu m$ 。

```
contact name=drain resist=1e20
```

5、电极间短接

contact可以定义电极间的短接，短接电极的电压将是一样的。

回忆一下之前介绍用ATLAS命令生成结构，里面有个electrode状态，其中就有name参数。相信大家在这里不难挖掘出电极短接的方法，就是将两个或多个electrode的name设置成一样，而使用不同的位置参数。

同样的，在由ATHENA进行工艺仿真后在不同金属区的位置上设置的electrode的name一样，当仿真器切换到ATLAS时，名称相同的电极也是短接的。

上面提到的两种方法都是要电极的名称一致，而用contact状态就可以在电极名称不一致时实现短接。

```
contact name=base common=collector
:
solve vbase=0.5
```

上面的例句是将基极和集电极短接，在计算电学特性时这两个电极将是等电位的（solve将base电压加到0.5V时collector也是0.5V电压）。这在Gommel Plot仿真中很有用。

common参数除了可以得到电极短接的效果，也可以将电极联系起来而电极间有一恒定的电压差。

下面的例句中gate1的电压始终等于gate2电极的电压加上0.1V。

```
contact name=gate1 common=gate2 factor=0.1
```

6、开路接触

有三种方法可是实现电极的开路。第一种方法是在生成结构时将electrode删掉，也就是没有电极，那自然是开路的了。第二种方法是在contact定义中采用很大的接触电阻（如 $10^{20}\Omega$ ），接触电阻相当于上拉电阻的作用。第三种方法是contact定义中使用电流控制，然后将电流大小设置得很小很小或是0。

3.4.2 材料特性

所有的材料都被分为半导体、绝缘体和导体三大类。每一类都有特定的参数，如半导体有电子亲和势、能带间隙、状态密度、少子寿命和饱和速度等。器件仿真时很多材料都有默认的参数，附录A中列出了材料所属的类型。

材料的特性是用material状态设置的。

material参数可以分成几大类。区域参数，能带结构参数，BQP参数，迁移率模型参数，复合模型参数，碰撞离化参数，Klaassen模型参数，载流子统计模型参数，能量平衡参数，热载流子注入参数，晶格温度相关参数，氧化材料参数，光生成参数，激光器参数，噪声参数，有机传输参数，激子参数，混杂参数，导体材料和缺陷生成参数等等。

这些参数都对应一定的物理模型，由一系列方程来表示这些量。模型方程及变量的默认参数在ATLAS手册的physics部分有详细说明。表3-1是常用的材料参数及简单说明。

表 3-1: 材料特性的常用参数及其说明

区域材料参数	简单说明
mateiral	材料名称
name	在结构生成时特定区域的名称
region	在结构生成时特定区域的序号
能带结构参数	简单说明
affinity	电子亲和势(eV)
align	不同禁带宽度材料接触时导带不连续参数（后将给出详细说明）
d.tunnel	肖特基隧穿模型中定义最大隧穿距离
eg300	300K时的禁带宽度(eV)
nc300	300K时的导带密度
nv300	300K时的价带密度
ni.min	本征载流子允许的最小值
permittivity	材料的电学介电常数
迁移率模型参数	简单说明
mun	低电场时电子迁移率($cm^2/(V.s)$)（需指定迁移率浓度依赖模型）
mup	低电场时空穴迁移率($cm^2/(V.s)$)（需指定迁移率浓度依赖模型）
vsatn	电子饱和速度(cm/s)
vsatp	空穴饱和速度(cm/s)
复合模型参数	简单说明
augn	电子俄歇系数(cm^6/s)
augp	空穴俄歇系数(cm^6/s)
copt	材料的光学复合速率(cm^3/s)（设定模型时需使用model optr）
etrap	SRH复合时的陷阱能量(eV)
碰撞离化参数	简单说明
lambdae	电子平均自由程
lambdah	空穴平均自由程
opphe	光学声子能量(eV)
Klaassen模型参数	简单说明
taun0	SRH复合的电子寿命(s)
taup0	SRH复合的空穴寿命(s)
载流子统计模型参数	简单说明
eab	受主能级(eV)
edb	施主能级(eV)
热载流子注入参数	简单说明
ig.elinr	电子平均自由轨道(cm)（两次碰撞之间）
ig.hlinr	空穴平均自由轨道(cm)
ig.elinf	电子平均自由程长度(cm)
ig.hlinf	空穴平均自由程长度(cm)
导体参数	简单说明
drhodt	电阻率温度系数($\mu W.cm/K$)
resistivity	电阻率($\mu W.cm$)

晶格温度相关参数	简单说明
egalpha	禁带宽度随温度变化的 α 参数(eV/K)
egbeta	禁带宽度随温度变化的 β 参数(K)
lt.taun	电子寿命受晶格温度的影响指数因子(无量纲)
lt.taup	空穴寿命受晶格温度的影响指数因子(无量纲)
power	热学仿真时thermal3D某区域热源的热生成功率(W)
tc.a,tc.b,和tc.c	热导系数($cm.K/W$)
光生成参数	简单说明
j.elect	电流密度(A/cm^2)
j.magnet	磁流密度(V/cm^2)
imag.index	半导体的复光学折射率的虚部
index.file	从文件导入不同波长时的复折射率
permeability	磁导率
real.index	半导体的复光学折射率的实部
激光器参数	简单说明
alphaa	体吸收系数(cm^{-1})
alphar	线宽扩展因子(无单位)
混杂参数	简单说明
alattice	面晶格常数(\AA)
degeneracy	自旋退化因子(无量纲)
delta1,delta2和delta3	价带能量劈裂参数(eV)
dindexdt	折射率的温度系数(1/K)
drhodt	导体电阻率的温度系数($\mu\Omega.cm/K$)
mhh	重空穴有效质量(无量纲)
mlh	轻空穴有效质量(无量纲)
mc	导带有效质量(无量纲)
mv	价带有效质量(无量纲)
resistivity	导体电阻率($\mu\Omega.cm$)

材料的有些参数可以用C解释器写成函数表达式，再以文件的方式导入ATLAS进行仿真。程序提供了一些C解释器的一些模板，还有数学表达的缩写。C解释器在下一章Silvaco的高级特性部分进行介绍。如材料的光学折射率参数可用index.file（同f.index）参数导入文件。文件内包含了波长及材料在该波长下折射率信息。文件格式如下：

```

<n>
l(1) n(1) k(1)
l(2) n(2) k(2)
:      :      :
l(n) n(n) k(n)

```

<n>表示不同波长的光波总数目，l(n)表示第n个光波的波长(μm)，n(n)和k(n)分别对应材料在该波长时的折射率实部和虚部。




图 3-5: 材料能带结构参数示意图

图3-5为能带结构及其参数。材料1和材料2是两种不同的半导体，禁带宽度分别为 E_{g1} 和 E_{g2} ，电子亲和势分别为 χ_1 和 χ_2 。它们可以制作异质结器件。材料2和金属形成非整流接触。 ϕ_b 为势垒高度， ϕ_m 为金属功函数。其中 $E_{g1} < E_{g2}$ ， $\chi_1 < \chi_2$ 。下面的四个方程显示了这些能带结构参数之间的关系。

$$\Delta E_c = \chi_1 - \chi_2 \quad (3-1)$$

$$\Delta E_v = \Delta E_g - \Delta E_c \quad (3-2)$$

$$\Delta E_c = (\Delta E_{g2} - \Delta E_{g1}) \cdot align \quad (3-3)$$

$$\phi_b = \phi_m - \chi_s \quad (3-4)$$

表3-1介绍能带结构参数时提到align参数。align是异质结器件的重要参数，表征导带的不连续性³。从方程(3-3)可知align的含义。

材料参数设置例句：

```
material material=silicon eg300=1.12 mun=1100
```

```
material region=1 tauno=1e-6 taup0=1e-6
```

```
material name=base nc300=2e19
```

```
material material=AlGaAs align=0.6
```

```
material material=InGaAs align=0.6 mun0=12000 mup0=2000 vsat=2.e7 \
  taun0=1.e-8 taup0=1.e-8
```

函数文件描写材料参数：

```
material name=silicon f.index=myindex.c
```

³准确地表征异质结的不连续，除了之前的这些能带参数定义准确外，还有能带不连续处的 网格间距 也非常关键。如果间距太大，本来是突变的能带被拉直了，不止是没有二维电子气存在，后面所有的仿真结果都将是不可信的。

3.4.3 界面特性

interface状态用于定义界面电荷密度和表面复合速度。界面的类型默认是半导体-绝缘体界面，也可以是半导体和半导体之间的区域或半导体的边界区域。界面的主要参数是df、面电荷密度(cm^{-2})、位置参数x.min,x.max,y.min和y.max、复合速率S.N(电子表面复合速度)和S.P(空穴表面复合速度)，参数S.S,S.M,S.C可以是界面模型应用在半导体-半导体界面、半导体-金属界面以及半导体-导体界面。

界面定义的例句：

```
interface y.max=0.1 qf=-1e11
```

```
interface x.min=-4 x.max=4 y.min=-0.25 y.max=0.1 qf=1e11 \
s.n=1e4 s.p=1e4
```

3.4.4 物理模型

物理模型由状态models和impact指定。这些物理模型可以分成五组：迁移率模型，复合模型，载流子统计模型，碰撞离化模型和隧道模型。

模型设定的例句：

```
models conmob fldmob srh fermidirac
```

```
impact selb
```

针对特定的技术,有简便的方法配置相应模型。这些技术是MOS,BIPOLAR,PROGRAM和ERASE。models将会据此配置一些基本的迁移率、复合、载流子统计和隧道模型。模型参数为MOS时配置的基本模型有cvt,srh和fermidirac，为BIPOLAR时配置的基本模型有conmob,fldmob,consrh,auger和bgn。下面的例句中有print参数，所以在DeckBuild输出窗口将会列出此仿真语模型及其参数。

```
models mos print
```

```
models bipolar print
```

物理模型的指定也可以是对特定材料，这样在异质结器件仿真或其他的多半导体类型器件的仿真参数设置提供了很大灵活性。

```
models material=gaas fldmob evsatmod=1 ecritn=6e3 conmob
models mateiral=ingaas srh fldmob evsatmod=1 ecritn=3e3
```

上面例句中出现的模型参数简单地说明如下：conmob，迁移率受杂质浓度影响的模型；fldmob，迁移率受电场影响的模型；srh，Shockley-Read-Hall复合模型；fermidirac，费米-狄拉克统计模型；cvt，迁移率模型，包含N,T,E//和E⊥的影响；auger，俄歇传输模型；bgn，能带变窄模型；selb，参数受温度影响的模型。

物理模型的内容相当丰富，手册第三章Physics部分有详细的说明⁴。

⁴附录B中有简要的说明（本来希望详细地介绍，无奈内容实在太多，还易造成全书结构的不协调。）

3.5 数值计算方法

ATLAS仿真半导体器件是基于1到6个耦合(coupled)的非线性(non-linear)的,偏微分方程(partial differential equations)。ATLAS将在器件结构的网格点处对这些方程采用数值计算来获取器件的特性。

对非线性的代数系统采用迭代的方式求解,直到解满足要求或确认不收敛。非线性迭代始于初始猜测值,然后将问题线性化,再对线性子问题直接求解或迭代求解。

不同的解决步骤会导致收敛性、精确性、效率和坚固性上的差异。收敛性主要体现在是否收敛和收敛的快慢。精确性指计算得到的解答和真实结果的近似程度。效率主要体现在计算所耗费的时间多少。坚固性指应用在宽范围结构,或网格和初始猜测策略非最优时的收敛性上的能力。

离散化技术将不同的网格点产生一定的联系。离散化应该包含两重含义,一是通过建立网格使连续的结构离散化,二是数值计算上的算法使物理量离散化。离散化技术对仿真很重要,尤其是网格定义应是重中之重,初学者往往体会不到这一点。

非线性计算方法由method状态以及迭代和收敛准则相关的参数进行设定。参数是newton,gummel,block,也可以是这些参数的组合。

1、Newton迭代法

Newton迭代法的每一次迭代将非线性的问题线性化处理。离散化的“尺寸”较大,则所需的时间也会变长。如果初始猜测很成功的话,就能很快得到收敛的且比较满意的结果。

Newton迭代法是ATLAS漂移-扩散计算的默认方法。还有一些其他的计算需要采用Newton迭代法,它们是:含有集总元件时的DC计算,瞬态计算,curve tracing,频域的小信号分析。

Newton-Richardson方法是Newton迭代法的变体,当收敛放慢时它会计算新的系数矩阵。method的参数设置为autonr时会自动采用Newton-Richardson法。

如果经过很多步才能收敛,问题可能来自于:网格定义(网格中高宽比(或宽高比)很大的三角形太多),耗尽区扩展到已定义为欧姆接触的地方,或是初始猜测值很差。

2、Gummel迭代法

Gummel迭代法的每一步迭代都需解一系列相关的子问题(subproblems)。通过对一个方程的主要变量离散化来得到子问题,此时其他的变量保持在当前计算得到的值不变。Gummel迭代法收敛得要慢,但能容忍粗糙的初始猜测值。

Gummel迭代法不能用于含有集总元件或电流边界情形的求解。

有两种办法可以改善Gummel迭代计算。一、由于默认的Gummel迭代是阻尼的(步长在减小),可将参数dvlimit设置成负值或零让迭代成为非阻尼的。二、由于每一步Gummel迭代法其线性Poisson求解的数目限制为1,这会导致势(potential)更新时驰豫不足。“single-Poisson”求解模式可扩展Gummel迭代方法应用范围(更高的电流的情形),

这在低电流Bipolar仿真和MOS饱和区的仿真上很有用。实现的方法是method的参数设置为singlepoisson。

3、Block迭代法

在含有晶格加热或能量平衡方程时block迭代法很有用。Block迭代法计算一些子方程组(subgroup of equations)，子方程组由一些不同方程按不同的顺序组成。

在不等温的漂移-扩散仿真时指定block迭代法，则Newton迭代法将更新(update)势和浓度，去耦之后计算热流方程。

热流方程和载流子温度方程都包含时，Block法将先计算最初的温度，然后晶格温度去耦之后再迭代进行计算。

4、组合迭代法

newton迭代法,gummel迭代法和block迭代法可以单独使用来计算，有时也需要将这些方法联合起来使用。可以先用gummel迭代法，一定计算步数还不收敛再转为采用block迭代法或newton迭代法计算。gummel迭代法的次数由参数gum.init设定。

在包含晶格加热或能量平衡计算时可以先采用block迭代法然后newton迭代法的方式，block迭代法的次数上限用nblockit参数设置。

5、非线性迭代的收敛准则

当计算后的结果（主要是电势，浓度，晶格温度和载流子温度）在可容忍的范围内时非线性迭代就终止，即结果收敛。

数值计算方法method的例句如下。

基本的漂移-扩散计算：

```
method gummel block newton
```

晶格加热时的漂移-扩散计算：

```
method block newton
```

能量平衡计算：

```
method block newton
```

计算的载流子种类数。默认是2，也可以是1或0。载流子类型elec和hole分别表示电子和空穴。如果载流子类型设置为0时，将主要得到电势分布的仿真结果。

```
method carriers=2
```

```
method carriers=1 elec
```

```
method carriers=0
```

TRAP参数应用。如果计算开始时有分歧，则电极的偏置步长将从最初近似开始按照倍增因子atrap减少。ATLAS仿真时经常会碰到不收敛的报错信息“max trap more than 4”，可用maxtrap参数使步数加大，从而能够收敛。

```
method newton trap maxtrap=10
```

计算方法在手册的第19章“Numerical Techniques”有详细的介绍。

3.6 获取器件特性

实际器件的特性获得都要通过仪器进行测试得到。测试结果一般都是端电流电压特性。可改变电信号（直流、交流、瞬态以及特性波形等等）、环境的温度、光照、压力或磁场等得到端电流电压随这些量的变化。

ATLAS进行器件仿真时也可以按照这些方式进行仿真，除了能得到端的电学特性外，还能得到器件内部的信息（浓度分布，电势分布，电流密度...），这是实际的测试仪器难以做到的。UTMOST III可以直接导入ATLAS仿真的结果（也可以是实际仪器测试的结果，如KEITHLEY），从而提取器件对应的Spice模型的参数。Windows版本没有UTMOST III，所以本书也没有考虑写这部分的内容。

在仿真开始时所有的电极都是零偏的，之后才会按照设置的方式将电流或电压步进式地加上去。步进的步长是需要考虑的，步长太大就容易不收敛（由于计算方法中的初始猜测策略）。

电压和电流的施加使用solve状态，log和save是将计算得到结果分别保存为日志文件和结构文件。

不同的器件类型关心不同的特性。本书主要关心电流控制型器件、电压控制型器件、光电器件和霍尔器件等几类。从下面起分四小节介绍直流、交流、瞬态和一些高级特性的获取，至于这些获取方式可以用于获取哪些器件的哪些特性会有提示。

3.6.1 直流特性

从例句学习器件特性的获取。

所有电极的电压加为0V：

```
solve init
```

经过solve之后保存的结构文件中将包含有电学信息（电势，电流密度，电极的电流电压等）。直接从某一电压开始计算，则上面的语句solve init将自动加入。

基极电压加到0.1V：

```
solve vbase=0.1
```

按一定步长进行扫描（如果vfinal不是整数个步长后的值，则会自动调整）：

```
solve vdrain=0.0 vstep=0.05 vfinal=5.0 name=drain
```

将之前计算得到的结果作为计算的初始近似：

```
solve previous
```

基极电压从0.1V加到0.2V再经过一系列步骤加到2.0V（可以得到BE结的I-V特性）：

```
solve vbase=0.1
solve vbase=0.2
:
solve vbase=2.0
```

漏电压加到0.1V然后在此基础上将栅压从0.1V一直加到3.0V（可得转移特性，从保存的日志文件中可抽取出跨导随栅压的特性曲线，抽取方法在1.3.5节已进行过详细介绍。

```
solve vdrain=0.1
solve vgate=0.1 vstep=0.1 vfinal=3.0 name=gate
```

Gummel Plot特性仿真

```
solve vcollector=2
solve vbase=0.0 vstep=0.1 vfinal=2 name=base
```

GP图也可以按照之前讲接触定义时的方法，将基极和集电极短接，再扫描电压。

```
contact name=base common=collector
solve vbase=0.0 vstep=0.1 vfinal=2 name=base
```

BJT的CE结击穿特性的仿真须将基极开路，开路接触在介绍接触时也提到了。可以将基极定义成电流控制电极，再将电流设置成很小的值就行了。

```
impact selb
method trap climit=1e-4 maxtrap=10
#
solve init
#
solve vbase=0.025
solve vbase=0.05
solve vbase=0.2
#
contact name=base current
solve ibase=3.e-15
#
log outfile=gp.log master
#
solve vcollector=0.2
solve vstep=0.2 vfinal=5 name=collector
solve vstep=0.5 vfinal=10 name=collector compl=5.e-11 e.comp=3
```

注意上面的例句中有log状态，log是将计算得到的端电压电流保存在日志文件(log file)中。log状态中master参数表示AC数据和I-V数据都将保存在内，master参数的默认值是ture。上例solve中参数comp为限流致5e-11A，参数e.comp将限流的电极数设置为3。没有solve状态时将结束写日志，也可用下面语句结束写日志。

```
log off
```

电流控制型器件(BJT,HBT)的输出特性仿真，是一个Ib一条曲线的。如果你够细心，你会发现上面仿真击穿特性的语句实际上就是输出特性曲线中Ib=0（3e-15 ≈0）的那一

条。按照这个思路，在解得的每一个 I_b 值后保存一下结构文件（结构文件中里面要有当时完整的电学信息），再在扫描集电极电压时导入相应的结构文件即可得到输出特性。

```
solve init
solve vbase=0.05 vstep=0.05 vfinal=0.8 name=base
contact name=base current
#
solve ibase=2.e-6
save outf=bjt_ib_1.str master
solve ibase=4.e-6
save outf=bjt_ib_2.str master
#
load inf=bjt_ib_1.str master
log outf=bjt_ib_1.log
solve vcollector=0.0 vstep=0.25 vfinal=5.0 name=collector
#
load inf=bjt_ib_2.str master
log outf=bjt_ib_2.log
solve vcollector=0.0 vstep=0.25 vfinal=5.0 name=collector
```

上面语句中bjt_ib_*.str为一定基极电流下保存的结构文件，bjt_ib_*.log为对应的输出特性曲线。save状态中的master参数将计算得到的电学特性保存在结构文件中。

电压控制型器件(MOS,MESFET,HMET)的输出特性仿真和电流控制型器件的方法类似。有时栅电压跨度大，为了收敛需要将method的参数maxtrap设置大一些。

```
solve init
solve vgate=1.1 outf=solve_tmp1
solve vgate=2.2 outf=solve_tmp2
#
load infile=solve_tmp1
log outf=mos_ids_1.log
solve name=drain vdrain=0 vstep=0.3 vfinal=3.3
#
load infile=solve_tmp2
log outf=mos_ids_2.log
solve name=drain vdrain=0 vstep=0.3 vfinal=3.3
```

3.6.2 交流小信号特性

交流仿真的语法和直流仿真的语法很相似，只是添加了频率相关的参数。有两种交流仿真类型，一是频率不变只变直流偏置，一是变频率直流偏置不变。

频率不变，直流偏置在变（能得到特定频率下的CV特性）。

```
solve vbase=0 vstep=0.05 vfinal=2.0 name=base ac freq=1e6
```

变交流频率的例子（能得到两端口的电容随频率变化的特性）。频率从1GHz到11GHz，以1GHz为步长。

```
solve vbase=0.7 ac freq=1e9 fstep=1e9 nstep=10
```

在初始频率的基础上按倍数增加，下面的例子中频率是从1MHz开始增加，每一次是原来的两倍，总共增加10次。这样最后为 $2^{10} \times 1\text{MHz} = 1.024\text{GHz}$ 。

```
solve vbase=0.7 ac freq=1e6 fstep=2 mult.f nstep=10
```

```
solve prev ac freq=1e6 fstep=2 mult.f nstep=10
```

也可直流偏置和交流频率一起改变，这会在每一个直流偏置点都对频率进行扫描。

```
solve vgate=0 vstep=0.05 vfinal=1 name=gate ac freq=1e6 \
fstep=2 mult.f nsteps=10
```

3.6.3 瞬态特性

瞬态仿真用于时间相关的测试或响应。瞬态仿真可以由逐段线性方式，指数函数方式和正弦函数方式获得。

```
solve vgate=1.0 ramptime=1e-8 tstop=1e-7 tstep=1e-10
```

上面例子的含义：电压从当前值在10ns内线性增加到1.0V，步长为0.1ns。

3.6.4 高级特性

1、Curcetrace

Curvetrace可以设置复杂的扫描方式，自动得到I-V特性。curvetrace和solve联合使用可用于击穿电压仿真、CMOS栓锁仿真和二次击穿仿真。

```
curvetrace contr.name=cathode step.init=0.5 nextst.ratio=1.2 \
mincur=1e-12 end.val=1e-3 curr.cont
solve curvetrace
```

上例中contr.name定义电极名称，step.init为开始的电压步长，当电流值超过mincur时电压按nextst.ratio增加，最后(curr.cont指定为电流控制)达到或超过end.val时停止扫描。

可参照工艺仿真时的machine参数来体会Curvetrace。

2、S参数仿真

S参数仿真是基于交流分析的，只是log状态时需设定参数s.param,inport和outport。Z方向的宽度width(μm)。有两个输入端时用in2port表示第二个输入端（同理，有out2port），rin表示输入电阻(Ω)。

```
log outf=ac.log s.param inport=gate outport=drain width=100
solve ac.analysis direct frequency=1.e9 fstep=2.e9 nsteps=20
```

```
log outf=ac.log s.param inport=gate outport=drain \
  in2port=source out2port=source width=100 rin=100
solve ac.analysis direct frequency=1.e9 fstep=2.e9 nsteps=20
```

可以参照上面例句得到RF分析的其他参数设置。如扫描方式:h.param,y.param,z.param,abcd.param,gain; 噪声参数: noise; 寄生元件参数: rout,rground,lin,lout(Henrys)等。

S参数仿真可以得到增益随频率变化的特性, 需指定参数gain。

3、霍尔效应仿真

霍尔效应主要是要添加进磁场, 施加磁场在models状态中设置。

```
models srh fldmob bz=1.0
model temperature=300
solve init
#
probe name=hall1 x=0.0 y=0.5 potential
probe name=hall2 x=1.0 y=0.5 potential
probe name=reference x=0.5 y=0.5 potential
#
log outf=hall_effect.log
solve name=anode vanode=0.0 vstep=0.05 vfinal=1.0
```

bz为Z方向的磁场大小(Tesla), 同样可以设置bx和by。probe状态允许输出日志文件中保存特定位置的分立的值, 上例中为输出potential。由hall1,hall2和reference的电势可以得到霍尔电压。不难理解上面的仿真语句能得到 I_x 和 V_{hall} 的曲线。

仿真温度可以在models状态中设置, 默认为300K。如果需改变温度仿真可以在每一个温度点仿真一下得到一个结果, 这样就能间接得到某特性随温度变化的曲线。

4、光电特性仿真

光电(探测器)特性的仿真和霍尔效应仿真相似, 主要就是加光照。对光照的条件(波长和光强等)进行改变。

光束的定义用beam状态。主要参数有方向参数, 波长, 强度分布和光线的几何分布参数, 反射参数等。下面从例句介绍beam的常用设置。

定义光束的方向及(单色光)波长。(x.orign,y.orign)为光出射的点, angle为从X轴正向往Y轴负方向偏转的角度, 默认0°, 90°即表示从器件顶部垂直于表面往下照射。

```
beam num=1 x.orign=5 y.orign=-2 angle=90 wavelenght=.8
```

光束是复合光, 波长范围由开始波长、结束波长以及波长数目定义。

```
beam num=1 x.orign=5 y.orign=-2 angle=90 wavel.start=.5 wavel.end=1.7 \
  wave.num=13
```

考虑光在器件界面的反射（也包括前反射和后反射），以及反射次数和最小光强(min.power)的限制，光强的单位为 W/cm^2 。

```
beam num=2 x.orign=1 y.orign=-1 angle=90 wavelength=1.5 \
back.refl front.refl reflect=5 min.power=0.01
```

光沿点(x.orign,y.orign)往angle方向的分布。rays为射线束数目，默认是1，也就是说单色光且未定义rays参数时，光线只有一条。

```
beam num=3 x.orign=2 y.orign=-0.5 angle=90 wavelength=0.9 \
rays=101 gaussian mean=0 xsigma=0.25
```

对于光电器件，需要考虑导带电子和价带空穴直接复合并发射光子的情形。模型optr即起到此作用。output状态中有参数opt.intens时，结构文件将包含有光强分布的信息。beam状态没有定义光强，光强在solve状态中指定。b1=2表示第一束光的光强为 $2W/cm^2$ ，同理可知b2表示第二束光的光强（注意和磁场的bx，by和bz相区别）。

```
models optr fldmob fermi
method newton carr=2
output con.band val.band opt.intens
solve init
#
solve prev b1=2
```

光谱响应（光电流对波长的响应）需要改变单色光波长。

```
beam num=1 x.orign=2.5 y.orign=-1.0 angle=90.0 wavelength=.4
#
solve init
solve vcathode=0.05 vstep=0.05 vfinal=1.0 name=cathode
#
log outf=spectral_Response.log master
solve prev b1=2 lambda=0.5
solve prev b1=2 lambda=0.6
solve prev b1=2 lambda=0.7
solve prev b1=2 lambda=0.8
solve prev b1=2 lambda=0.9
solve prev b1=2 lambda=1.0
solve prev b1=2 lambda=1.2
```

5、热学特性仿真

介绍霍尔效应时曾提到，通过model状态的参数temperature可设置仿真时的温度。在不同的温度下对器件的特性进行仿真，可得到特性随温度的变化关系。

ATLAS的组件Giga是自加热仿真器，能仿真晶格的自加热效应。

晶格加热效应仿真:

模型中设定lat.temp参数时可仿真晶格加热效应，但至少要定义一个热接触，热接触参数thermcontact。

thermcontact语法:

```
thermcontact number=<n> <position> [ext.temper=<n>] [alpha=<n>]
```

position为热接触的位置，可由x.min,x.max,y.min,y.max,z.min和z.max(μm)来定义位置。如果热接触是已定义的电极，则只需要设定elec.number参数即可，定义electrode的先后顺序决定了number。位置默认为边界(boundary)。

ext.temp为仿真时的环境温度（同models状态的temperature参数）。

alpha($a = 1/R_{TH}$)为热阻的倒数，单位 $W/(cm^2 \cdot K)$ 。

热接触定义的例子:

```
thermcontact num=1 y.min=0.5 ext.temp=300 alpha=1
```

```
thermcontact num=2 elec.num=4 ext.temp=350
```

SOI-MESFET的晶格加热仿真的例子，其中arora为迁移率模型。

```
models arora consrh auger bgn fldmob lat.temp
impact selb
thermcontact number=1 y.min=0.6 ext.temper=300
```

电子和空穴温度仿真

使用hcte模型时ATLAS会计算电子和空穴的温度，hcte为能量平衡模型。

```
models hcte
```

hcte.el和hcte.ho分别表示只计算电子温度和只计算空穴温度。

```
models hcte.el
```

```
models hcte.ho
```

3.7 仿真结果分析

ATLAS的仿真结果可以有多种方式得到，DeckBuild界面实时输出，仿真中保存的日志文件，抽取的器件特性，用Tonyplot显示结果以及它的function功能。

3.7.1 实时输出

下面的内容为仿真进行到solve时，DeckBuild输出窗口输出的实时计算结果以及各个电极的电流。

第一栏project指明使用的仿真初始，这些可以是previous,local或init。第二栏direct指明是求解类型，可以是direct和iterative。第三栏i和j表示迭代计算的当前次数，i表示去耦之后的外循环迭代数，j表示内循环数。m指明计算方法（一个字母含义：G=gummel，

B=block, N=newton, A=newton with autonr, S=coupled Poisson-Schrodinger solution)。xnrom和rhsnorm为方程计算时的误差，虚线下的值是以10为底的对数表示误差大小。有星号的数字表明已经达到误差容限。Va栏列出了接触表面的电压。

proj			psi	n	p	psi	n	p
direct			x	x	x	rhs	rhs	rhs
i	j	m	-5.0*	-5.0*	-5.0*	-26.0*	-17.3*	-17.3*

1	1	G	-1.819			-18.64		
2	1	G	-3.510			-20.94		
3	1	G	-7.82*			-25.24		
4	1	G		-3.398			-13.08	
5	1	G			-9.55*			-17.07
1	1	G	-8.55*			-24.24		
2	3	G		-10.2*			-16.18	
3	1	G			-14.4*			-17.6*
Electrode	Va(V)		Jn(A/um)		Jp(A/um)		Jc(A/um)	Jt(A/um)
=====								
gate	0.000e+000		0.000e+000		0.000e+000		0.000e+000	0.000e+000
source	0.000e+000		-5.617e-013		-1.838e-035		-5.617e-013	-5.617e-013
drain	1.000e-001		5.618e-013		2.136e-023		5.618e-013	5.618e-013
substrate	0.000e+000		-8.335e-019		-8.333e-017		-8.416e-017	-8.416e-017
Total								-7.537e-018

当模型里指定print时，DeckBuild的实时输出窗口将显示当前仿真使用的物理参数，材料参数(能带参数，复合参数，隧穿参数，迁移率参数)，各材料区域使用的模型等。如果相关的参数在ATLAS进行了设置则会替换掉默认值。

用print显示出这些量，一方面提供了一些默认参数的数值，另一方面可以验证和自己设置的是否有差别。

下面是print设置后输出窗口显示的例子，结构定义是从图3-3的结构基础上将区域2的材料改为“sic”得到，所有材料都采用默认设置。

从语句中很容易就能看出各个量代表的含义，竖点(“:”)表示有些具体的参数省去了，希望能从全貌上领会仿真时可以进行哪些设置。

```

constants:
Boltzmann's constant   = 1.38066e-023 J/K
Elementary charge      = 1.60219e-019 C
Permittivity in vacuum = 8.85419e-014 F/cm
Temperature             = 300 K
Thermal voltage        = 0.025852 V

```

3.7 仿真结果分析

REGIONAL MATERIAL PARAMETERS:

Region	:	1	2	3	4
Material	:	Silicon	SiC-6H	Conductor	Conductor
Type	:	semicond.	semicond.	metal	metal
Average Composition Fraction					
X-composition	:	0	0	0	0
Y-composition	:	0	0	0	0
Epsilon	:	11.8	9.66		
Band Parameters					
Eg (eV)	:	1.08	2.9		
Chi (eV)	:	4.17	0		
Nc (per cc)	:	2.8e+019	7.68e+018		
Nv (per cc)	:	1.04e+019	4.76e+018		
ni (per cc)	:	1.45e+010	2.64e-006		
Gc	:	2	2		
Gv	:	4	4		
Ed (eV)	:	0.044	0.044		
Ea (eV)	:	0.045	0.045		
:	:	:	:		
Recombination Parameters					
taun0	:	1e-007	1		
taup0	:	1e-007	1		
:	:	:	:		
Band-to-band tunneling Parameters					
bb.a	:	4e+014	4e+014		
bb.b	:	1.9e+007	1.9e+007		
:	:	:	:		
Saturation Velocities					
vsatn (cm/s)	:	1.03e+007	2e+007		
vsatp (cm/s)	:	1.03e+007	1e+007		
:	:	:	:		

REGIONAL MODEL FLAGS:

Region:	1	2	3	4
SRH	F	F		
consrh	F	F		
klasrh	F	F		
Auger	F	F		
klaaug	F	F		
optr	F	F		
bgn	F	F		

```

incomplete      F  F
bbt              F  F
bbtauto         F  F
bbthurkx        F  F
bbtkane         F  F
tat(local)      F  F
tat(coulombic)  F  F
impact          F  F
Boltzmann       T  T
Fermi-Dirac     F  F
Laser gain model
gain model      0  0  0  0
REGIONAL MOBILITY MODEL SUMMARY:
  Region #1:
    Model for Electrons:
      Concentration Independent Mobility
      Temperature = 300 Kelvin
      mu = 1000
      tmu = 1.5
    Model for Holes:
      :
  Region #2:
    :
  Contacts :
  Name      Num  Work fn  Resist.  Capacit.  Induct.
              (eV)    (Ohms)   (Farads) (Henrys)
  anode      1   0.000   0.000E+00 0.000E+00 0.000E+00
  cathode    2   0.000   0.000E+00 0.000E+00 0.000E+00

```

实时输出窗口可以得到很多仿真的动态信息，仿真的时候一定要随时留意。注意仿真时的报错，可从实时输出结果判断怎样改正仿真。这些方法要认真体会。


3.7.2 日志文件

ATLAS仿真能保存的结果主要有结构文件、log文件和抽取得到的dat文件。

log文件中包含有端的电学信息(电流和电压)，还有在log状态设置时声明的量（如s参数扫描时有gain参数，则log文件将包含有电流增益的数据）。

在Win系统中直接打开log文件，很难看清其数据结构。查看数据有两种好的办法：一、用Tonyplot工具查看；二、用Tonyplot将其导出成csv格式文件。

Tonyplot只能查看，但图形的显示以及数据的获取还是很麻烦，建议导出csv格式文件，效果如图3-6所示。



	A	B	C	D	E	F	G
1	Frequency	Gate Voltage	Gate Current	Source Voltage	Source Current	Drain Voltage	Drain Current
2	1.00E+11	0.00E+00	0.00E+00	0.00E+00	-6.34E-13	5.00E-01	6.32E-13
3	1.00E+11	5.00E-02	0.00E+00	0.00E+00	-2.20E-12	5.00E-01	2.20E-12
4	1.00E+11	1.00E-01	0.00E+00	0.00E+00	-7.73E-12	5.00E-01	7.73E-12
5	1.00E+11	1.50E-01	0.00E+00	0.00E+00	-2.76E-11	5.00E-01	2.76E-11
6	1.00E+11	2.00E-01	0.00E+00	0.00E+00	-9.96E-11	5.00E-01	9.96E-11
7	1.00E+11	2.50E-01	0.00E+00	0.00E+00	-3.64E-10	5.00E-01	3.64E-10
8	1.00E+11	3.00E-01	0.00E+00	0.00E+00	-1.35E-09	5.00E-01	1.35E-09
9	1.00E+11	3.50E-01	0.00E+00	0.00E+00	-5.01E-09	5.00E-01	5.01E-09
10	1.00E+11	4.00E-01	0.00E+00	0.00E+00	-1.86E-08	5.00E-01	1.86E-08
11	1.00E+11	4.50E-01	0.00E+00	0.00E+00	-6.67E-08	5.00E-01	6.67E-08
12	1.00E+11	5.00E-01	0.00E+00	0.00E+00	-2.19E-07	5.00E-01	2.19E-07
13	1.00E+11	5.50E-01	0.00E+00	0.00E+00	-6.09E-07	5.00E-01	6.09E-07
14	1.00E+11	6.00E-01	0.00E+00	0.00E+00	-1.38E-06	5.00E-01	1.38E-06
15	1.00E+11	6.50E-01	0.00E+00	0.00E+00	-2.61E-06	5.00E-01	2.61E-06

图 3-6: log文件导出成csv格式以查看仿真数据

3.7.3 DeckBuild抽取

DeckBuild有内建的QUICKMOS和QUICKBIP可以在工艺生成结构时就抽取器件特性，但是这些是一维的简陋的抽取。在第1.3.5节已经详细介绍了DeckBuild的抽取。

器件抽取是在ATLAS仿真得到log文件后，抽取log文件内的电学信息。可以是以数字形式显示在实时输出窗口，如 V_t 抽取。也可以抽取出曲线，保存在dat文件中。

Tonyplot能将结果导出成数据结构清晰的csv格式文件，那么抽取岂不显得尴尬？

我认为抽取的一大优势是能对数据进行分析 and 计算。如：统计功能(min)；拟合功能(slope)；计算功能(i.“collector”/i.“base”)。抽取的另一个巨大优势是抽取得到的值可以作为优化的目标。优化功能已经在2.5节进行了详细介绍。

3.7.4 Tonyplot显示

Tonyplot可以显示结构文件，器件仿真得到的log文件以及抽取出的dat文件等。Tonyplot已经在1.3.4进行过详细介绍，这里就不重复了。

总 结

这一章讲解Silvaco的二维器件仿真。器件仿真没有图形界面，而且语法和参数之多，使得不可能像第二章介绍工艺仿真那样详细。主要思路是通过例句来学习常用的语法设置。在讲特性获取时尽可能地对各种特性的获取方式都照顾到。器件仿真的物理模型及相关参数将在附录B中做简要介绍。

第四章 高级的特性

前两章介绍了二维工艺仿真和二维器件仿真，Silvaco也能进行三维工艺仿真和三维器件仿真。

Silvaco有很强的可扩展功能。这些功能包括自定义材料，C解释器描写参数等。

4.1 三维仿真

4.1.1 三维工艺仿真

SSuprem3是一维工艺仿真器，ATHENA可以进行二维仿真，Victory仿真器可以进行三维仿真。

VICTORY能仿真所有的工艺流程：刻蚀、淀积、光刻、氧化、离子注入和扩散等。VICTORY仿真是基于ATHENA语法的。

由于条件有限，没有接触过VICTORY，所以没有办法在这里介绍。

4.1.2 三维器件仿真

三维器件仿真可以用VICTORY，也可由ATLAS完成。

ATLAS中三维器件仿真的主要模块有：

- DEVICE3D：硅化合物材料和异质结仿真
- GIGA3D：不等温仿真
- MIXEDMODE3D：器件-电路混合仿真
- TFT3D：TFT仿真
- QUANTUM3D：量子效应仿真
- LUMINOUS3D：光探测器仿真

仿真之前需定义结构。有3种方法可以得到三维结构。

- 1、用三维工艺仿真器VICTORY直接得到三维结构。
- 2、用ATLAS命令来生成三维结构。
- 3、用DevEdit3D来生成三维结构。

介绍二维器件仿真中结构的定义时曾提过ATLAS命令生成结构。其中结构的维度参数有x.mesh,y.mesh。如果由命令生成三维结构就只需指定参数z.mesh，相应的位置(如电极位置)参数都需定义成三维坐标。

```
region num=1 material=silicon x.min=0 x.max=3 y.min=0 y.max=5.0 \
z.min=0 z.max=3
```

```
electrode name=anode x.min=-1 x.max=1 y.min=0 y.max=3 z.min=0 z.max=1
```

```
doping uniform p.type conc=1e20 junc=.5 z.min=0 z.max=2
```

类似的，三维结构还可以用DevEdit3D编辑得到，结构编辑方法可在二维结构编写的基础上添加Z轴的坐标。DevEdit3D有图形化的界面(linux下DevEdit和DevEdit3D都是图形化界面)，图4-1即是DevEdit3D的界面。其中已经编辑好了一个二极管，图中显示了不同的区域和网格。图4-3为编辑完成后得到的三维二极管结构。




图 4-1: DevEdit3D界面

DevEdit3D的语法和DevEdit相近，简单示意如下：

```
region reg=1 name=silicon mat=silicon color=0xffc000 pattern=0x3 \
  z1=0 z2=5 points="3,0.6 6.9,0.6 6.9,2.8 3,2.8"
impurity id=1 region.id=1 imp=Phosphorus color=0x906000 \
  x1=0 x2=0 y1=0 y2=0 \
  peak.value=1e+16 ref.value=0 z1=0 z2=5 comb.func=Multiply \
  rolloff.y=both conc.func.y=Constant \
  rolloff.x=both conc.func.x=Constant \
  rolloff.z=both conc.func.z=Constant
```

器件编辑器除了可以完全定义一个结构之外，还可以在现有的结构(如ATHENA仿真得到的结构)基础上进行编辑从而得到三维结构。

下面的例句是用三维器件编辑器导入二维结构文件(ATHENA仿真得到的)mos2d.str，然后进行编辑。

```
go devedit simflags="-3d"
# load file with specified Z extent
init inf=mos2d.str z1=0 z2=1.1
region reg=1 mat=Silicon z1=0 z2=1.1
region reg=2 mat="Silicon Oxide" z1=0 z2=1.1
region reg=3 mat=PolySilicon z1=0.3 z2=0.8
```

上面语句中z1和z2参数是将原二维结构扩展到三维时设定Z轴的范围。




图 4-2: DevEdit3D导入的原二维结构以及编辑得到三维结构的二维剖面图

图4-2(a)为ATHENA仿真得到的二维结构，这是MOS垂直于沟道方向的结构（电极只有多晶硅栅Gate）。图4-2(b)是将(a)结构沿Z轴扩展（请注意：上一页例句中在Z轴扩展时材料silicon, “silicon oxide”和polysilicon的扩展范围是不一样的）后得到三维结构，再从三维结构中Cutplane(如图4-4)得到的二维剖面。

和DevEdit一样，DevEdit3D中的区域是由“point”连起来的，相比用ATLAS命令得到结构而言有很大的灵活性。

结构定义中也有些限制。二维仿真的网格点数最多为20,000个(ATLAS也有网格点数为100,000的版本，不过需要获得Sivaco office的支持)。三维仿真的网格点最多可以达到40,000,000个，但任意Z平面的点不超过20,000，Z平面的数目不超过2,000。ATLAS 二维和三维仿真的区域数不超过1,000，电极数不超过50。

三维结构生成后，需设置材料参数和物理模型。材料参数和物理模型的设置和二维器件仿真的状态是通用的。这些状态是：models, impact, material, mobility, interface和contact。

器件结构、材料参数、物理模型和计算方法（可参考二维器件仿真的设置）设置好即可仿真其电学、热学、磁学、光学等特性。

器件特性的获取方式还是要参考二维器件仿真，只是位置参数需改成三维坐标。

图1-7是二维MOS仿真得到的特性曲线，显示的单位是V和A，实际上应该是 $V/\mu m$ 和 $A/\mu m$ （系统示例的显示设置有的就设置成 $A/\mu m$ ，如BJT输出特性曲线时每一条曲线对应的ibase, AC仿真时的电容单位为 $F/\mu m$ ）。也就是说实际上是线密度，乘以器件Z轴的长度(μm)就能得到三维器件的特性（注意这是器件结构在Z轴方向没有差异下的近似结果）。三维器件仿真得到的结果直接就是V和A。

4.1.3 Tonyplot3D

三维结构的数据量大得多(所以仿真速度要慢)，相应的文件体积也大。三维结构的显示需要使用Tonyplot3D。

图4.1.3为Tonyplot3D显示的器件结构，图中还按等温线显示出了电场分布。





图 4-3： Tonyplot显示器件结构

介绍Tonyplot的时候提到Cutline可得到一维信息，同样Tonyplot3D可由Cutplane得到二维信息。图4-4为Cutplane及其预览的效果，Cutplane得到的二维结构如图4-2(b)所示。



(a) Tonyplot3D的Cutplane

(b) Cutplane预览

图 4-4： Tonyplot3D的Cutplane及其预览

4.2 C解释器

Silvaco有内建的C语言解释器¹，语法和C语言相似，读起来很方便。可由C解释器编写的函数文件来描述仿真中的参数。

下面是定义迁移率随组分、温度和掺杂变化的一个例子。

```
#include <stdio.h>
#include <stdlib.h>
#include <math.h>
#include <ctype.h>
#include <malloc.h>
#include <string.h>
#include <template.h>
/*
 * Composition, temperature and doping dependent electron mobility
 * Statement: MATERIAL
 * Parameter: F.CONMUN
 * Arguments:
 * xcomp composition fraction x
 * ycomp composition fraction y
 * temp temperature (K)
 * nd net concentration of donors (per cc)
 * na net concentration of acceptors (per cc)
 * *mun electron mobility (cm2/Vs)
 */
int comun(double xcomp,double ycomp,double temp,double nd,double na,
          double x,double y,double taun,double *mun)
{
    double mu0, ntot;
    ntot = nd+na;
    if(xcomp < 0.1)
        mu0 = (9000.0-2e4*xcomp);
    else if(xcomp < 0.3)
        mu0 = (90.0 + 1.1e5*(xcomp-0.3)*(xcomp-0.3));
    else if(xcomp < 0.5)
        mu0 = (90.0 + 4e4*(xcomp-0.3)*(xcomp-0.3));
    else
        mu0 = (200.0-2.0/(xcomp-0.3));
    *mun = (mu0*(0.34+0.66/(1.0+ntot/9.85e16,0.55)));
    return(0);          /* 0 - ok */
}
```

¹SILVACO C - INTERPRETER (SCI)

编写的时候需要指明变量，然后对变量的关系进行描述，最后得到目标变量表达式。return(0)中的“0”表示OK,如果是“1”则fail。

C解释器编写的文件保存为*.lib文件，在仿真中调用时需在材料参数处指明函数文件：

```
material f.conmun="my_mun.lib"
```

#include <***.h> 为采用的模板，这些模板中有函数文件编写的规范表述。

这些模板文件的位置：

X:\sedatools\lib\Atlas***.R\common\SCI

X:\sedatools\lib\Atlas***.R\common 目录下的template.lib包含有很多已编好的函数文件。从这个模板文件中能找到你需要的信息。

ATLAS中可用的C解释器编写的函数文件描述的参数列表在ATLAS手册中附录C-Interpreter Functions部分。

注意将C解释器得到函数文件（F.***）和之前介绍折射率文件（波长和折射率）和power.file（强度和波长）进行区别。

总 结

这一章粗略地介绍了Silvaco的一些高级特性。包括三维仿真，由C解释器编辑函数文件来描述仿真中的参数。三维仿真很多可以参考二维仿真的描述，写得文字虽少，实则内容相当丰富。希望各位能结合第三章，看到更深层的东西。

附录 A 材料系统

A.1 材料类型

ATLAS中所有的材料都是半导体、绝缘体和半导体等三类材料中的一种。每一类材料都有相应的特性。

1、半导体：

半导体材料必须设定能带参数（ $E_g, N_c, N_v, \chi, \text{align}$ 等）。要用Tonyplot显示能带结构，必须在output状态中设置con.band和val.band。仿真时所有选择的模型的相应方程都会进行计算。当模型中设置了参数print时，半导体材料以及采用的模型都将在实时输出窗口中列出来。

如果半导体区域被定义成电极（如多晶硅栅），那么将会当导体对待。

2、绝缘体：

绝缘体材料只会计算Poisson方程和晶格热方程。绝缘体要设定介电常数。

3、导体：

所有导体材料必须定义成电极，并且所有电极也都会认为是导体。如果结构内含有导体区域，则会自动当成未命名的电极。如果含有未知材料则会认为它就是绝缘体材料。

电学仿真时，只有电极边界的网格点会计算其特性。电极内部的网格点只在光电仿真时计算光线轨迹和光的吸收。

表A-1为ATLAS仿真时已知的材料及其类型。

表 A-1: ATLAS材料

单质半导体								
Silicon	Poly	Germanium	Diamond					
二元化合物半导体								
AlAs	AlN	AlP	AlSb	BeTe	CdS	CdSe	CdTe	GaAs
GaN	GaP	GaSb	HgS	HgSe	HgTe	IGZO	InAs	InN
InP	InSb	PbS	PbSe	PbTe	ScN	SiC-3C	SiC-4H	SiC-6H
SiGe	SnTe	ZnO	ZnS	ZnSe	ZnTe			
三元化合物半导体								
AlAsSb	AlGaAs	AlGaN	AlGaP	AlGaSb	AlPAs	AlPSb	CdZnTe	GaAsP
GaAsSb	GaSbAs	GaSbP	HgCdTe	InAlAs	InAlP	InAlSb	InAsP	InAsSb
InGaAs	InGaN	InGaP	InGaSb	InPSb				
四元化合物半导体								
AlGaAsP	AlGaAsSb	AlGaNAs	AlGaNP	AlInNAs	AlInNP	InAlAsP	InAlGaAs	InAlGaN
InAlGaP	InGaAsP	InGaNAs	InGaNP					

表A-1:ATLAS材料（续）

绝缘体							
Air	Al2O3	Ambient	BPSG	BSG	CuInGaSe2	HfO2	HfSiO4
InAlAsSb	InGaAsSb	InPAsSb	Nitride	Oxide	OxyNitride	Sapphire	Si3N4
SiN	SiO2	Vacuum					
导体							
AlSi	AlSi	AlSiCu	AlSiTi	Aluminum	Cobalt	Conductor	Contact
Copper	CoSi	Gold	Iron	ITO	Lead	Molybdenum	MoSi
Nickel	NiSi	Palladium	PaSi	Platinum	Polysilicon	PtSi	Silver
Tantalum	TaSi	Tin	TiSi	Titanium	TiW	Tungsten	WSi
ZrSi							
有机物							
Alq3	IGZO	NPB	Organic	Pentacene	PPV	Tetracene	TPD

A.2 自定义材料

Silvaco提供了用户自定义材料功能。可以定义程序中没有包含的材料。

上一节也提到Silvaco将材料分成半导体、绝缘体和导体三组。所以自定义的材料也必须是其中的一种。

定义类型用user.group参数。

user.group=semiconductor
user.group=insulator
user.group=conductor

材料名称用user.material参数。

user.material=my_material

新定义的材料采用默认的类型。

user.default=default_atlas_material

最近稀磁半导体的研究热情很高，我就举一个自定义GaMnAs材料的例子吧。主要想通过语句从形式上了解自定义材料功能的使用方法，所以GaMnAs的参数该定义成什么值并没有关心（因为模型里设置了print，如果要查看仿真时的参数，在实时输出窗口对比就可以了）。图A-1所示结构中，中间的层即是GaMnAs材料。

go atlas
mesh
x.m l=0.0 spac=0.1
x.m l=5.0 spac=0.1
y.m l=0.0 spac=0.05
y.m l=3.2 spac=0.05
y.m l=5.0 spac=0.1


```

region num=1 material=GaAs y.max=0.1
region num=2 user.material=GaMnAs y.min=0.1 y.max=0.5
region num=3 material=GaAs y.min=0.5
#
elec name=anode top
elec name=cathode bottom
#
doping region=1 uniform conc=2.0e19 p.type
doping region=2 uniform conc=1.0e16 n.type
doping region=3 uniform conc=5.0e18 n.type
#
material material=GaMnAs user.group=semiconductor permittivity=13.2 \
  eg300=1.44 affinity=4.07 augn=5e-30 augp=9.9e-32 vsat=7.7e6
material num=2 mun0=8000 mup0=400 taun0=1.e-9 taup0=2.e-8
#
model fldmob srh auger bgn print temperature=300
output con.band val.band band.para
#
solve init
save outfile=my_GaMnAs.str
tonyplot my_GaMnAs.str
quit

```




图 A-1: 自定义材料GaMnAs

附录 B 物理

本附录介绍atlas器件仿真的物理部分，包括器件物理方程和模型及模型的参数。本打算详细介绍，但内容太多，这里只简要地提及(很多内容省略了)，详见手册的物理部分。

这章附录的另一个目的是想体验一下L^AT_EX在数学公式排版上的应用。

B.1 基本半导体方程

经过多年的研究，半导体器件特性的数学计算已经能得到较满意的效果。这些计算是基于一系列数学方程来描述的物理模型。

器件仿真的通用框架是泊松方程和连续性方程。其中 $J_n, J_p, G_n, G_p, R_n, R_p$ ，迁移率，载流子浓度，禁带变窄，少子寿命，光生成速率等等参数都有专门的模型来定义。不同的模型表达式会有差别。将基本的方程中的量去耦（器件仿真时介绍计算方法时提到过变量去耦），然后相应的模型求出这些量，再带入方程进行计算。

电流密度方程和电荷传输模型通常采用玻尔兹曼近似。这些由不同的传输模型，如漂移-扩散模型，能量平衡传输模型和水力学模型等设定。电荷传输模型主要受所选的生成-复合模型的影响。电荷传输模型和生成-复合模型使用一些和载流子统计相关的概念。

泊松方程

泊松方程描述静电势和空间电荷密度的分布：

$$\text{div}(\varepsilon \nabla \psi) = \rho \quad (\text{B-1})$$

ψ 是静电势， ε 是介电常数， ρ 为空间电荷密度。电场可由电势的梯度求得：

$$\vec{E} = -\nabla \psi \quad (\text{B-2})$$

载流子连续性方程

电子和空穴的连续性方程由式(B-3)和式(B-4)两个方程定义：

$$\frac{\partial n}{\partial t} = \frac{1}{q} \text{div} \vec{J}_n + G_n - R_n \quad (\text{B-3})$$

$$\frac{\partial p}{\partial t} = -\frac{1}{q} \text{div} \vec{J}_p + G_p - R_p \quad (\text{B-4})$$

传输方程

漂移-扩散模型

$$\vec{J}_n = -q \mu_n n \nabla \phi_n \quad (\text{B-5})$$

$$\vec{J}_p = -q \mu_p p \nabla \phi_p \quad (\text{B-6})$$

注意 ϕ_n 和 ϕ_p 是准费米能级， μ_n 和 μ_p 分别是电子和空穴迁移率。

对于Boltzmann近似，载流子浓度 n 和 p ，可由下面的式子表示：

$$n = n_{ie} \exp \left[\frac{q(\psi - \phi_n)}{kT_L} \right] \quad (\text{B-7})$$

$$p = n_{ie} \exp \left[\frac{-q(\psi - \phi_p)}{kT_L} \right] \quad (\text{B-8})$$

n_{ie} 是有效本征载流子浓度， T_L 是晶格温度。可由上两式得到准费米能级。

$$\phi_n = \psi - \frac{kT_L}{q} \ln \frac{n}{n_{ie}} \quad (\text{B-9})$$

$$\phi_p = \psi + \frac{kT_L}{q} \ln \frac{p}{n_{ie}} \quad (\text{B-10})$$

综合式(B-5),(B-6),(B-9)和(B-10)可得电流密度和电势的表达式：

$$\vec{J}_n = qD_n \nabla n - qn\mu_n \nabla \psi - \mu_n n (kT_L \nabla (\ln n_{ie})) \quad (\text{B-11})$$

$$\vec{J}_p = -qD_p \nabla p - qp\mu_p \nabla \psi + \mu_p p (kT_L \nabla (\ln n_{ie})) \quad (\text{B-12})$$

式(B-11)和式(B-12)中最后一项为有效本征载流子浓度的梯度，这是受禁带变窄的影响。

于是式(B-2)的电场将由式(B-13)和式(B-14)的有效电场取代：

$$\vec{E}_n = -\nabla \left(\psi + \frac{kT_L}{q} \ln n_{ie} \right) \quad (\text{B-13})$$

$$\vec{E}_p = -\nabla \left(\psi - \frac{kT_L}{q} \ln n_{ie} \right) \quad (\text{B-14})$$

所以，总的漂移-扩散方程的表达式如式(B-15)和式(B-16)：

$$\vec{J}_n = qn\mu_n \vec{E}_n + qD_n \nabla n \quad (\text{B-15})$$

$$\vec{J}_p = qn\mu_p \vec{E}_p - qD_p \nabla p \quad (\text{B-16})$$

上面的推导应用了爱因斯坦关系：

$$D_n = \frac{kT_L}{q} \mu_n \quad (\text{B-17})$$

$$D_p = \frac{kT_L}{q} \mu_p \quad (\text{B-18})$$

如果采用费米-狄拉克统计，则式(B-17)为以下形式：

$$D_n = \frac{\left(\frac{kT_L}{q} \mu_n \right) F_{1/2} \left\{ \frac{1}{kT_L} [\varepsilon_{Fn} - \varepsilon_C] \right\}}{F_{-1/2} \left\{ \frac{1}{kT_L} [\varepsilon_{Fn} - \varepsilon_C] \right\}} \quad (\text{B-19})$$

能量平衡传输模型

玻尔兹曼传输方程中如果载流子温度(或能量)与电流密度发生耦合，则漂移-扩散模型需进行改进，以考虑相应的影响。

式(B-20)致式(B-23)是电流密度和能流密度的表达式:

$$\vec{J}_n = qD_n \nabla n - q\mu_n n \nabla \psi + qnD_n^T \nabla T_n \quad (\text{B-20})$$

$$\vec{S}_n = -K_n \nabla T_n - \left(\frac{k\delta_n}{q} \right) \vec{J}_n T_n \quad (\text{B-21})$$

$$\vec{J}_p = -qD_p \nabla p - q\mu_p p \nabla \psi - qnD_p^T \nabla T_p \quad (\text{B-22})$$

$$\vec{S}_p = -K_p \nabla T_p - \left(\frac{k\delta_p}{q} \right) \vec{J}_p T_p \quad (\text{B-23})$$

T_n 和 T_p 为电子和空穴温度, S_n 和 S_p 是载流子传递给晶格的能流(热)。

位移电流方程

时域仿真时, 位移电流的计算结果可保存在结构文件和日子文件中, 也会在实时输出窗口显示。

$$\vec{J}_{dis} = \varepsilon \left(\frac{\partial E}{\partial t} \right) \quad (\text{B-24})$$

B.2 基本的载流子统计理论

Fermi-Dirac和Boltzmann统计

Fermi-Dirac统计的状态密度 $f(\varepsilon)$, T_L 是晶格温度, ε 是能量。

$$f(\varepsilon) = \frac{1}{1 + \exp\left(\frac{\varepsilon - E_F}{kT_L}\right)} \quad (\text{B-25})$$

Boltzmann统计的状态密度:

$$f(\varepsilon) = \exp\left(\frac{E_F - \varepsilon}{kT_L}\right) \quad (\text{B-26})$$

有效状态密度

费米-狄拉克统计的电子和空穴密度表达式:

$$n = N_C F_{1/2} \left(\frac{E_F - E_C}{kT_L} \right) \quad (\text{B-27})$$

$$p = N_V F_{1/2} \left(\frac{E_V - E_F}{kT_L} \right) \quad (\text{B-28})$$

玻尔兹曼近似时, 电子和空穴密度表达式:

$$n = N_C \exp\left(\frac{E_F - E_C}{kT_L}\right) \quad (\text{B-29})$$

$$p = N_V \exp\left(\frac{E_V - E_F}{kT_L}\right) \quad (\text{B-30})$$

N_C 和 N_V 是电子和空穴的有效状态密度:

$$N_C(T_L) = 2 \left(\frac{2\pi m_e^* k T_L}{h^2} \right)^{\frac{3}{2}} = \left(\frac{T_L}{300} \right)^{NC.F} NC300 \quad (B-31)$$

$$N_V(T_L) = 2 \left(\frac{2\pi m_h^* k T_L}{h^2} \right)^{\frac{3}{2}} = \left(\frac{T_L}{300} \right)^{NV.F} NV300 \quad (B-32)$$

表 B-1: 状态密度的参数

状态	参数	默认值	单位
material	NC300	2.8×10^{19}	cm^{-3}
material	NV300	1.04×10^{19}	cm^{-3}
material	NCF	1.5	
material	NVF	1.5	

本征载流子浓度

由式(B-29)和式(B-30)可得本征载流子浓度 n_{ie} :

$$np = n_{ie}^2 \quad (B-33)$$

本征载流子浓度 n_{ie} 由Boltzmann统计得:

$$n_{ie} = \sqrt{N_C N_V} \exp \left(\frac{-E_g}{2kT_L} \right) \quad (B-34)$$

$E_g = E_C - E_V$ 为禁带宽度。对于本征材料 $p = n$, 则 E_F :

$$E_F = E_i = -q\psi_i = \frac{E_C + E_V}{2} + \left(\frac{kT_L}{2} \right) \ln \left(\frac{N_V}{N_C} \right) \quad (B-35)$$

电子浓度和空穴浓度由本征载流子浓度表达如下, ψ 是内建势, ϕ 为费米势($E_F = q\phi$)

$$n = n_{ie} \exp \left[\frac{q(\psi - \phi_n)}{kT_L} \right] \quad (B-36)$$

$$p = n_{ie} \exp \left[\frac{-q(\psi - \phi_p)}{kT_L} \right] \quad (B-37)$$

常规的禁带模型

温度依赖的带隙能量:

$$\begin{aligned} E_g(T_L) &= E_g(0) - \frac{EGALPHA(T_L^2)}{T_L + EGBETA} \\ &= EG300 + EGALPHA \left[\frac{300^2}{300 + EGBETA} - \frac{T_L^2}{T_L + EGBETA} \right] \end{aligned} \quad (B-38)$$

一般的三元化合物能带模型

$$E_g = EG2300 * x + EG1300 * (1 - x) - EG12BOW * x * (1 - x) \quad (B-39)$$

表 B-2: 常规能带参数

状态	参数	默认值	单位
material	EG300	1.08	eV
material	EGALPHA	4.73×10^{-4}	eV/K
material	EGBETA	636	K

$$E_{g1} = EG1300 + EG1ALPH \left[\frac{300^2}{300 + EG1BETA} - \frac{T_L^2}{T_L + EG1BETA} \right] \quad (B-40)$$

$$E_{g2} = EG2300 + EG2ALPH \left[\frac{300^2}{300 + EG2BETA} - \frac{T_L^2}{T_L + EG2BETA} \right] \quad (B-41)$$

$$E_g = E_{g2} * x + E_{g1}(1 - x) - EG12BOW * x * (1 - x) \quad (B-42)$$

禁带变窄时的有效本征载流子浓度:

$$n_{ie}^2 = n_i^2 \exp \left(\frac{\Delta E_g}{kT} \right) \quad (B-43)$$

禁带变窄模型方程:

$$\Delta E_g = BGN.E \left\{ \ln \frac{N}{BGN.N} + \left[\left(\ln \frac{N}{BGN.N} \right)^2 + BGN.C \right]^{\frac{1}{2}} \right\} \quad (B-44)$$

表 B-3: 禁带变窄模型及其参数

状态	参数	默认值(Slotboom)	默认值(Klaassen)	单位
material	BGN.E	9.0×10^{-3}	6.92×10^{-3}	V
material	BGN.N	1.0×10^{17}	1.3×10^{17}	cm^{-3}
material	BGN.C	0.9	0.5	—

$$\vec{E}_n = -\nabla \left(\psi + \frac{kT_L}{q} \ln n_{ie} \right) \quad (B-45)$$

$$\vec{E}_p = -\nabla \left(\psi - \frac{kT_L}{q} \ln n_{ie} \right) \quad (B-46)$$

$$\chi_{eff} = \chi + \Delta E_g \times ASYMMETRY \quad (B-47)$$

空间电荷

完全离化、陷阱和缺陷引起的空间电荷:

$$\text{div}(\epsilon \nabla \psi) = q (n - p - N_D^+ + N_A^-) - Q_T \quad (B-48)$$

$$N_D^+ = \frac{N_D}{1 + GCB \exp \left(\frac{\epsilon_{Fn} - (E_C - E_{DB})}{kT_L} \right)} \quad (B-49)$$

$$N_A^- = \frac{N_A}{1 + GVB \exp\left(\frac{E_V + E_{AB} - \varepsilon_{F_p}}{kT_L}\right)} \quad (\text{B-50})$$

B.3 边界物理

边界物理的内容略去不提，下面是边界物理部分的内容（详见手册）。

欧姆接触，肖特基接触，浮接触，电流边界条件，绝缘接触，上拉元件边界，分布式接触电阻

B.4 物理模型

ATLAS有很多物理模型，合理的设定物理模型对仿真准确性紧密相关。

迁移率模型

ATLAS中迁移率模型有浓度依赖迁移率(conmob)，浓度和温度依赖迁移率模型(analytic)，Arora模型(arora)，载流子-载流子散射模型(ccsmob)，平行电场依赖模型(fldmob)，Tasch模型(tasch)，Klaassen模型(kla)，Shirahata模型(shi)，Watt扩展模型(mod.watt)，Lombardi模型(cvt)，Yamaguchi模型(yamaguchi)，Watt模型(watt)，等迁移率模型。

低场迁移率模型:

$$\mu_{n0} = MUN \left(\frac{T_L}{300} \right)^{-TMUN} \quad (\text{B-51})$$

$$\mu_{p0} = MUP \left(\frac{T_L}{300} \right)^{-TMUP} \quad (\text{B-52})$$

表 B-4: 低场迁移率模型参数

状态	参数	默认值	单位
mobility	mun	1000	$\text{cm}^2/(\text{V} \cdot \text{s})$
mobility	mup	500	$\text{cm}^2/(\text{V} \cdot \text{s})$
mobility	tmun	1.5	K
mobility	tmup	1.5	K

载流子生成-复合模型

ATLAS中载流子生成-复合模型有Shockley-Read-Hall(SRH)复合模型，SRH浓度依赖的寿命模型(consrh)，Klaassen浓度依赖的寿命模型(klasrh)，Scharfetter浓度依赖的寿命模型(schsrh)，陷阱增强俄歇复合模型(trap.auger)，陷阱增强隧穿模型(trap.tunnel)，光生成-辐射复合模型(optr)，俄歇复合模型(auger)，Klaassen温度依赖的俄歇复合模

型(klaaug)，带隙变窄俄歇复合模型(augkp)，温度和浓度系数俄歇复合模型(hnsaug)，表面复合（由界面状态interface中的参数s.n和s.p设定）等。

下面简单的给出SRH复合和俄歇复合的表达式以及参数的默认值。

平衡的情况下本征载流子浓度满足下面的关系：

$$n_0 p_0 = n_i^2 \quad (\text{B-53})$$

Shockley-Read-Hall(SRH)复合模型：

$$R_{SRH} = \frac{pn - n_{ie}^2}{TAUP0 \left[n + n_{ie} \exp\left(\frac{ETRAP}{kT_L}\right) \right] + TAUN0 \left[p + n_{ie} \exp\left(\frac{-ETRAP}{kT_L}\right) \right]} \quad (\text{B-54})$$

表 B-5: SRH复合模型参数

状态	参数	默认值	单位
material	etrap	0	eV
material	taun0	1×10^{-7}	s
material	taup0	1×10^{-7}	s

俄歇复合模型：

$$R_{Auger} = AUGN (pn^2 - nn_{ie}^2) + AUGP (np^2 - pn_{ie}^2) \quad (\text{B-55})$$

表 B-6: 俄歇复合模型参数

状态	参数	默认值	单位
material	augn	8.3×10^{-32}	cm^6/s
material	augp	1.8×10^{-31}	cm^6/s

碰撞离化模型

只要电场足够高，都能使载流子加速到速度够高，和原子碰撞时可产生电子-空穴对。

碰撞离化模型有几何情形碰撞离化模型(gradqfl)，Selberherr碰撞离化模型(selb)

$$G = \alpha_n |\vec{J}|_n + \alpha_p |\vec{J}|_p \quad (\text{B-56})$$

G 是电子空穴对的生成速率， $\alpha_{n,p}$ 是离化系数， $J_{n,p}$ 为电流密度。

物理模型还有以下内容。物理意义的描述和方程非三言两语能说得清楚（详见手册），请允许我偷偷懒哈^_^。

带-带隧穿模型，栅电流模型，器件级的可靠性模型，铁电体介电常数模型，外延应力模型，极化模型，压力影响硅带隙模型，应力硅电场迁移率模型，光电子模型，光学指数模型，磁场下载流子传输模型，各向异性介电常数模型。

索引

- 3d device simulator, [93](#)
- AC solution, [85](#)
- advice, [18](#)
- any shape etch, [40](#)
- athena, [21](#)
- atlas, [65](#)
- atlas ability, [66](#)
- atlas command define structure, [69](#)
- atlas material, [99](#)
- atlas mesh, [69](#)
- bake, [53](#)
- base on physics, [1](#)
- binary compound semiconductors, [99](#)
- Block Iteration, [81](#)
- BreakDown, [82](#)
- C - Interpreter, [97](#)
- command, [8](#)
- conductors, [100](#)
- contact, [74](#)
- Convergence Criteria, [81](#)
- Current Boundary Conditions, [74](#)
- curvetrace, [85](#)
- data file, [91](#)
- DC solution, [82](#)
- deposit, [35](#)
- devddit3d, [94](#)
- DevEdit define structure, [72](#)
- develop, [54](#)
- diffuse, [33](#)
- divisions, [26](#)
- doping, [70](#)
- dry etch, [40](#)
- electrode, [70](#)
- energy band parameters, [77](#)
- epitaxy, [43](#)
- etch, [37](#)
- expose, [52](#)
- External Capacitors, [74](#)
- External Inductors, [74](#)
- External Resistors, [74](#)
- extract, [12](#)
- extract in atlas , [91](#)
- filter, [48](#)
- Floating Contacts, [74](#)
- gas flow, [33](#)
- go, [8](#)
- grid, [25](#)
- Gummel Iteration, [80](#)
- Gummel Plot, [83](#)
- Hall Effect, [86](#)
- Ic/Vce, [84](#)
- Id/Vds, [84](#)
- illumination, [46](#)
- image, [50](#)
- implant, [30](#)
- init, [27](#)
- insulators, [100](#)
- interface, [79](#)
- layout, [49](#)
- log file, [91](#)
- mask, [46](#)
- Material Properties, [75](#)
- mesh, [25](#)
- Monte Carlo implant, [32](#)
- Monte Carlo plasma etch, [40](#)
- Newton Iteration, [80](#)

node, [26](#)
numerical computation, [1](#)
Numerical Methods, [80](#)

Open Circuit Contact, [75](#)
optimizer, [60](#)
optimizer parameter, [62](#)
optimizer result, [62](#)
optimizer setup, [60](#)
optimizer targets, [62](#)
Optoelectronics, [86](#)
OPTOLITH, [45](#)
organics, [100](#)
oxygen, [34](#)

parameter type, [67](#)
photo, [45](#)
physics model, [79](#)
plasma etch, [39](#)
point, [26](#)
polish, [44](#)
print, [89](#)
print.mom, [32](#)
process simulation roadmap, [23](#)
projection, [47](#)

quaternary compound semiconductors, [99](#)

rate.deposit, [36](#)
rate.develop, [54](#)
rate.polish, [44](#)
region, [70](#)
RIE etch, [38](#)
Run-Time Output, [88](#)

S - parameter, [85](#)
Schottky Contacts, [74](#)
SCI, [97](#)
set, [9](#)
Shorting Contacts Together, [75](#)

Silvaco, [2](#)
single element semiconductors, [99](#)
Spectral Response, [87](#)
statement, [67](#)
structure, [29](#)
SVDP implant, [31](#)

TCAD, [1](#)
temperature ramped when diffuse , [34](#)
ternary compound semiconductors, [99](#)
tonyplot, [10](#)
tonyplot3d, [96](#)
transfer property, [83](#)
transient solution, [85](#)

victory, [93](#)

wet etch, [38](#)