微电子工艺实验与仿真
指导手册

谢 生

电子科学与技术实验教学中心

实验一 SILVACO ATHENA 入门

实验目的：熟悉 Silvaco Athena 工艺仿真软件的基本操作

实验内容：1、启动 Silvaco Athena 工艺仿真软件；

2、结合软件自带 Examples，熟悉 Deckbuild 界面的基本操作；

3、创建一个仿真用的网格。

实验步骤：

1、启动 Silvaco Athena 工艺仿真软件

双击桌面图标 S.EDA Tools，出现如下界面



双击 Deckbuild 进入 Deckbuild 集成运行环境界面



2、加载 Athena Examples，熟悉 Deckbuild 界面的基本操作

单击 Example/Athena Examples 选择一个工艺输入文件



3、创建一个 Si(100)晶向、仿真面积为 $10\times1\mu m^2$ 的网格区

```
### ** This example is to demonstrate the initialization grid ** ###
# Start Process Simulator "ATHENA"
go athena

#Define mesh for X plane
line x loc=0.00    spac=0.10
line x loc=10.00 spac=0.10

#Define mesh for y plane
line y loc=0.00    spac=0.02
line y loc=0.30    spac=0.03
line y loc=0.60    spac=0.04
line y loc=1.00    spac=0.05

#Initialisation of Si Sub of crystal orientation of 100 with Phosphorus dopants added, space.mult=2.0 to speed up the process
init silicon c.phosphor=1.0e14 orientation=100 two.d space.mult=2.0

# save the initialization grid, and then display by tonyplot
structure outfile=grid.str
tonyplot -st grid.str

#
quit
```

在 Deckbuild 集成运行环境界面的命令输入区输入下面一段脚本

仿真结束后，自动启动 Tonyplot 图形显示，设置网格如图 1 所示。

右键单击，选择 Display….，出现图 2 所示显示界面，熟悉各工具的作用。设计的最终仿真网格如图 3 所示。



图 1

图 2



读者可自行更改仿真区域大小及网格间距，观察网格疏密变化。

要插入的网格线定义点在位置 0.0 处，相邻网格线定义点的网格线间隔为 0.02，当两个定义点所设定的网格线间隔不同时，系统会自动将网格间距从较小值渐变到较大值。

实验二 硅氧化

一、实验目的：熟悉氧化工艺过程，理解晶向、掺杂、氧化氛围等因素对氧化的影响

二、实验内容：1、晶向对氧化的影响；

2、干氧和湿氧的影响；

3、不同掺杂类型和浓度的影响，及杂质在 SiO_2/Si 界面的再分布。

三、氧化模型：

在一维热氧化模型中，采用 Deal-Grove 热氧化公式

$$x^2 + Ax = B\left(t + \tau\right)$$

其中，x 是氧化厚度，t 是氧化时间，τ 是初始氧化层厚度 d_0 引起的时间坐标平移，A 和 B 与线性和抛物线形生长系数 K_L、K_p 以及归一化氧分压 P_{O2} 有关。

$$A = P_{O2}k_p / k_L \qquad B = P_{O2}k_p \qquad \tau = \frac{x_0{}^2 + Ax_0}{B}$$

在相对低的掺杂浓度下，k_p 和 k_l 只同硅晶向及氧化氛围有关，且他们是温度的简单激活函数。

在二维解析模型中，氧化模型考虑氧化时发生两种机制：(1) 氧化剂通过 SiO_2 扩散到 SiO_2/Si 界面；(2) 氧化物的流动由氧化物生产过程的体膨胀和相关联的应力逐步建立所引起，氧化物流是一种粘滞流。

氧化剂扩散由稳态通用扩散方程描述

$$D \cdot \nabla^2 C = \frac{\partial C}{\partial t} \cong 0$$

式中，D 和 C 分别为氧化剂的扩散常数和浓度。

氧化物流动由一个简化的 Navier-Stokes 方程描述

$$\mu \cdot \nabla^2 v = \nabla p$$

式中，μ, v, p 分别为氧化物的粘滞度、速度和压力。

原则上在一定的边界条件下解上述氧化剂扩散和氧化物流动方程就可得到氧化周期结束时 SiO_2 边界，从而满足二维氧化工艺模拟的需求。但由于 SiO_2 边界是一个连续移动的边界，解上述方程还需结合处理边界值的数值技术来得到所需的解。

在 SUPREM-4 中常用的氧化模型有 COMPRESS 模型和 VISCOUS 模型。关于氧化模型的详细介绍请参考 SILVACO ATHENA 用户手册中的氧化模型部分。

四、实验步骤：

1、晶向对氧化的影响

在 Deckbuild 集成运行环境界面的命令输入区输入下面一段脚本，并运行

```
# ** This example is to demonstrate the effects of wafer orientation on oxidation **
# Start Process Simulator "ATHENA"
go athena

#*** This part deck is to simulation the oxidation of Si(100) wafer ***
#Define mesh for X plane
line x loc=0.00    spac=0.10
line x loc=10.00 spac=0.10

#Define mesh for y plane
line y loc=0.00    spac=0.02
line y loc=0.30    spac=0.03
line y loc=0.60    spac=0.04
line y loc=1.00    spac=0.05

#Initialisation of Si Sub of crystal orientation of 100 with Phosphorus dopants added
init silicon c.phosphor=1.0e14 orientation=100 two.d
# Oxidation simulation for Si (100)
method compress
diffuse time=10 temperature=900 dryO2 T.rate=20 pressure=1 HCl.pc=0
diffuse time=40 temperature=900 dryO2    pressure=1 HCl.pc=0
diffuse time=10 temperature=1100 dryO2 T.rate=-20 pressure=1 HCl.pc=0
extract name="Tox_100" thickness oxide mat.occno=1 x.val=5

# save the structure
```

```
struc outfile=Oxid_100.str
tonyplot -st Oxid_100.str

#*** This part deck is to simulation the oxidation of Si(111) wafer ***
#Define mesh for X plane
line x loc=0.00    spac=0.10
line x loc=10.00 spac=0.10

#Define mesh for y plane
line y loc=0.00    spac=0.02
line y loc=0.30    spac=0.03
line y loc=0.60    spac=0.04
line y loc=1.00    spac=0.05

#Initialisation of Si Sub of crystal orientation of 111 with Phosphorus dopants added
init silicon c.phosphor=1.0e14 orientation=111 two.d

# Oxidation simulation for Si (111)
method compress
diffuse time=10 temperature=900 dryO2 T.rate=20 pressure=1 HCl.pc=0
diffuse time=40 temperature=900 dryO2    pressure=1 HCl.pc=0
diffuse time=10 temperature=1100 dryO2 T.rate=-20 pressure=1 HCl.pc=0
extract name="Tox_111" thickness oxide mat.occno=1 x.val=5

# save the structure
struc outfile=Oxid_111.str
tonyplot -add Oxid_111.str Oxid_100.str

quit
```

仿真过程中，注意观察命令执行窗口有无错误或警告！若无错误，查看 EXTRACT 命令抽取的（100）和（111）晶向的氧化层厚度。

```
Tox100=5583.45 angstroms (0.558345 um)    X.val=5
Tox111=5669.44 angstroms (0.566944 um)    X.val=5
```

仿真结束后，系统自动启动 Tonyplot 图形显示，两个晶向的 SiO_2/Si 界面如下图所示。



2、干氧和湿氧的影响

在 Deckbuild 集成运行环境界面的命令输入区输入下面一段脚本，并运行

```
### ** This example is to demonstrate the effects of ambient atmosphere ** ###
# Start Process Simulator "ATHENA"
go athena

#*** This part deck is to simulation the dryO2 oxidation of Si(100) wafer ***
#Define mesh for X plane
line x loc=0.00    spac=0.10
line x loc=10.00 spac=0.10

#Define mesh for y plane
line y loc=0.00    spac=0.02
line y loc=0.30    spac=0.03
line y loc=0.60    spac=0.04
line y loc=1.00    spac=0.05

#Initialization of Si Sub of crystal orientation of 100 with Phosphorus dopants added
init silicon c.phosphor=1.0e14 orientation=100 two.d

# DryO2 Oxidation simulation for Si (100)
method compress
diffuse time=20 temperature=900 dryO2 T.rate=20 pressure=1 Hcl.pc=0
diffuse time=10 temperature=1100 dryO2 T.rate=-20 pressure=1 Hcl.pc=0
extract name="Tox_dryO2" thickness oxide mat.occno=1 x.val=5
```

```
# save the structure
struc outfile=Oxid_dryO2.str
#tonyplot -st Oxid_dryO2.str

#*** This part deck is to simulation the wetO2 oxidation of Si(100) wafer ***
#Define mesh for X plane
line x loc=0.00    spac=0.10
line x loc=10.00 spac=0.10

#Define mesh for y plane
line y loc=0.00    spac=0.02
line y loc=0.30    spac=0.03
line y loc=0.60    spac=0.04
line y loc=1.00    spac=0.05

#Initialisation of Si Sub of crystal orientation of 100 with Phosphorus dopants added
init silicon c.phosphor=1.0e14 orientation=100 two.d
# WetO2 Oxidation simulation for Si (100)
method compress
diffuse time=20 temperature=900 wetO2 T.rate=20 pressure=1 Hcl.pc=0
diffuse time=10 temperature=1100 wetO2 T.rate=-20 pressure=1 Hcl.pc=0
extract name="Tox_wetO2" thickness oxide mat.occno=1 x.val=5

# save the structure
struc outfile=Oxid_wetO2.str
tonyplot -add Oxid_wetO2.str Oxid_dryO2.str

 quit
```

仿真过程中，注意观察命令执行窗口有无错误或警告！若无错误，查看 EXTRACT 命令抽取的 DryO2 和 WetO2 的氧化层厚度。

```
Tox_dryO2=981.386 angstroms (0.0981386 um)    X.val=5
Tox_wetO2=4906.63 angstroms (0.490663 um)    X.val=5
```

仿真结束后，系统自动启动 Tonyplot 图形显示，DryO2 和 WetO2 氧化的 SiO_2/Si 界面如下图所示。



3、不同掺杂类型和浓度的影响，及杂质在 SiO_2/Si 界面的再分布

在 Deckbuild 集成运行环境界面的命令输入区输入下面一段脚本，并运行

```
### ** This example is to demonstrate the effects of dopants on oxidation ** ###
# Start Process Simulator "ATHENA"
go athena

#*** This part deck is to simulation the oxidation of phosphor doping ***
#Define mesh for X plane
line x loc=0.00    spac=0.10
line x loc=10.00 spac=0.10

#Define mesh for y plane
line y loc=0.00    spac=0.02
line y loc=0.30    spac=0.03
line y loc=0.60    spac=0.04
line y loc=1.00    spac=0.05

#Initialization of Si Sub of crystal orientation of 100 with Phosphorus dopants added
init silicon c.phosphor=1.0e14 orientation=100 two.d

# Oxidation simulation for Si (100)
```

```
method compress
diffuse time=20 temperature=900 dryO2 T.rate=20 pressure=1 Hcl.pc=0
diffuse time=30 temperature=1100 wetO2 pressure=1 Hcl.pc=0
diffuse time=10 temperature=1100 dryO2 T.rate=-20 pressure=1 Hcl.pc=0
extract name="Tox_P" thickness oxide mat.occno=1 x.val=5

# save the structure
struc outfile=Oxid_P.str
#tonyplot -st Oxid_P.str

# *** This part deck is to simulation the oxidation of boron doping ***
#Define mesh for X plane
line x loc=0.00    spac=0.10
line x loc=10.00 spac=0.10

#Define mesh for y plane
line y loc=0.00    spac=0.02
line y loc=0.30    spac=0.03
line y loc=0.60    spac=0.04
line y loc=1.00    spac=0.05

#Initialization of Si Sub of crystal orientation of 100 with boron dopants added
init silicon c.boron=1.0e14 orientation=100 two.d

# Oxidation simulation for Si (100)
method compress
diffuse time=20 temperature=900 dryO2 T.rate=20 pressure=1 Hcl.pc=0
diffuse time=30 temperature=1100 wetO2 pressure=1 Hcl.pc=0
diffuse time=10 temperature=1100 dryO2 T.rate=-20 pressure=1 Hcl.pc=0
extract name="Tox_B" thickness oxide mat.occno=1 x.val=5

# save the structure
struc outfile=Oxid_B.str
#tonyplot -st Oxid_B.str

#*** This part deck is to simulation the oxidation of Gallium doping ***
#Define mesh for X plane
line x loc=0.00    spac=0.10
line x loc=10.00 spac=0.10

#Define mesh for y plane
line y loc=0.00    spac=0.02
line y loc=0.30    spac=0.03
line y loc=0.60    spac=0.04
```

```
line y loc=1.00    spac=0.05

#Initialisation of Si Sub of crystal orientation of 100 with Gallium dopants added
init silicon c.gallium=1.0e14 orientation=100 two.d

# Oxidation simulation for Si (100)
method compress
diffuse time=20 temperature=900 dryO2 T.rate=20 pressure=1 Hcl.pc=0
diffuse time=30 temperature=1100 wetO2 pressure=1 Hcl.pc=0
diffuse time=10 temperature=1100 dryO2 T.rate=-20 pressure=1 Hcl.pc=0
extract name="Tox_Ga" thickness oxide mat.occno=1 x.val=5

# save the structure
struc outfile=Oxid_Ga.str
tonyplot -add Oxid_P.str oxid_B.str oxid_Ga.str

quit
```

仿真结束后，**注意检查命令执行窗口有无错误或警告！**若无错误，查看 EXTRACT 命令抽取的 DryO2 和 WetO2 的氧化层厚度。

```
Tox_P=4948.11 angstroms (0.494811 um)    X.val=5
Tox_B=4940.63 angstroms (0.494063 um)    X.val=5
Tox_Ga=4949.05 angstroms (0.494905 um)    X.val=5
```

仿真结束后，系统自动启动 Tonyplot 图形显示，两个晶向的 SiO_2/Si 界面如下图所示。



选中 Oxid_P.str 结构图，选择 Tool/Cutlines，出现如下对话框，选择垂直线，在想观测位置划线，出现如下的杂质分布图





P 在 SiO_2/Si 界面的再分布





读者可自行更改脚本中衬底掺杂浓度、氧化时间、温度、压力、HCl 百分含量等参数，观察不同参数对氧化的影响。

练习：1、保持其他参数不变，比较 pressure=1 和 2 时的氧化层厚度；

2、保持其他参数不变，比较 Hcl.pc=0 和 15 时的氧化层厚度

实验三 光刻与几何刻蚀

一、实验目的：了解光刻与几何刻蚀的基本过程

二、实验内容：1、光刻窗口的定义；

2、氧化硅的刻蚀。

三、实验步骤：

步骤一 建立 X/Y 方向长为 $10\times1\mu m^2$ 的仿真网格

步骤二 硅单晶衬底、晶向<100>、P 掺杂浓度 $1.0\times10^{14}cm^{-3}$

步骤三 干氧气氛生长 0.1μm 的氧化层

步骤四 旋涂 0.1μm 厚的光刻胶

步骤五 根据光刻板上的几何图形窗口【坐标 (3,-0.15) (3,-0.05) (7,-0.05) (7,0.15)】刻蚀光刻胶

步骤六 以光刻胶为掩膜刻蚀下层氧化硅，刻蚀坐标(3,-0.05) (3,0.04) (7,0.04) (7,-0.05)

步骤七 去除光刻胶

仿真文件如下：

```
### ** This example shows the lithography and etch stage of wafer fabrication ** ###
# Start Process Simulator "ATHENA"
go athena

#Define mesh for X plane
line x loc=0.00    spac=0.10
line x loc=10.00 spac=0.10

#Define mesh for y plane
line y loc=0.00    spac=0.03
line y loc=0.30    spac=0.02
line y loc=0.60    spac=0.04
line y loc=1.00    spac=0.05

#Initialisation of Si Sub of crystal orientation of 100 with Phosphorus dopants added
init silicon c.phosphor=1.0e14 orientation=100 two.d

#Saving structure
struct outfile=lith_0.str
```

```
#Growing thin layer of oxide through compressible model:dry oxidation
method compress
diffus time=20 temp=1100 dryo2 press=1.00 hcl.pc=15

#Saving structure
struct outfile=lith_1.str

#Extraction of oxide thichness
extract name="tox" thickness material="SiO~2" mat.occno=1 x.val=5

#Application of photoresist
deposit photoresist thick=0.1
#Saving structure
struct outfile=lith_2.str

# Patterning photoresist
etch photores start x=3 y=-0.15
etch cont x=3 y=-0.05
etch cont x=7 y=-0.05
etch done x=7 y=-0.15

#Saving structure
struct outfile=lith_3.str

# Patterning oxide
etch oxide start x=3 y=0.04
etch cont x=3 y=-0.05
etch cont x=7 y=-0.05
etch done x=7 y=0.04

#Saving structure
struct outfile=lith_4.str

#Photoresist removal
etch photoresist all

#Saving final structure
struct outfile=lith_5.str

# display the structures with different stage
tonyplot -add lith_0.str lith_1.str lith_2.str lith_3.str lith_4.str lith_5.str

quit
```

仿真运行前，在语句 extract name="tox" thickness material="SiO~2" mat.occno=1 x.val=5 后设置断点。根据抽取的氧化层厚度，计算光刻胶及氧化硅刻蚀窗口坐标。

tox=907.402 angstroms (0.0907402 um) X.val=5

仿真结束后，Tonyplot 自动运行，工艺各阶段的截面图如下图所示。


Tonyplot
File Edit Plot Tools Production Help
ATHENA
Data from ith_0.str
ATHENA
Data from ith_1.str
ATHENA
Data from ith_2.str
ATHENA
Data from ith_3.str
ATHENA
Data from ith_4.str
ATHENA
Data from ith_5.str
Microns
Tonyplot 3.6.17.R © Silvaco 2012

实验四 热扩散

一、实验目的：熟悉基本扩散工艺过程，理解预沉积和再分布两种扩散的杂质浓度分布曲线的异同。

二、实验内容：1、恒定表面浓度扩散及其分布；

2、恒定杂质总量扩散及其分布；

三、扩散模型：

扩散工艺模拟器是基于三个基本方程来计算扩散后的杂质分布的。第一个方程为流量方程，其一维表达式为

$$J_i = -D\frac{dC_i}{dx} + Z_i\mu_i C_i E$$

其中，Z_i 是电荷状态，μ_i 是杂质迁移率，E 是电场。下标 i 表示 SUPREM 仿真软件的计算格点数。

第二个方程是连续性方程，其关系式为

$$\frac{dC_i}{dt} + \frac{dJ_i}{dx} = G_i$$

式中，G_i 是杂质的产生/复合率。

最后一个方程是泊松方程，即

$$\frac{d}{dx}\left[\varepsilon_s E\right] = q\left(p - n + N_D^+ - N_A\right)$$

式中，ε_s 是介电常数，n 和 p 分别是电子和空穴浓度，N_D^+ 和 N_A^- 则表示电离施主和受主浓度。

SUPREM 按照用户指定的一维格点同时求解上面的方程组，而 B、P、As 和 Sb 元素的 E_a 和 D_0 值则由查询表得到。

四、实验步骤：

步骤一 建立 X/Y 方向长为 $10\times1\mu m^2$ 的仿真网格

步骤二 硅单晶衬底、晶向<100>、P 掺杂浓度 $1.0\times10^{14}cm^{-3}$

步骤三 干氧气氛生长 0.1μm 的氧化层

步骤四 光刻、刻蚀氧化层，形成扩散窗口

步骤五 去除光刻胶

步骤六 恒定表面浓度扩散；

步骤七 恒定杂质总量扩散；

仿真文件如下：

```
### This input file is to simulate the 2-step diffusion ###
# Start Process Simulator ATHENA
Go Athena

# Define mesh for x coordinates
Line x loc=0.0 spac=0.02
Line x loc=0.5 spac=0.03
Line x loc=1.0 spac=0.10

# Define mesh for y coordinates
Line y loc=0.0 spac=0.02
Line y loc=0.3 spac=0.03
Line y loc=0.6 spac=0.04
Line y loc=1.0 spac=0.05

# N-type (100) silicon substate
Init silicon c.phosphor=1.0e12 orientation=100 two.d

# deposition of oxide mask
Deposit oxide thick=0.04

# Patterning of oxide mask
Etch oxide start x=0.35 y=0.00
Etch cont x=0.35 y=-0.04
Etch cont x=0.65 y=-0.04
Etch done x=0.65 y=0.00

# saving structure
Struct outfile=ox_mask.str

# Plotting structure
Tonyplot    ox_mask.str

# Pre-deposition diffusion of boron atoms
Method compress init.time=0.10 fermi
```

```
Diffuse time=30 temp=1000 press=1.0 c.boron=1.0e15

# Saving structure
Struct outfile=predep.str
Tonyplot    predep.str

# Drive-in diffusion for 40min at 1000 C
Method compress init.time=0.1 fermi
Diffuse time=40 temp=1000 dryo2 press=1.0 hcl.pc=0

#
Struct outfile=dirve_in.str
Tonyplot    dirve_in.str

quit
```

仿真结束后，Tonyplot 自动运行，工艺各阶段的截面图如下图所示。



刻蚀氧化层后的扩散窗口(ox_mask.str)



恒定表面浓度扩散后的杂质浓度分布(predep.str)



推进扩散后的截面图形和杂质浓度分布(drive_in.str)

读者可自行更改脚本中扩散时间、温度和压力等参数，观察不同参数对扩散结深和杂质分布的影响。

练习：1、保持其他参数不变，比较推进时间 t=60min 和 t=90min 时的杂质分布曲线；

2、保持其他参数不变，将衬底掺杂类型变换为 P 型，将掺杂剂更改为磷时，观测扩散后的杂质分布和结深。

3、保持其他参数不变，改变推进扩散温度和时间，观测如下图所示的杂质分布。



实验五 离子注入

一、实验目的：熟悉离子注入工艺过程，理解注入能量、注入剂量、倾斜角等等因素对离子注入分布的影响

二、实验内容：1、注入能量对离子分布的影响；

2、注入剂量对离子分析的影响；

3、不同倾斜角时的杂质浓度分布再分布。

二、实验步骤：

步骤一 建立 X/Y 方向长为 $10\times1\mu m^2$ 的仿真网格

步骤二 硅单晶衬底、晶向<100>、P 掺杂浓度 $1.0\times10^{14}cm^{-3}$

步骤三 积淀 0.05μm 厚的氮化硅

步骤四 刻蚀氮化硅，形成局部氧化窗口

步骤五 局部氧化

步骤六 去除氮化硅

步骤七 以氧化硅为掩膜，实现自对准 B 注入，注入剂量 $7.0*10^{19}cm^{-3}$，注入能量 40eV，倾斜角为 0

步骤七 热退火，杂质再分布和晶格恢复

步骤八 显示注入后的杂质分布

仿真文件如下：

```
### ** This example shows a localized implantation stage of wafer fabrication ** ###
# Start Process Simulator "ATHENA"
go athena

#Define mesh for X plane
line x loc=0.00    spac=0.10
line x loc=10.00 spac=0.10
```

```
#Define mesh for y plane
line y loc=0.00    spac=0.03
line y loc=0.30    spac=0.02
line y loc=0.60    spac=0.04
line y loc=1.00    spac=0.05

#Initialisation of Si Sub of crystal orientation of 100 with Phosphorus dopants added
init silicon c.phosphor=1.0e14 orientation=100 two.d space.mult=2.0

#
Deposit nitride thick=0.05

#
Etch nitride start x=0.0 y=0.0
Etch cont x=0.0 y=-0.05
Etch cont x=2.0 y=-0.05
Etch done x=2.0 y=0.0

#

Etch nitride start x=8.0 y=0.0
Etch cont x=8.0 y=-0.05
Etch cont x=10.0 y=-0.05
Etch done x=10.0 y=0.0

# saving structure
Struct outfile=nitride_mask.str

# Plotting structure
Tonyplot nitride_mask.str

#
Diffuse time=90 temp=900 weto2 press=3.0 hcl.pc=0

#
Etch nitride all

#
Implant boron dose=7.0e16 energy=40 tilt=0 rotation=0 crystal gauss lat.ratio1=1.0 \
lat.ration2=1.0

#Saving structure
struct outfile=implant1.str
tonyplot implant1.str

# redistribution
Diffuse time=20 temp=1000 nitro press=1.0

#Saving structure
struct outfile=implant2.str
tonyplot implant1.str

quit
```

仿真结束后，Tonyplot 自动运行，工艺各阶段的截面图如下图所示。



离子注入后的截面图



离子注入后的杂质浓度分布

实验六 CMOS 工艺集成

一、实验目的：熟悉 N-WELL CMOS 工艺的关键制备过程

二、实验内容：根据 CMOS 器件的工艺流程，逐步运行下列工艺步骤的脚本文件，并观察器件的表面形貌和内部参数。

三、实验步骤：

在工艺步骤之前先建立仿真网格（X: 20μm, Y:5μm），然后初始化硅衬底（晶向<100>，B 掺杂@$4\times10^{16}cm^{-3}$）

步骤一 生长清洁氧化层，目的是清除表面沾污，为器件制备创建一个洁净区（详见课本图 2-19）。

步骤二 去除清洁氧化层。

步骤三 淀积氧化物掩膜，厚度 1.50μm。

步骤四 旋涂 0.5μm 厚的光刻胶

步骤五 根据光刻板上的几何图形窗口【坐标 (11,-1.195) (11,-2.0) (17,-2.0) (17,-1.495)】刻蚀光刻胶

步骤六 以光刻胶为掩膜刻蚀下层氧化硅，刻蚀坐标(11,-1.495) (11,0.005) (17,0.005) (17,-1.495)，去除光刻胶

步骤七 N-well P^+离子注入

步骤八 N-well 离子注入推进与退火，抽取 N-well 结深

步骤九 去除氧化物掩膜

步骤十 淀积氮化物

步骤十一 刻蚀氮化物窗口，以便在窗口区生长场氧隔离区

步骤十二 以干-湿-干氧化方式顺序生长场氧化层，抽取场氧厚度

步骤十三 去除氮化物掩膜

步骤十四 常压低温（900℃）、干氧生长薄栅氧，为了提高栅氧质量，通入 HCl 气体

步骤十五 淀积氮化物掩膜

步骤十六 刻蚀左侧氮化物，暴露 NMOS 器件区域，以便 B^+离子注入调整 NMOS 器件的阈值电压

步骤十七 B^+离子注入，调整 NMOS 器件阈值电压

步骤十八 去除氮化物掩膜

步骤十九 淀积氮化物掩膜

步骤二十 刻蚀右侧氮化物，暴露 PMOS 器件区域，以便 BF_2 离子注入调整 PMOS 器件的阈值电压

步骤二十一 BF_2 离子注入，调整 PMOS 器件的阈值电压

步骤二十二 去除氮化物掩膜

步骤二十三 淀积 NMOS 多晶硅栅(n-type, $N=1\times10^{21}cm^{-3}$)

步骤二十四 NMOS 多晶硅栅的刻蚀

步骤二十五 淀积氮化物掩蔽牺牲层

步骤二十六 刻蚀右侧氮化物掩蔽牺牲层，暴露 PMOS 器件区域

步骤二十七 淀积 PMOS 多晶硅栅(p-type)

步骤二十八 PMOS 多晶硅栅刻蚀

步骤二十九 以氮化物掩蔽牺牲层和 PMOS 多晶硅栅为掩膜，注入 PMOS 源/漏区

步骤三十 去除氮化物掩蔽牺牲层

步骤三十一 淀积氮化物掩蔽牺牲层

步骤三十二 刻蚀左侧氮化物掩蔽牺牲层，暴露 NMOS 器件区域

步骤三十三 NMOS 源/漏注入

步骤三十四 去除氮化物掩蔽牺牲层

步骤三十五 注入后退火，在激活杂质的同时消除晶格损伤

步骤三十六 淀积磷硅玻璃

步骤三十七 刻蚀 NMOS 和 PMOS 器件的源、漏接触孔

步骤三十八 淀积 Al 电极

步骤三十九 Al 淀积反刻

仿真文件如下：

```
# Fabrication of an N-well CMOS inverter #
# start ATHENA
go athena

# Define mesh for X and Y coordinates
line x loc=0.0 spac=0.2
line x loc=20.00 spac=0.2

line y loc=0.0 spac=0.2
line y loc=1.0 spac=0.1
line y loc=2.0 spac=0.2
line y loc=5.0 spac=0.4

#Initialization of silicon sub. of orientation of 100 with B doping
init silicon c.boron=4.0e16 orientation=100 two.d space.mult=2.0

# save the initiazed mesh structure of silicon substrate
struc outfile=init.str

# Step1_Grow a cleaning oxide, the added HCl is to creat a thick cleaning region
method compress
diffuse time=20 temp=900 dryO2 press=1.0 hcl.pc=3.0

# save the structure of cleaning oxde
struc outfile=step1.str

# extract the thickness of cleaning oxide
extract name="Tox" thickness oxide mat.occno=1 x.val=10

# Step2_Remove the cleaning oxide
etch oxide all

# save the structure of step2
struc outfile=step2.str

# ********************************************************************
### *** the following steps(3-9) is to creat N-well ***
```

```
# Step3_deposit oxide-mask
deposit oxide thick=1.5 division=15
# save the structure of step3
struc outfile=step3.str

# Step4_Spin on the coat of photoresist
deposit photoresist thick=0.5 division=5

# save the structure of step4
struc outfile=step4.str

# Step5_Patterning of photoresist
etch photo start x=11.00 y=-1.495
etch cont    x=11.00 y=-2.0
etch cont    x=17.00 y=-2.0
etch done    x=17.00 y=-1.495

# save the structure of patterning of photoresist
struct outfile=step5.str

# Step6_etching the window for N-well implant and remove the photoresist
etch oxide start x=11.00 y=-1.495
etch cont    x=11.00 y=0.005
etch cont    x=17.00 y=0.005
etch done    x=17.00 y=-1.495

etch photoresist all

# save the structure of N-well window
struc outfile=step6.str

# Step7_ion implant for N-well
implant phosphor dose=3.0e15 energy=100 tilt=0 rotation=0 crystal gauss

# extract the junction depth of P implant
extract name="xj_implant" xj material="Silicon" mat.occno=1 x.val=14 junc.occno=1

# save the structure of N-well implant
struc outfile=step7.str

# Step8_N-well drive-in and aneal
method compress
diffuse time=100 temp=1150 nitrogen press=0.1

# extract the junction depth of drive-in and aneal
extract name="xj_drive_in" xj material="Silicon" mat.occno=1 x.val=14 junc.occno=1

# save the structure of implant drive-in
struc outfile=step8.str

# Step9_remove the oxide-mask
etch oxide all

# save the structure of Step9
struc outfile=step9.str
# ************************************************************
```




```
# ***********************************************************************
### *** the following steps(10-13) is to grow field oxide ***
# Step10_deposit nitride-mask
deposit nitride thick=0.1 division=10

# save the structure of nitride-mask deposition
struc outfile=step10.str

# Step11_etch the windows for the growth of field oxide
# Etch the leftmost field oxide
etch nitride start x=0.0 y=0.005
etch cont x=0.0 y=-0.1
etch cont x=2.0 y=-0.1
etch done x=2.0 y=0.005

# Etch the center fielf oxide
etch nitride start x=8.0 y=0.005
etch cont x=8.0 y=-0.1
etch cont x=12.0 y=-0.1
etch done x=12.0 y=0.005

# Etch the rightmost field oxide
etch nitride start x=18.0 y=0.005
etch cont x=18.0 y=-0.1
etch cont x=20.0 y=-0.1
etch done x=20.0 y=0.005

# save the structure of field_oxide windows
struc outfile=step11.str

# Step12_grow the field oxide
method compress
diffuse time=10 temp=900 T.rate=20 dryO2 press=1.0 hcl.pc=1
```

```
diffuse time=70 temp=1100 wetO2 press=1.0 hcl.pc=1
diffuse time=10 temp=1100 T.rate=-20 dryO2 press=1.0 hcl.pc=1

# Extract the thickness of field oxide
extract name="Tox_field oxide" thickness oxide mat.occno=1 x.val=10
# Extract the junction depth of N-well after field oxide growth
extract name="xj_N_well" xj material="Silicon" mat.occno=1 x.val=15 junc.occno=1

# save the structure of field_oxide with Nitride_mask
struc outfile=step12.str

# Step13_Nitride removal
etch nitride all

# save the structure of field_oxide
struc outfile=step13.str
# ********************************************************************
```




```
# Step14_Gate Oxide Growth
diffuse time=20 temp=900 dryO2 press=1.0 hcl.pc=3

# Extract Gate oxide thickness
extract name="Tox_gate" thickness oxide mat.occno=1 x.val=5

# save the structure of Gate_oxide
struc outfile=step14.str

# ***************** Threshold adjustment for NMOS ******************
# Step15_deposit Nitride_mask
deposit nitride thick=0.3 division=15

# save the structure of nitride_mask for threshold adjustment for NMOS
struc outfile=step15.str

# Step16_Nitride mask patterning
etch nitride left p1.x=7.455

# save the structure of nitride_mask window for threshold adjustment
struc outfile=step16.str
```

```
# Step17_Vt adjustment for NMOS
implant boron dose=1.2e12 energy=10 tilt=0 rotation=0 crystal gauss

# save the structure of threshold adjustment for NMOS
struc outfile=step17.str

# Step18_Nitride remaval
etch nitride all

# save the structure of nitride_mask removal
struc outfile=step18.str

# ************** Threshold adjustment for PMOS ************************
# Step19_deposit Nitride_mask for PMOS Vt adjustment
deposit Nitride thick=0.3 division=15

# Step20_Nitride mask patterning for exposure of PMOS
etch nitride right p1.x=12.545

# save the structure of nitride_mask window for exposure of PMOS
struc outfile=step20.str

# Step21_BF2 implant for Vt adjustment of PMOS
implant BF2 dose=2.0e11 energy=20 tilt=0 rotation=0 amorph gauss

# save the structure of BF2 implant for Vt adjustment of PMOS
struc outfile=step21.str

# Step22_Nitride removal
etch nitride all

# save the structure of PMOS BF2 implant for Vt adjustment
struc outfile=step22.str

# Step23_NMOS poly-Si gate deposition (doped to reduce resisitivity)
deposit poly thick=0.04 division=1 c.phosphor=1.0e21

#
struc outfile=step23.str

# Step24_NMOS gate patterning
etch poly right p1.x=5.5
etch poly left p1.x=4.5

# save the strucutre of gate paterning of NMOS
struc outfile=step24.str

# Step25_Sacrificial Nitride mask deposition for withstand the implant of PMOS S/D
deposit nitride thick=0.5 division=5
#
struc outfile=step25.str

# Step26_Etch Sacrificial Nitride for exposure the region of PMOS
etch nitride right p1.x=10
#
struc outfile=step26.str
```

```
# Step27_PMOS poly_Si gate deposition (doped B)
deposit poly thick=0.04 division=1 c.boron=1.0e21
#
struc outfile=step27.str

# Step28_PMOS gate patterning
etch poly right p1.x=15.5
etch poly left p1.x=14.5
#
struc outfile=step28.str

# Step29_PMOS S/D implant
implant BORON dose=1.0e18 energy=5 tilt=7 rotation=0 crystal gauss
#
struc outfile=step29.str

# Step30_Sacrificial Nitride removal
etch nitride all
#
struc outfile=step30.str

# Step31_Nitride deposition
deposit nitride thick=0.5 division=5
#
struc outfile=step31.str

# Step32_Nitride etching for exposure of NMOS
etch nitride left p1.x=10
#
struc outfile=step32.str

# Step33_NMOS S/D implant
implant arsenic dose=1.0e19 energy=5 tilt=7 rotation=0 crystal gauss
#
struc outfile=step33.str

# Step34_Nitride removal
etch nitride all
#
struc outfile=step34.str

# Step35_Aneal to activate impurity and
diffuse time=5 temp=600 nitrogen press=1

# Extract the S/D junction depths of NMOS and PMOS respectively
extract name="xj_NMOS" xj material="Silicon" mat.occno=1 x.val=3.5 junc.occno=1
extract name="xj_PMOS" xj material="Silicon" mat.occno=1 x.val=13.5 junc.occno=1

# save the structure of S/D region
struc outfile=step35.str
```




```
# Step36_deposit P-doped glass and reflow
deposit oxide thick=0.05 division=5 c.phosphor=1e21
#
struc outfile=step36.str

# Step37_etch the contact hole for S/D region
# NMOS Source contact definition
etch oxide start x=3 y=0.0085
etch cont x=3 y=-0.055
etch cont x=4 y=-0.055
etch done x=4 y=0.0085

# NMOS Drain contact definition
etch oxide start x=6 y=0.0085
etch cont x=6 y=-0.055
etch cont x=7 y=-0.055
etch done x=7 y=0.0085

# PMOS Source contact definition
etch oxide start x=13 y=0.0085
etch cont x=13 y=-0.055
etch cont x=14 y=-0.055
etch done x=14 y=0.085

# PMOS Drain contact definition
etch oxide start x=16 y=0.0085
etch cont x=16 y=-0.055
etch cont x=17 y=-0.055
etch done x=17 y=0.085

# save the structure of contact hole
struc outfile=step37.str
```

```
# Step38_First metal deposition
deposit alumin thick=0.1 division=2
#
struc outfile=step38.str

# Step39_Al Elctrode etching
etch aluminum start x=4.5 y=-0.092
etch cont x=4.5 y=-0.194
etch cont x=5.5 y=-0.194
etch done x=5.5 y=-0.092

etch aluminum start x=14.5 y=-0.092
etch cont x=14.5 y=-0.194
etch cont x=15.5 y=-0.194
etch done x=15.5 y=-0.092

# Isolate the NOMS and PMOS for simulation purpose
#etch aluminum start x=9.2 y=-0.462
#etch cont x=9.2 y=-0.564
#etch cont x=10.8 y=-0.564
#etch done x=10.8 y=-0.462

# save the structure of Al electrode
struc outfile=step39.str

# Step40_nitride passivation deposition
#deposit nitride thick=0.3 division=3
#
#struc outfile=step40.str

# Step41_Pad hole definition
#etch nitride start x=0.0 y=-0.562
#etch cont x=0.0 y=-0.863
#etch cont x=2.0 y=-0.863
#etch done x=2.0 y=-0.562

#etch nitride start x=18.0 y=-0.562
#etch cont x=18.0 y=-0.863
#etch cont x=20.0 y=-0.863
#etch done x=20 y=-0.562
#
#struc outfile=step41.str

# Defining electrodes for NMOS gate, source & drain
electrode name=source x=3.5
electrode name=gate x=5.0
electrode name=drain x=6.5

# Extract of Vt of NMOS device
extract name="Vt_NMOS" 1dvt ntype x.val=5 qss=1e10

# Defining electrodes for PMOS gate, source & drain
electrode name=drain x=13.5
electrode name=gate x=15
electrode name=source x=16.5
```

```
# Electrode for CMOS
electrode name=subtrate backside

# Extract of Vt of pMOS device
extract name="Vt_pMOS" 1dvt ptype x.val=15 qss=1e10

# save the final CMOS structure
struc outfile=CMOS_final.str

#
tonyplot -add step38.str CMOS_final.str

quit
```



实验七 NMOS 器件性能仿真

一、实验目的：本模拟实验实现 CMOS 双阱工艺中一部分的 NMOS 管的制备工艺，并且通过器件仿真软件 ATLAS 模拟器件的转移特性和 $I_d \sim V_{ds}$ 输出特性曲线。

二、实验步骤：

仿真文件如下：

```
############     Process simulation    ###################
go athena

# Define the mesh for X and Y coordinates
line x loc=0.0 spac=0.1
line x loc=0.2 spac=0.006
line x loc=0.4 spac=0.006
line x loc=0.6 spac=0.01
#
line y loc=0.0 spac=0.002
line y loc=0.2 spac=0.005
line y loc=0.5 spac=0.05
line y loc=0.8 spac=0.15

# initiate the Si substrate
init silicon orientation=100 c.phos=1e14 two.d space.mul=2

# ******* P-well formation **********************
# First time oxidation for pad oxide
diffus time=30 temp=1000 dryo2 press=1.00 hcl=3
#
etch oxide thick=0.02
#
# P-well Implant
#
implant boron dose=8e12 energy=100 pears

# P-well drive-in and aneal
diffus temp=950 time=100 weto2 hcl=3
#
# N-well implant not shown -
#
# well drive-in and aneal starts here
diffus time=50 temp=1000 t.rate=4.000 dryo2 press=0.10 hcl=3
diffus time=220 temp=1200 nitro press=1
diffus time=90 temp=1200 t.rate=-4.444 nitro press=1

# etch the oxide coating
etch oxide all

# sacrificial "cleaning" oxide
diffus time=20 temp=1000 dryo2 press=1 hcl=3
```

附录:常用工艺仿真命令

说明：本文件仅给出了本课程仿真实验中用到的部分仿真命令及工艺参数，对于想深入了解工艺命令的读者，请查阅SILVACO ATHENA User's Manual的Statements部分。

LINE

LINE specifies a mesh line during grid definition.

Syntax

LINE

X|Y LOCATION=<n> [SPACING=<n>] [TAG=<c>] [TRI.LEFT|TRI.RIGHT]

关键参数说明

X|Y ——指明水平或垂直方向的网格线

LOCATION=<n> ——指明沿指定坐标轴方向网格线的位置

SPACING=<n> ——指明网格间隔，microns

Example

```
#Define mesh for X plane
line x loc=0.00    spac=0.10
line x loc=10.00 spac=0.10

#Define mesh for y plane
line y loc=0.00    spac=0.03
line y loc=0.30    spac=0.02
line y loc=0.60    spac=0.04
line y loc=1.00    spac=0.05
```

GO

GO starts the simulator. Each ATHENA input file should begin with a GO statement.

Example

```
Go Athena
```

Go Atlas

Go SSUPREM3

INITIALIZE

INITIALIZE specifies the initial starting material and background doping levels.

Syntax

INITIALIZE

[MATERIAL] [ORIENTATION=<n>] [ROT.SUB=<n>] [C.FRACTION=<n>]

[C.IMPURITIES=<n>|RESISTIVITY=<n>][C.INTERST=<n>][C.VACANCY=<n>]

[BORON|PHOSPHORUS|ARSENIC|ANTIMONY] [NO.IMPURITY]

[ONE.D|TWO.D|AUTO] [X.LOCAT=<n>] [CYLINDRICAL]

[INFILE=<c>] [STRUCTURE|INTENSITY]

[SPACE.MULT=<n>] [INTERVAL.R=<n>] [LINE.DATA] [SCALE=<n>] [FLIP.Y]

[DEPTH.STR=<n>] [WIDTH.STR=<n>]

Description

This command sets up the mesh from either a rectangular specification or from a previous structure file. The statement also initializes the background doping concentration in all regions.

关键参数说明

MATERIAL ——指明初始化材料

ORIENTATION=<n> ——指明衬底材料晶向(100)、(110) & (111)

C.IMPURITIES=<n> ——指明掺杂类型及浓度

ONE.D|TWO.D|AUTO ——指明进行一维还是二维仿真，或者根据工艺流程自动选择。如果选择1D计算，则需要 **X.LOCAT=<n>** 参数指定1D计算的具体位置。

SPACE.MULT=<n> ——为**LINE**语句定义的仿真网格指明倍增器，以加快仿真速度

Example

init silicon orientation=100 c.phosphor=1.0e14 two.d space.mult=1.0

DIFFUSE

DIFFUSE — runs a time temperature step on the wafer and calculates oxidation, silicidation and diffusion of impurities.

Syntax

DIFFUSE

TIME=<n> [HOURS|MINUTES|SECONDS]

TEMPERATURE=<n> [T.FINAL=<n>|T.RATE=<n>]

[DRYO2|WETO2|NITROGEN|INERT] [HCL.PC=<n>] [PRESSURE=<n>]

[F.02=<n>|F.H2=<n>|F.H20=<n>|F.N2=<n>|F.HCL=<n>]

[C.IMPURITIES=<n>]

[DUMP][DUMP.PREFIX=<c>] [TSAVE=<n>] [TSAVE.MULT=<n>]

[B.MOD=<c>] [p.MOD=<c>] [AS.MOD=<c>] [IC.MOD=<c>] [VI.MOD=<c>]

[NO.DIFF] [REFLOW]

Description

This command specifies diffusion and/or oxidation/silicidation steps. Any impurities present in the wafer are diffused if they have non-zero diffusivities. The oxidation and diffusion control parameters are contained in the associated METHOD, OXIDE, and SILICIDE statements. Default coefficients are in the ATHENAMOD file available from the DeckBuild Commands menu under Models. To change model coefficients, refer to the appropriate IMPURITY statement for information.

关键参数说明

Time ——以 Hours/Minutes/Seconds 为单位的氧化或扩散时间，默认单位为 MIN

Temperature ——以℃为单位的炉温，该数值因设定在 700~1200℃范围内,超出此范围仿真结果不准确

T.final ——氧化或扩散过程中的最终炉温

T.rate ——从 Temperature 到 T.final 温度间的温升速度℃/min

DryO2 ——干氧

WetO2 ——湿氧

Nitrogen ——氮气

Inert ——惰性气氛

HCl.pc ——HCl 的百分含量

Pressure ——以大气压为单位的活性粒子的分压

[F.02=<n>|F.H2=<n>|F.H20=<n>|F.N2=<n>|F.HCL=<n>] ——氧化或扩散氛围中氧气、氢气、水汽及HCl组分的相对流量

[C.IMPURITIES=<n>] ——预扩散过程中的杂质浓度

如 C.Boron=1.0e20 atoms/cm^3　C.Phosphor

Oxidation Example

Diffuse time=30 Temp=900 Dryo2 Press=1 T.final=1100 T.rate=20

Predeposition Example

Diffuse time=30 Temp=1000 C.Boron=1.0e20

EXTRACT

Extract命令用于抽取工艺和器件的物理参数和电学特性。可以按以下格式输入，也可在Deckbuild中的Commond/Insert extract插入抽取命令，并修改相关参数。

Syntax

EXTRACT
[name=<QSTRING>][<Value_Type>|<Curve_Type>][<MATERIAL>][mat.occno=<EXPR>][
y.val=<EXPR>|x.val=<EXPR>|region=<QSTRING>]

Example

1、抽取氧化层厚度

Extract name=”gateox” thickness oxide mat.occno=1 x.val=0.5

2、抽取结深

Extract name=”nxj” xj silicon mat.occno=1 x.val=0.5 junc.occno=1

3、抽取扩散区的方块电阻

Extract name=”Sheet rho” sheet.res material=”silicon” mat.occno=1 x.val=0.5 region.occno=1

ETCH

ETCH simulates an etch process.

Syntax

ETCH
[MATERIAL] [NAME.RESIST]
[ALL|DRY] [THICKNESS=<n>] [ANGLE=<n>] [UNDERCUT=<n>]
[LEFT|RIGHT|ABOVE|BELOW] [P1.X=<n>] [P1.Y=<n>] [P2.X=<n>] [P2.Y=<n>]

[START|CONTINUE|DONE] [X=<n>][Y=<n>]

[INFILE=<c>] [TOP.LAYER] [NOEXPOSE]

[MACHINE=<c>] [TIME=<n>] [HOURS|MINUTES|SECONDS]

[DT.FACT=<n>] [DT.MAX=<n>] [DX.MULT=<n>]

[MC.REDEPO] [MC.SMOOTH = <n>] [MC.DT.FACT = <n>] {MC.MODFNAME = <c>]

Description

ATHENA provides two different etch simulation methods. The first is geometrical etching available within any ATHENA module. The second is physical etching available only in ELITE.

关键参数说明

Material ——指明要刻蚀的材料

Name.resist ——指明刻蚀的光刻胶类型

All ——指明刻蚀全部的指定材料

Thickness ——指明刻蚀厚度，单位：microns

Left/Right/Above/Below ——提供梯形截面刻蚀的快速方法

P1.X/P1.Y/P2.X/P2.Y ——指明**Left/Right/Above/Below**刻蚀的刻蚀界限

Start/Continue/Done ——指明任意形状的刻蚀

X/Y ——指明**Start/Continue/Done**过程中形状的坐标点

Simple Geometrical Etch Example

Etch nitride left p1.x=0.5

Arbitrary Geometrical Shape Etch Example

Etch oxide start x=0.0 y=0.0

Etch continue x=1.0 y=0.0

Etch continue x=1.0 y=1.0

Etch done x=0.0 y=1.0

IMPLANT

IMPLANT specifies an ion implantation process step.

Syntax

IMPLANT

[GAUSS|PEARSON|FULL.LAT|MONTECARLO|BCA] [CRYSTAL|AMORPHOUS]

IMPURITY ENERGY=<n> DOSE=<n> [FULL.DOSE]

[TILT=<n>] [ROTATION=<n>] [FULLROTATION]

[PLUS.ONE] [DAM.FACTOR=<n>] [DAM.MOD=<c>] [PRINT.MOM]

[X.DISCR=<n>] [LAT.RATIO1] [LAT.RATIO2] [S.OXIDE=<n>]

[MATCH.DOSE|RP.SCALE|MAX.SCALE][SCALE.MOM] [ANY.PEARSON]

[N.ION=<n>] [MCSEED=<n>] [TEMPERATURE=<n>] [DIVERGENCE=<n>] [IONBEAMWIDTH=<n>]

[IMPACT.POINT=<n>] [SMOOTH=<n>] [SAMPLING] [DAMAGE] [MISCUT.TH][MISCUT.PH]

[TRAJ.FILE=<n>] [N.TRAJ=<n>]

[Z1 = <n>] [M1 = <n>]

Description

This statement simulates ion implantation using different analytical and Monte Carlo models.

关键参数说明

GAUSS|PEARSON|FULL.LAT|MONTECARLO|BCA ——指明离子注入模型

CRYSTAL|AMORPHOUS ——指明注入材料的晶体类型

IMPURITY ENERGY=<n> DOSE=<n> ——指明注入的杂质种类、注入能量和剂量

TILT=<n> ROTATION=<n> ——指明注入材料的倾斜角度(默认为7°)和旋转角度(默认为30°)

DEPOSIT

DEPOSIT deposits a layer of specified material. DEPOSITION is a synonym for this statement.

Syntax

DEPOSIT

MATERIAL [NAME.RESIST=<c>] THICKNESS=<n>

[SI_TO_POLY] [TEMPERATURE=<n>]

[DIVISIONS=<n>] [DY=<n>][YDY=<n>] [MIN.DY=<n>] [MIN.SPACE=<n>]

[C.IMPURITIES=<n>] [F.IMPURITIES=<n.] [C.INTERST=<n>] [F.INTERST=<n>]

[C.VACANCY=<n>] [F.VACANCY=<n>] [C.FRACTION=<n>] [F.FRACTION=<n>]

[MACHINE=<c>] [TIME=<n>] [HOURS|MINUTES|SECONDS]

[N.PARTICLE=<n>] [OUTFILE=<c>] [SUBSTEPS=<n>][VOID]

关键参数说明

Material ——指明淀积材料

Name.resist ——指明淀积的光刻胶类型

Thickness ——指明淀积层的厚度，microns

Divisions ——指明淀积层的垂直网格间隔数，默认值为1

C.impurities ——指明淀积层的掺杂浓度，cm^{-3}

Deposition Example

Deposit oxide thick=0.1 divisions=4

Deposit oxide thickness=0.1 div=6 C.boron=1.0e20

STRUCTURE

STRUCTURE writes the mesh and solution information, aerial image information, or flips or mirrors the structure. SAVEFILE is a synonym for this statement.

Syntax

STRUCTURE

[OUTFILE=<c>][INFILE=<c>][OPC=<n>]

[FLIP.Y][MIRROR][LEFT|RIGHT|TOP|BOTTOM]

[INTENSITY][MASK][REMOVE.GAS] [SIGE.CONV][TWO.DIM]

Description

This statement writes the entire mesh and solution set to a file. The saved data is from the current set of solution and impurity values.

OUTFILE=<c> ——指明要写的文件名

Example

Structure outfile=test.str

TONYPLOT

Tonyplot starts the graphical post-processor TONYPLOT.

Example

Tonyplot –st myfile.str

Tonyplot –overlay myfile1.str myfile2.str

Tonyplot –add myfile1.str myfile2.str

QUIT

QUIT terminates execution of ATHENA. The EXIT, STOP and BYE statements QUIT statement.

Syntax

QUIT

Description

All statements after a QUIT statement will not be checked or executed.